

**Voss**

**MSX**  
**EI**  
**Manual**  
**Escolar**

**UN LIBRO DATA BECKER**

**EDITADO POR FERRE MORET, S.A.**

### **EL CONTENIDO:**

Escrito para alumnos de los últimos cursos de EGB y de BUP, este libro contiene multitud de programas para resolver problemas y de aprendizaje, descritos de una forma muy completa y fácil de comprender. Proporcionan un aprendizaje intensivo, a la vez que divertido, con el MSX.

Del contenido:

- Teorema de Pitágoras
- Progresiones geométricas
- Escritura cifrada
- Crecimiento exponencial
- Verbos irregulares
- Igualdades cuadráticas
- Movimiento pendular
- Estructura de moléculas
- Repaso de vocablos
- Cálculo de intereses

Una pequeña panorámica sobre los fundamentos del tratamiento informático, un somero repaso a los principales elementos del BASIC y una introducción en los aspectos básicos del análisis de problemas completan el libro.

### **ESTE LIBRO HA SIDO ESCRITO POR:**

Werner Voss, profesor de estadística en la Universidad de Bochum. Multitud de publicaciones dentro del campo de la estadística y del tratamiento informático.

ISBN 84-86437-26-1

Voss

# **MSX EI Manual Escolar**

**UN LIBRO DATA BECKER**  
**EDITADO POR FERRE MORET, S.A.**

## ESTRUCTURACION

=====

0.	Prefacio.....	1
1.	Elementos básicos del lenguaje BASIC.....	7
1.1	Consideración previa .....	7
1.2	Conceptos básicos que se debería conocer.....	8
1.3	Sobre el funcionamiento de los ordenadores.....	12
1.4	Análisis de problemas.....	14
1.5	Impresión de resultados.....	17
1.6	Asignaciones de valores.....	22
1.7	Editor del BASIC.....	25
1.8	Entrada de informaciones.....	26
1.9	Bifurcaciones del programa.....	28
1.10	Bucles de programa.....	32
1.11	La utilización de memorias externas.....	34
1.12	Indicaciones complementarias.....	36
2.	Matemáticas.....	39
2.1	Consideración previa.....	39
2.2	El teorema de Pitágoras.....	41
2.3	M.C.D. y m.c.m.....	50
2.4	Comprobación de números primos.....	59
2.5	Ecuación de segundo grado.....	68
2.6	Número e - base del logaritmo neperiano.....	74
2.7	Cálculo de porcentajes.....	80
2.8	Dado.....	85
2.9	Ejemplos de la exactitud de cálculo.....	92

	Página
3. Química .....	94
3.1 Otras instrucciones BASIC.....	94
3.2 La molécula del agua.....	103
3.3 Ecuación química.....	109
3.4 Cálculos estequiométricos.....	116
3.5 El sistema periódico de los elementos.....	122
4. Física .....	131
4.1 Consideración previa.....	131
4.2 Programación de gráficos.....	132
4.3 Instrucciones BASIC para gráficos .....	133
4.4 El dinamómetro .....	144
4.5 El principio de Arquímedes .....	151
4.6 Movimiento pendular.....	157
4.7 Reproducción óptica.....	163
4.8 Paralelogramo de fuerzas.....	168
4.9 La ley de Ohm.....	174
5. Idiomas .....	179
5.1 Consideración previa y complementos BASIC.....	179
5.2 Verbos ingleses irregulares.....	182
5.3 Vocablos franceses.....	189
5.4 Test de vocablos ingleses.....	196
5.5 Escritura en clave.....	203
6. Biología/Ecología .....	211
6.1 Consideración previa.....	211
6.2 Crecimiento sin control.....	212
6.3 Crecimiento controlado.....	217


	Página
6.4 Contaminación ambiental .....	223
7. Geografía/Historia .....	231
7.1 Consideración previa.....	231
7.2 Fechas históricas.....	232
7.3 Las comunidades autónomas.....	238
7.4 El desarrollo de la población en diversas naciones.....	245
8. Economía.....	255
8.1 Consideración previa.....	255
8.2 Cálculo de intereses.....	256
8.3 Amortización de hipoteca.....	262
8.4 Media aritmética.....	269
8.5 Distribución de frecuencias.....	273
8.6 El poder económico de las regiones de la R.F.A....	282
9. Matemáticas II : Gráfico de alta resolución.....	291
9.1 consideración previa .....	291
9.2 Instrucciones para gráficos.....	293
9.3 La recta .....	296
9.4 El círculo.....	299
9.5 La curva senoidal.....	303

	Pagina
9.6 Tabla matemática 1.....	306
9.7 Tabla matemática 2.....	309
9.8 Tabla matemática 3.....	313
9.9 Elipse.....	316
9.10 Distribución normal según Gaus.....	320
9.11 Regresión.....	324
9.12 Permutaciones.....	332
9.13 Combinaciones.....	336
9.14 Lotería.....	342
9.15 Distribución binomial.....	347
9.16 Solución de ecuaciones.....	358
9.17 Elecciones.....	362
10. Epílogo.....	367
Índice general.....	369

Voss	Capítulo	0	:	Prefacio	Página
	Punto	-	:	-	1
Colegio					

Prefacio

=====



La llegada al mercado de los microordenadores domésticos dotados del sistema operativo MSX, ha suscitado un indudable interés. El MSX brinda la posibilidad - ya de por sí interesante -, de que toda una variedad de microordenadores de diferentes marcas puedan ejecutar los mismos programas. Pero no solo el importante hecho de la compatibilidad atrae nuestra atención. El intérprete BASIC de que están dotados estos microordenadores, ha sido desarrollado por la firma norteamericana MICROSOFT, lo cual es de por sí una garantía de la calidad del mismo. El BASIC-MSX es potente, el número de instrucciones que soporta es elevado, y por ello su aprendizaje puede parecer algo complicado. De hecho, no lo es en absoluto, como veremos.

El usuario de un microordenador doméstico debe orientar su utilización más allá de lo lúdico. Utilizar un ordenador solo para jugar es infrutilizarlo. Por muy fascinantes que puedan ser los programas elaborados - juegos de fútbol, guerras espaciales, carreras de coches, partidas de ajedrez, luchas contra extraños monstruos, etc.- no hay nada más excitante que ser uno mismo el que programa. Quien analiza por primera vez un problema y consigue que el programa introducido lo resuelva en cuestión de segundos y sin errores, puede congratularse. Un logro de este tipo prueba la utilidad del ordenador, es decir, que nos libere del pesado trabajo mental, siendo ésta su principal aplicación, que determina su creciente importancia.

Voss	Capítulo	0 : Prefacio	Página
	Punto	- : -	2
Colegio			

Por este motivo, en el presente libro queremos enseñarle al alumno cómo aplicar el ordenador, en concreto un MSX, para resolver los problemas fundamentales de la enseñanza escolar.

En resumidas cuentas, con este libro se persiguen simultáneamente cuatro objetivos:

1. Se toman los problemas escolares más normales, por ejemplo de matemáticas, física u otras asignaturas y se proponen medios para una solución.
2. La descripción de estos medios es presentada al lector de forma que pueda aplicarlos al ordenador.
3. Se informará al lector sobre la utilización del lenguaje de programación BASIC destinado a aquellos programas necesarios para solucionar los problemas antes mencionados.
4. De esta forma, el lector dispondrá de una colección de programas, que le será útil en la escuela.

Comprenderá el lector que la utilidad de este libro se hace patente especialmente cuando prueba él mismo los programas presentados al igual que cuando ensaya con modificaciones de éstos o con otros programas.

Voss	Capítulo	0 : Prefacio	Página
	Punto	- : -	3
Colegio			

Como en muchos otros ámbitos, dentro del campo de la programación y de la utilización de ordenadores se cumple lo siguiente:

'' La práctica hace al maestro ''

En los capítulos de este libro, se cubren los siguientes campos:

1. Problemas de matemáticas
2. Problemas de química
3. Problemas de física
4. Ejercicios de lenguas
5. Problemas de biología/ecología
6. Problemas de geografía e historia
7. Problemas de economía
8. Otros problemas

En estos capítulos se contienen algunas explicaciones acerca de las bases y condiciones de aplicación del ordenador y se discuten, también, los elementos básicos del lenguaje de programación BASIC.

Los lectores que posean suficientes conocimientos de BASIC pueden prescindir de estas explicaciones.

Debemos hacer la siguiente aclaración con respecto a los programas que desarrollamos y presentamos en los capítulos siguientes:

Por una parte, se trata de programas que deben resolver determinados problemas (Ejemplo: Programa para calcular un ''máximo común divisor'').

Voss	Capítulo	0 : Prefacio	Página
	Punto	- : -	4
Colegio			

Estos programas pueden "alimentarse" con diferentes datos de partida, y sirven siempre para el mismo fin. Por otro lado, existen también programas sólo de entrenamiento: Por ejemplo, un programa que sirve para aprender vocablos Ingleses, que no resuelve ningún problema, sino que sólo "entrena".

En todo caso, estos programas tienen aplicaciones más generales cuando se proveen de material de partida más variado.

Se están preparando una cassette y un diskette de programas para este libro, donde se encuentran todos los programas aquí discutidos, de modo que el lector que quiera utilizarlos se pueda ahorrar el trabajo de teclearlos.

La siguiente observación es de gran importancia para las próximas explicaciones:

En los programas donde se resuelven los problemas del colegio hemos prescindido conscientemente de utilizar los elementos más avanzados y complejos del lenguaje BASIC-MSX. Por el contrario, nos hemos esforzado en utilizar los elementos básicos y sencillos del BASIC. El motivo es evitar sobrepasar, tanto las posibilidades del lector como las del principiante, en materia de programación y conseguir, de este modo, que tenga la ocasión de centrarse en los problemas.

Esta limitación tiene como consecuencia el que algunos programas parezcan demasiado simples pero, aun con los medios más sencillos funcionan y éste era nuestro objetivo.

Voss	Capítulo	0 : Prefacio	Página
	Punto	- : -	5
Colegio			

No obstante, los programas más complejos de los últimos capítulos, requieren la aplicación de otros elementos sintácticos, que serán tratados en caso concreto.

Rogamos también al lector tenga en cuenta que algunos de los programas presentados figuran simplemente a modo de ejemplos:

Si, por ejemplo, en un programa de repaso de vocablos sólo se prevén 10, el lector, para poder utilizarlo de forma concreta, deberá ampliar primero la lista de éstos. Este procedimiento naturalmente no modifica la estructura básica del programa - que es lo que a fin de cuentas importa-.

Voss	Capítulo	1	: BASIC	Página
	Punto	1	: Consideración	7
Colegio			previa	



## Capítulo 1 : Elementos básicos del lenguaje BASIC

---

### 1.1 Consideración previa

---

En este capítulo, se presentan y se explican además de algunos conceptos básicos del proceso de datos, que volverán a aparecer más tarde, los elementos básicos más importantes del lenguaje BASIC.

Aquellos lectores que ya conozcan suficientemente el BASIC pueden saltarse este capítulo.

Sin embargo, los lectores que no tengan ninguna experiencia como programadores pensarán que las explicaciones de este capítulo son algo someras. A éstos les diremos que en los próximos capítulos se aclarará la utilización, el funcionamiento y el efecto de estos elementos sintácticos mientras se aplican de forma práctica.

Visto de este modo, este capítulo contiene sólo una Introducción.



Voss	Capítulo	1 : BASIC	Página
	Punto	2 : Conceptos básicos	8
Colegio			

## 1.2 Conceptos básicos que deberían conocerse

Un ordenador moderno no sólo sabe calcular. Sabe, asimismo, manipular letras, palabras y textos y puede, entre otras cosas, generar representaciones gráficas, producir música así como ejercer tareas de control. Ello significa que los ordenadores, que hoy en día ya pueden adquirirse por un precio bastante módico, no sólo disponen de la misma capacidad de cálculo que un ordenador grande de los años sesenta, que era económicamente inaccesible, sino que también su aplicación es más universal.

**Datos** : Bajo "datos" puede entenderse en el sentido más amplio la palabra "informaciones". Concretamente puede tratarse de:

- cifras y números
- valores
- letras
- símbolos (caracteres especiales)
- palabras y textos

**Proceso** : Bajo "proceso de datos" se entienden los procedimientos (frecuentemente cálculos), que sirven para registrar estos datos, para almacenarlos, para valorarlos o analizarlos y para ofrecer resultados.

Cuando el proceso de datos en la forma descrita no se realiza manualmente, sino con la ayuda de ordenadores, es necesario conocer otros conceptos nuevos:

Voss	Capítulo	1 : BASIC	Página
	Punto	2 : Conceptos básicos	9
Colegio			

**bit** : Bajo bit "binary information digit" se entiende la unidad de información más pequeña (ver antes : "datos" = "informaciones"). La unidad de información más pequeña se llama bit, porque un bit puede tener únicamente dos contenidos distintos. Visto desde la perspectiva del fabricante de ordenadores esto significa: un bit puede almacenar sólo dos contenidos de información. Estos contenidos se denominan usualmente 0 y 1.

**byte** : El byte es la reunión de una serie de bits. Normalmente son ocho los bits que constituyen un byte. De este modo, un byte está en condiciones de almacenar una serie de 8 ceros y/o unos.

**Carác-** : Distinguímos tres grupos de caracteres


ter	1. caracteres numéricos	= cifras
	2. caracteres alfabéticos	= letras
	3. otros caracteres	= caracteres especiales

Generalmente, cada uno de estos caracteres se representa en una serie de 8 ceros y/o unos en el ordenador, es decir, que un byte acepta un carácter.

**Campo** : Una serie de caracteres (p.e. palabras o números) constituyen un campo.

~~X~~ 1 = Caracter

La palabra PRINT, por ejemplo, consta de 5 caracteres, ocupa 5 bytes en el ordenador, y como estos 5 símbolos pertenecen a una misma palabra, constituyen un campo.



Lo mismo ocurre con el número 178, que es almacenado en un campo de tres posiciones (se deduce que también pueden existir campos de una sola posición).

Algunos caracteres

**Variable** : Una variable es una magnitud que puede tomar diversos valores - precisando - un nombre, bajo el cual pueden seleccionarse diversos valores (ver abajo).

Variable = Con un nombre diversos valores

Por ejemplo, bajo el nombre "tamaño" podemos hallar los valores 178, 185, 167 etc., detrás del nombre "número de niños" los valores 1, 2, 0, 4 etc., bajo el nombre "localidad" por ejemplo los valores "Palencia", "Alcántara", "Palafrugell".

**Valor** : Como puede deducirse de la definición anterior, bajo "valor" entendemos una determinada característica que puede adquirir una variable dada. No es necesario que sean números, también puede tratarse de palabras o textos como en el ejemplo de la variable anterior "localidad".

Valor = localidad  
Variable = Palencia Alcántara etc

Cada uno de estos valores ocupa un campo del ordenador (ver arriba). También puede llamarse posición de memoria.

**String** : Llamamos "cadena" a un texto que puede representar un valor de una variable (ver arriba); genéricamente se trata de una cadena de caracteres.

**Tabla** : La totalidad de los valores que puede tomar una variable (números o palabras) se llama "tabla".

**Registro** : Varios <sup>palabras</sup> campos relacionados constituyen un registro. Por ejemplo, todos los datos referentes a una persona (nombre, sexo, edad, domicilio, etc.) forman un registro. En este ejemplo, el registro es una serie de números y/o palabras relacionados entre sí.

Una serie de palabras solas puede también constituir un registro, por ejemplo la instrucción:

```
PRINT "PEPE", "ANA", "MAXIMO"
```

El registro se denomina también "record".

**File** : Varios registros relacionados forman un fichero. Obtendremos un fichero por ejemplo si confeccionamos los registros de todos los empleados de una empresa. Podemos obtener igualmente un fichero si contemplamos todas las instrucciones de un programa en conjunto. El fichero se llama también "file".

Voss	Capítulo	1 : BASIC	Página
Colegio	Punto	1 : Funcionamiento	12

### 1.3 Referente al funcionamiento de los ordenadores

Ya explicamos lo que se entiende bajo "proceso de datos". De ello se deduce qué actividades (funciones) debe realizar un equipo de proceso de datos:

1. Recepción de datos ( entrada de datos)
2. Almacenamiento de datos E entrada
3. Proceso de datos propiamente dicho  
(p.e. cálculos o traslado de textos  
y parecidos) P proceso
4. Salida de datos S salida

La entrada de información, su procesado y su salida hacen necesarios una serie de pasos organizativos, que no es necesario tratar aquí, porque generalmente los resuelve el mismo ordenador. De ello se encarga el llamado sistema operacional.

Así, el ordenador debe obtener nuestras informaciones, para que el proceso de los datos pueda ponerse en marcha y proporcionar resultados.

Las informaciones que debemos darle se clasifican en tres grupos:

1. Los datos a procesar
2. Las instrucciones  
(Pasos, que son necesarios para someter los datos  
introducidos a los procedimientos de valoración  
requeridos)
3. Comandos  
(Informaciones e instrucciones que damos al  
sistema operacional del ordenador; ver arriba)

Voss	Capítulo	1 : BASIC	Página
Colegio	Punto	1 : Funcionamiento	13

De acuerdo con los diferentes grupos de cometidos que debe resolver el ordenador distinguiremos entre:

1. Aparatos de entrada (p.e. el teclado)
2. Ambito del procesado (la llamada unidad central)
3. Aparatos de salida (pantalla o impresora)

Aparte de esto existen también las llamadas memorias externas (o periféricas). Sirven para almacenar a largo plazo los programas y los datos. Generalmente se trata de la cassette y de los llamados diskettes, que se reproducen o leen en los correspondientes aparatos reproductores.


#### 1.4 Análisis de problemas

Resolver un problema concreto con ayuda de un ordenador sólo es posible si se ha hallado de antemano su solución. El lector deberá recordar siempre esta evidencia. Esto quiere decir que la solución, a pesar de la modernidad de los ordenadores, sólo puede hallarse mentalmente y no en la máquina; el ordenador únicamente se ocupa de realizar los pasos necesarios.

Para que un ordenador pueda efectuar esta tarea debe indicársele el procedimiento de resolución en forma de instrucciones individuales. Esto lleva consigo que deba descomponerse mentalmente el problema a resolver en una sucesión lógica de pasos individuales - este procedimiento se denomina análisis del problema.

En un análisis de problema resulta útil representar la sucesión de pasos gráficamente en forma de un diagrama de flujo.

Veamos el siguiente ejemplo ilustrativo:



Fuente : W. VOSS : BASIC aprendido rápida- y fácilmente con el alphaTronic PC, Darmstadt 1984, (Heim-verlag), pag. 8.

Voss	Capítulo	1 : BASIC	Página
	Punto	4 : Análisis de	16
Colegio		problemas	

En los capítulos siguientes utilizaremos una y otra vez el instrumento "análisis del problema", porque de este modo, la programación, es decir, la redacción de las secuencias de instrucciones para el ordenador, se facilita considerablemente, evitándose los errores de programación.

Voss	Capítulo	1 : BASIC	Página
	Punto	5 : Visualización	17
Colegio		de resultados	

### 1.5 Visualización de resultados

---

Se le denomina programa a una sucesión de instrucciones que hacen que un ordenador resuelva por sí mismo un problema dado. Debe ser escrito en un lenguaje que el ordenador pueda comprender. Este es el BASIC.

BASIC =  
 Beginner's  
 All-Purpose  
 Symbolic  
 Instruction  
 Code

Todo programa BASIC consta de una sucesión de instrucciones (statements) que deben ser introducidos línea a línea en el ordenador. Una línea se llama sentencia.

Regla 1: Cada sentencia BASIC requiere un número.

Regla 2: Cada sentencia debe concluirse pulsando la tecla RETURN.

Regla 3: Una sentencia puede contener varias instrucciones, que hay que separar mediante ":".

*Las instrucciones hay que meterlas entre*  
 ( " ) - ( : ) = ( )

Voss	Capítulo	1 : BASIC	Página
	Punto	5 : Visualización	18
Colegio		de resultados	

Instrucción 1:

nn END

La Instrucción END cierra el programa BASIC. Sirve asimismo para separar una subrutina del resto del programa BASIC.

Instrucción 2:

nn PRINT	{ Sin argumento Número Operación Cadena Nombre de variable
----------	--

La Instrucción PRINT trae los resultados a la pantalla. Se distinguen cinco posibilidades:

1. Sin argumento : Se obtiene una línea en blanco
2. Número : Se visualiza el número
3. Operación : Se visualiza el resultado de una operación
4. Cadena : Se visualiza la cadena (cadena de caracteres puesta entre comillas)
5. Nombre de variable : El contenido de este campo de la memoria es visualizado

La palabra PRINT puede abreviarse con un ?.

Voss	Capítulo	1 : BASIC	Página
	Punto	5 : Visualización	19
Colegio		de resultados	

El siguiente programa BASIC utiliza todas estas posibilidades:

```

5 CLS
10 PRINT 3
20 PRINT
30 PRINT 4.8/2
40 ?
50 ? "SUSI"
60 ?
70 ?X
80 END

```

Este programa BASIC sólo se ejecuta si tecleamos el comando RUN:

Comando 1:

RUN

Para el ejemplo anterior, después del comando RUN obtendremos en pantalla los siguientes resultados:

```

3
2.4
SUSI
0

```

Voss	Capítulo	1 : BASIC	Página
Colegio	Punto	5 : Visualización de resultados	20

Regla 4: Los comandos no reciben ningún número de sentencia

Si después de la impresión del resultado deseamos ver nuevamente nuestro programa en pantalla, necesitamos el siguiente comando:

Comando 2:

-----  
LIST

La instrucción PRINT requiere algunas puntualizaciones adicionales:

Las diferentes posibilidades enunciadas en la instrucción 2 pueden utilizarse tanto mezcladas entre sí como en una sola instrucción PRINT.  
En este caso deben separarse mediante comas y puntos .

Regla 5: Si se separan con comas las diferentes impresiones dentro de la instrucción PRINT, la impresión prosigue por el principio de la siguiente zona de la pantalla ( se tabulan 14 espacios ); cuando se utiliza el punto y coma, la impresión se produce justo a continuación.

Regla 6: Por regla general, toda instrucción PRINT produce un salto de una línea. Si la instrucción contiene la palabra PRINT sola, se produce un único salto de línea.

Regla 7: Una coma o un punto y coma al final de una instrucción PRINT reprimen el salto de línea.

Voss	Capítulo	1 : BASIC	Página
Colegio	Punto	5 : Visualización de resultados	21

La instrucción PRINT genera impresiones de resultados, que comienzan en el margen izquierdo de la pantalla. Para cambiar esto puede utilizarse la instrucción LOCATE:

Instrucción 3:

-----  
nn LOCATE S,Z

Esta instrucción mueve el cursor a la columna S y a la línea Z de la pantalla y las impresiones posteriores comienzan allí.

S pueden tomar valores entre 0 y 39, Z puede tomar valores entre 0 y 23.

Para ello debe imaginarse la pantalla dividida en cuarenta columnas y 24 líneas, situándose la línea 0 arriba del todo y la columna 0 a la izquierda.

En el siguiente punto trataremos el concepto del nombre de variable, que ya jugaba un papel en la instrucción PRINT, además hablaremos sobre particularidades de las operaciones realizables.

Para borrar un programa que ya no nos es de utilidad antes de introducir uno nuevo, se requiere el comando siguiente, que borra el contenido de memoria del program y pone a cero todas las variables:

Comando 3:

-----  
NEW

Voss	Capítulo	1	: BASIC	Página
	Punto	6	: Asignaciones	22
Colegio			de valores	

### 1.6 Asignaciones de valores

Como ya tendremos ocasión de ver, resulta muy útil para su aplicación, el dotar de nombres a las posiciones de memoria, con el fin de almacenar diversos valores bajo estos nombres.

En principio basta con distinguir entre dos tipos de variable:

1. Variables numéricas
2. Variables de cadena

Las variables numéricas pueden tomar números como valores, como por ejemplo:

3.5 4 -17.01 7 18. -3 0.04 0

-7,5 20 etc.

(En lugar de la coma decimal, al programar se utiliza un punto).

Las variables de cadena tienen secuencias de caracteres como valor y deben colocarse entre comillas, por ejemplo:

"ANTONIO" "00" "SEVILLA" "B-30" etc.

Regla 8: Los nombres de las variables numéricas constan de uno o dos caracteres alfanuméricos; el primero debe ser una letra. Los nombres de las variables de cadena deben acompañarse de un \$.

Voss	Capítulo	1	: BASIC	Página
	Punto	6	: Asignaciones	23
Colegio			de valores	

### Ejemplos:

1. Variables numéricas : X B1 NM
2. Variables de cadena : F\$ C2\$ ZZ\$

Después de aclarar estos conceptos hablaremos sobre la instrucción LET, que sirve para asignar valores de modo que pueda disponerse de los mismos a lo largo de todo el programa:

#### Instrucción 4:

nn	LET	nom. variable	=	{	número
					operación
					cadena
				}	nombre de variable

Puede omitirse también la palabra LET.

Esta instrucción permite calcular de la misma manera que se hacía con la instrucción PRINT.

Los operadores para calcular son los siguientes:

- + Adición
- Sustracción
- MOD Resto de división de enteros
- \ División de enteros
- \* Multiplicación
- / División
- Cambio de signo
- Potenciación

La prioridad disminuye en sentido descendente.



Voss	Capítulo	1 : BASIC	Página
	Punto	6 : Asignaciones de valores	24
Colegio			

El rango de los operadores es análogo al que conocemos de la escuela; en caso necesario deben colocarse paréntesis.

Muchos cálculos pueden simplificarse mediante las llamadas funciones incorporadas. Las más importantes son:

#### Funciones importantes

Nombre	Cometido
SIN (X)	Calcula el seno de X en radianes
COS (X)	Calcula el coseno de X en radianes
TAN (X)	Calcula la tangente de X en radianes
LOG (X)	Calcula el logaritmo neperiano de X
EXP (X)	Ofrece la función exponencial en base e
SQR (X)	Ofrece la raíz cuadrada de X
ABS (X)	Ofrece el valor absoluto de X
FIX (X)	Ofrece la parte entera de X
RND (X)	Generación de números aleatorios
CHR\$ (X)	Ofrece el carácter cuyo código ASCII es X
BIN\$ (X)	Convierte X en su expresión binaria
HEX\$ (X)	Convierte X en su expresión hexadecimal
OCT\$ (X)	Convierte X en su expresión octal
LEN (X\$)	Ofrece el número de caracteres de X\$
ATN (X)	Calcula la arcotangente de X

Ahora podemos prescindir de los ejemplos. Le serán suficientes los capítulos siguientes.

Voss	Capítulo	1 : BASIC	Página
	Punto	6 : Editor del BASIC	25

Los ordenadores MSX utilizan un editor de pantalla completa, que puede incluso rivalizar con los editores BASIC de otros microordenadores con vocación profesional. Esta característica facilita de gran forma la escritura de programas BASIC. Un editor de pantalla completa permite la escritura, modificación, o inserción de un programa, en toda la superficie de la pantalla, y no, como ocurre en otros microordenadores, en una zona concreta de la misma. Es conveniente que antes de escribir un programa, se familiarice con los comandos del editor BASIC de su microordenador.

La escritura de un programa BASIC queda rígidamente determinado por unas normas concretas. Si se comete un error mecanográfico o de sintaxis de un determinado comando o sentencia, el programa no podrá nunca funcionar correctamente. En tal caso deberá ser corregido. Para hacerlo, no es necesario volver a reescribir toda la línea en que se ha cometido el error. El editor BASIC permite su corrección sin más operación, que la derivada de posicionar el cursor sobre el mismo y escribir sobre el error la corrección pertinente. En caso necesario podrá hacer uso de las teclas BS (retroceso), DEL (borrado) e INS (inserción). Su editor MSX dispone de más funciones, cuyo fin podrá consultar en el manual de programación de su ordenador. Comentaremos brevemente a continuación, la función de los comandos AUTO, DELETE y RENUM, que complementan al editor de BASIC.

- AUTO : Su cometido es generar automáticamente el número de línea
- DELETE : Permite el borrado de líneas de un programa.
- RENUM : Su función es volver a numerar las líneas de un programa.

Consulte la sintaxis de estos comandos en el manual de programación de su ordenador MSX.

Voss	Capítulo	1	: BASIC	Página
	Punto	7	: Entrada de	26
Colegio			Informaciones	

### 1.7 Entrada de Informaciones

Para introducir informaciones resulta muy adecuada la siguiente instrucción:

#### Instrucción 5:

```
nn INPUT Lista de var.
```

Esta instrucción puede parecer complicada, pero es fácil de comprender:

Cuando el ordenador llega a esta instrucción, interrumpe el proceso del programa y produce un signo de interrogación en la pantalla. Entonces espera a que el usuario introduzca tantos valores (separados por comas) como nombres haya en la lista de variables (es decir, como mínimo un valor, en el caso de que se trate de un solo nombre).

#### Ejemplos:

```
INPUT X      El ordenador pide un número, que
              será almacenado en el campo X.
```

```
INPUT A,B,C  El ordenador pide tres números.
```

Voss	Capítulo	1	: BASIC	Página
	Punto	7	: Entrada de	27
Colegio			Informaciones	

```
INPUT N$     El ordenador pide una cadena (p.e.
              "ANTONIO"), que será almacenada en
              el campo N$.
```

En cualquier caso, toda variable de cadena debe ser dimensionada previamente (ver pag. 31).

Regla 9: Los nombres en la lista de variables de la instrucción INPUT se separan mediante comas. Los valores a introducir también hay que separarlos de la misma forma.

#### Ejemplo:

```
5 CLS
10 PRINT "CUATRO VALORES POR FAVOR : ";
20 INPUT A,B,C,D
30 LET S=A+B+C+D
40 AM=S/4
50 PRINT:?:? "PROMEDIO" = ";AM
60 END
```

Vemos claramente como este programa calcula el promedio de cuatro valores a introducir.

Voss	Capítulo	1 : BASIC	Página
	Punto	8 : Bifurcaciones del programa	28
Colegio			

### 1.8 Bifurcaciones del programa

Mientras en un programa BASIC no aparezca ninguna bifurcación (llamada salto), el proceso tiene lugar en el orden de los números de sentencia asignados. Si no se desea respetar este orden son necesarios los saltos en el programa.

Distinguimos entre saltos condicionales y saltos no condicionales.

Veámos en primer lugar el salto condicional, que se realiza gracias a la siguiente instrucción:

Instrucción 6:

```

nn IF condición THEN
    Instrucción
    Número sentencia mm
  
```

Esta instrucción obliga al sistema operacional del ordenador a consultar: Si la condición expresada después del término IF se cumple, entonces se ejecuta la instrucción escrita detrás de THEN o se pasa a la sentencia número mm.

De no cumplirse esta condición, el control del programa pasa a la sentencia que sigue a la sentencia IF.

Voss	Capítulo	1 : BASIC	Página
	Punto	8 : Bifurcaciones del programa	29
Colegio			

Regla 10: Si se cumple la sentencia IF, se efectúa la instrucción detrás del THEN o se salta al número de sentencia ahí escrito. De no cumplirse esta condición, se procesa la siguiente sentencia.

Regla 11: Si hay otras instrucciones detrás del IF dentro de la misma sentencia, sólo se realizarán si se cumple la condición.

Ejemplo:

```

10 CLS
20 I=1
30 Q=I*I
40 W=SQR(I)
50 PRINT USING "#####.###";I,Q,W
60 I=I+1
70 IF I <=20 THEN 30
80 END
  
```

Este programa calcula los cuadrados (que aparecen en el campo Q; línea 30) de todos los números enteros del 1 al 20 (que aparecen en el campo I), y también sus raíces cuadradas (en el campo W; línea 40), imprimiendo cada grupo de tres valores en una línea de la pantalla.

La línea 70 resulta decisiva. Dicta que el programa vuelva una y otra vez a la línea 30, mientras en el campo I siga habiendo un valor menor de 21 tras elevar su contenido cada vez en 1 (línea 60).

En tanto esta condición se cumpla, se ejecutará la instrucción que sigue al THEN (que es el salto de vuelta a la línea 30); cuando ya no se cumpla se procesará la instrucción que sigue al IF (línea 80).

La alternativa al salto condicional la encarna el salto no condicional:

Instrucción 7:

```
nn GOTO mm
```

Quando el control del programa alcanza esta instrucción tiene lugar un salto a la sentencia número mm.

Para el siguiente programa de muestra, necesitamos otra instrucción:

Instrucción 8:

```
DIM nombre de var. (número 1),
    nombre de var. (...)
```

Ejemplo:

```
DIM X(20)
```

Esta instrucción mantiene libres para la variable X no sólo una posición, sino 21 (posiciones 0 a 20).

Para utilizar variables de cadena debe mantenerse libre un campo por cada carácter de la cadena, por ejemplo:

```
DIM A$(5)
```

Mantiene libres cinco campos para una cadena, de forma que pueda introducirse una cadena con una longitud máxima de cinco caracteres.

Siempre que se trabaje con variables de cadena habrá que escribir una instrucción DIM como ésta.

```
10 DIM A$(1)
20 CLS
30 ? "INTRODUZCA VALOR : ";
40 INPUT X:?:?:?
50 N=N+1
60 S=S+X
70 PRINT "OTRO VALOR (S/N) ";
80 INPUT A$
90 IF A$ = "S" THEN 20
100 ?:?:?:? "PROMEDIO = "; S/N
110 END
```

Este programa calcula la media aritmética de una cantidad cualquiera de valores a introducir. Después de introducir un valor, el ordenador pregunta al usuario si desea introducir otros. Si el usuario contesta 'S' ('sí'), tiene lugar un salto de la línea 90 a la línea 20 y se pide el siguiente valor.

Si, por el contrario, ya no quedan valores, es decir, si el usuario ya no contesta con "S" en la instrucción INPUT de la línea 80, prosigue la ejecución del programa a partir de la línea 100 (cálculo y visualización de la media).

La instrucción CLS borra la pantalla.

Voss	Capítulo	1 : BASIC	Página
	Punto	9 : Bucles	32
Colegio			

### 1.9 Bucles

A menudo resulta adecuado ejecutar determinadas partes del programa más de una vez. Para ello se necesitan los llamados bucles (loops). Estos bucles pueden generarse sencillamente con las instrucciones tratadas hasta el momento, tal como puede apreciarse en el programa anterior.

Las siguientes instrucciones simplifican la constitución de bucles, abriéndose las siguientes posibilidades adicionales.

```
Instrucción 9: nn FOR nombre variable Index =
                valor inic.
                TO valor final (STEP longitud)
```

La siguiente instrucción va unida a la anterior:

```
Instrucción 10: nn NEXT nombre variable indexada
```

La variable indexada que aparece en ambas instrucciones debe ser un número real. El valor inicial, el valor final y la longitud pueden ser números, variables o expresiones aritméticas.

Voss	Capítulo	1 : BASIC	Página
	Punto	9 : Bucles	33
Colegio			

''Cuando el ordenador llega a la instrucción FOR, se iguala la variable indexada al valor inicial y luego se ejecuta esta parte del programa situada entre FOR y NEXT con este valor.

Tras alcanzar el NEXT, la variable indexada salta al siguiente valor, que resulta del valor inicial + la longitud y se ejecuta nuevamente la misma parte del programa, esta vez con el segundo valor de la variable indexada.

Esto sigue hasta que, con una nueva adición de la longitud tras alcanzar el NEXT, se supera el valor final. Entonces se ejecuta la línea que sigue al NEXT.

Puede suprimirse el "STEP longitud" cuando la longitud deba ser igual a uno.'' (PRUST, 1982, pag. 64).

El programa siguiente imprime 10 veces la expresión "BUENOS DIAS" de forma uniforme, puesta una debajo de la otra:

```
5 CLS
10 FOR I=1 TO 10
20 PRINT "BUENOS DIAS"
30 NEXT I
40 PRINT
50 PRINT "FINAL"
60 END
```

### 1.10 La utilización de memorias externas

Siempre se desea guardar por un espacio de tiempo cualquier programa elaborado, de modo que se pueda utilizar en caso de necesidad (del mismo modo, en ocasiones se desean utilizar programas de otros autores). Esto es posible utilizando cintas de cassette o diskettes. Para ello deben conocerse los siguientes comandos.

Para cargar un programa BASIC a partir de un cassette necesitamos el :

Comando 4:

```
CLOAD "Nombre del programa"
```

Para grabar un programa BASIC en un cassette necesitamos el:

Comando 5:

```
CSAVE "Nombre del programa"
```

El cassette no es un recurso muy fiable. Para verificar si la grabación ha resultado correcta se usa el comando siguiente:

```
CLOAD? "Nombre del programa"
```

Se puede variar la velocidad de transmisión con el comando:  
SCREEN (modo gráfico),(tamaño sprite),(sonido teclas),  
(velocidad de transmisión),(tipo de impresora)

Si en el submandato (velocidad de transmisión), escribimos un 1, la velocidad quedará establecida en 2400 baudios, si por el contrario, se omite el submandato o se escribe un 0, la velocidad será de 1200 baudios.

Si al escribir el comando CLOAD no especificamos el nombre del programa, es decir, si escribimos CLOAD " ", el ordenador cargará el primer programa que se encuentre. Si por el contrario, especificamos el nombre del programa el ordenador solo cargará el programa cuyo título se corresponda con el que hemos escrito entre las comillas, e ignorará el resto de los programas que pudiera contener el cassette -aunque visualizará el nombre de los mismos a medida que se sucedan-.

Existen también otros comandos que permiten la carga o la grabación de información con distintos formatos del habitual. Estos son:

```
LOAD "CAS:nombre del programa o archivo"
```

Este comando carga en la memoria del ordenador programas o archivos almacenados en cinta magnética en formato ASCII.

```
SAVE "CAS:nombre del programa o archivo"
```

Este comando permite grabar en formato ASCII un programa BASIC o un archivo contenido en la memoria del ordenador.

```
BLOAD "CAS:nombre del programa"
```

Carga un programa de lenguaje máquina almacenado mediante una sentencia BSAVE en un área memoria especificado.

```
BSAVE "CAS:nombre del programa",dirección inicial,  
dirección final,dirección de comienzo de la  
ejecución
```

Almacena un programa en formato binario (código máquina).

### 1.11 Otros aspectos

---

Para cerrar este capítulo Introductorio veámos unos pocos y útiles aspectos adicionales.

La presentación de un programa mejora sustancialmente si durante la ejecución del mismo y antes de la impresión de los resultados, se borra la pantalla. Esto es posible gracias a la instrucción:

```
nn CLS
```

A menudo resulta necesario parar el proceso de un programa en casos como la programación del llamado bucle sin fin. Pulsando las teclas CTRL + STOP se para la ejecución del programa.

Finalmente quiero mencionaremos la Instrucción

```
Instrucción 11:      nn STOP,
```

---

mediante la cual el desarrollo del programa se interrumpe por sí mismo. Para activar nuevamente el programa interrumpido necesitamos el comando:

```
Comando 6:
-----
CONT
```

Finalmente quisiera añadir que a menudo resulta útil incluir observaciones explicativas en un programa (títulos, comentarios, explicaciones, etc.). Esto es posible con la siguiente Instrucción:

Instrucción 12:

---

```
nn REM texto
```

Esta Instrucción no influye en la ejecución del programa, pero aparecerá en cualquier listado que se efectúe del mismo.

Espero que estas indicaciones sirvan para que incluso el principiante tenga una idea de los elementos básicos del lenguaje BASIC. De esta manera estará en condiciones de comprender las aplicaciones que siguen. Si aún así tuviera problemas de comprensión, sólo puedo darle un consejo:

P R O B A R

Probando los programas e intentando analizar por qué el ordenador hace precisamente una cosa y no otra se aprende más que de cualquier otra forma.

# 2

## Capítulo 2 : Matemáticas

---

### 2.1 Consideración previa

---

Si queremos una buena aplicación para el ordenador, no hay nada más indicado que las matemáticas. Por esta razón empezaremos por ellas. Más tarde veremos, no obstante, que las matemáticas no son el único campo de aplicación.

En todo caso, buscaremos unos ejemplos adecuados pues todos sabemos que las matemáticas cubren un espectro de grados de dificultad muy amplio - empezando con las operaciones básicas y pasando por el cálculo infinitesimal.

Tratarlos problemas de "dificultad media", es decir, nos orientaremos hacia problemas como los que se presentan en el último nivel de EGB y en el primer curso de bachillerato.



Voss	Capítulo	2	: Matemáticas	Página
	Punto	1	: Consideración	40
Colegio			previa	

Las matemáticas son perfectas para demostrar las posibilidades del ordenador, sin que la presentación de los problemas tratados ocupe demasiado tiempo debido a un eventual grado de complejidad de los mismos.

Este libro no pretende exceder en las posibilidades del lector con los ejemplos de aplicación, simplemente se trata de presentar la capacidad del ordenador. Esto se conseguirá con problemas de estructura sencilla, adaptados a los problemas de los escolares más jóvenes.

En los programas de este y de los siguientes capítulos se seguirá la siguiente norma :

¡En los programas prescindiremos de corregir los errores de introducción del usuario!

Por ejemplo, en un problema matemático dado, si sólo pueden utilizarse números positivos (por ejemplo en las raíces), el correspondiente programa no tiene previsto el informarle al usuario en caso de haber introducido por error un número negativo.

Comenzaremos ahora con un ejemplo de grado de dificultad "medio".

Voss	Capítulo	2	: Matemáticas	Página
	Punto	2	: Pitágoras	41
Colegio				

## 2.2 Teorema de Pitágoras

---

Por medio del famoso teorema de Pitágoras puede mostrarse de forma clara, cómo hay que estructurar un problema.

Siempre se seguirán los siguientes pasos de trabajo:

### Pasos de trabajo:

---


1. PASO : Presentación del problema
2. PASO : Análisis del problema
3. PASO : Diagrama de flujo
4. PASO : Programa
5. PASO : Lista de variables
6. PASO : Descripción del programa
7. PASO : Resultados

Siguiendo esta estructuración pasemos ahora al problema citado:

1. PASO : Presentación del problema

El teorema de Pitágoras dice que en el triángulo rectángulo, la longitud del lado base (hipotenusa) es igual a la raíz cuadrada de la suma de los cuadrados de ambos catetos.

Esta figura ilustra lo dicho:



$$c = \sqrt{a^2 + b^2}$$

2. PASO : Análisis del problema

En este caso, el análisis del problema es muy sencillo:

Debe recordarse tanto aquí como en los siguientes problemas que cualquier proceso de datos funciona según el

Principio E P S

ya presentado.

E	=	Entrada
P	=	Proceso
S	=	Salida


Por esta razón deberemos fijar, en primer lugar, las informaciones que debe recibir el ordenador de nuestra parte, para la resolución del problema.

Las informaciones de entrada son lógicamente las longitudes de ambos catetos a y b.


En el proceso, estos dos valores deben ser elevados al cuadrado; debe efectuarse su suma y finalmente extraerse la raíz cuadrada de esta suma.

La salida consiste únicamente en visualizar en la pantalla el resultado obtenido (la longitud de la hipotenusa c).

3. PASO : Diagrama de flujo



Versión resumida del diagrama de flujo:



```

10 REM      PITAGORAS
20 COLOR 1,15,15:CLS
30 ? "PROGRAMA PARA DETERMINAR LA HIPOTE-";
40 ? "NUSA DE UN TRIANGULO RECTANGULO"
45 ?::?:
47 ? "      PROF. DR. W. VOSS, 1984"
50 ?::?:
60 INPUT "PRIMER CATETO : ";A:?
70 INPUT "SEGUNDO CATETO : ";B:?
80 LET Q1=A*A
90 LET Q2=B*B
100 LET S=Q1+Q2
110 LET C=SQR(S)
120 ?::?: "HIPOTENUSA C = ";C
130 END

```

5. PASO: lista de variables

- A : Primer cateto
- B : Segundo cateto
- C : Hipotenusa
- Q1 : Cuadrado del primer cateto A
- Q2 : Cuadrado del segundo cateto B
- S : Suma de los cuadrados de los catetos

6. PASO : Descripción del programa

Nos basamos en la primera versión, la más larga del programa:

- Línea 10 : Comentario
- Línea 20 : Se establecen los colores de la pantalla y se procede a su borrado.
- Línea 30-50: Impresión del título en pantalla
- Línea 60 : Se pide un valor para el primer cateto, que es almacenado en el campo A
- Línea 70 : Lo mismo con el segundo cateto en el campo B
- Línea 80 : Cálculo del cuadrado del primer cateto
- Línea 90 : Cálculo del cuadrado del segundo cateto
- Línea 100 : Suma de los dos cuadrados resultantes
- Línea 110 : Cálculo de la raíz de la suma
- Línea 120 : Impresión del resultado en pantalla
- Línea 130 : Final del programa

Voss	Capítulo	2 : Matemáticas	Página
	Punto	2 : Pitágoras	48
Colegio			

## 7. PASO : Resultados

Una vez iniciado el programa (en su primera versión) aparece el título en la pantalla:

PROGRAMA PARA DETERMINAR LA HIPOTENUSA DE UN TRIANGULO RECTANGULO

PROF. DR. W.VOSS, 1984

Dos líneas más abajo se pide el primer valor:

PRIMER CATETO : ?

A continuación, Introduzcamos por ejemplo el valor 3 y el ordenador reaccionará de la forma siguiente:

SEGUNDO CATETO : ?

Introduzcamos por ejemplo el valor 4, a lo que el ordenador contestará:

HIPOTENUSA C = 5

Este programa es adecuado para calcular, a partir de cualquier valor, la hipotenusa de un triángulo rectángulo. En este sentido se trata de un programa muy general.

Voss	Capítulo	2 : Matemáticas	Página
	Punto	2 : Pitágoras	49
Colegio			

Hemos tratado este problema muy extensamente con el fin de ilustrar el procedimiento a seguir en cada uno de los ejemplos. De ahora en adelante nos ahorraremos explicaciones, por lo menos cuando se trate de ejemplos tan sencillos como el Teorema de Pitágoras.

2.3 M.C.D. y m.c.m.

---

1. PASO : Presentación del problema

En muchos cálculos algebraicos resulta útil conocer el máximo común divisor (M.C.D.) y el mínimo común múltiplo (m.c.m.) de diferentes números, especialmente cuando aparecen quebrados.

El M.C.D. es el valor que es divisor, sin resto, de dos números sin que haya ningún otro número mayor que cumpla esta condición.

El m.c.m. es el número divisible entre los dos números de partida no habiendo otro número más pequeño para el que se cumpla esto.

Si tomamos por ejemplo los dos números  $Z_1=30$  y  $Z_2=40$ , el M.C.D. es 10 y el m.c.m. es 120. (El lector lo puede comprobar mediante cálculo mental).

2. PASO : Análisis del problema

¿Cómo hallar el M.C.D. y el m.c.m. de dos números dados?

Analizemos por medio de los números 30 y 40 cuál es la forma de proceder más adecuada para hallar el M.C.D.:

1. Paso :  $40 : 30 = 1$  resto 10

2. Paso :  $30 : 10 = 3$  resto 0

En este ejemplo se halla el M.C.D. de la forma siguiente:

1. Paso : División del número mayor entre el número menor; anotación del resultado.

2. Paso : División del número menor entre el resto; anotación del nuevo resto.

Si el nuevo resto es 0, el divisor que nos ha llevado al mismo es el M.C.D. buscado

Veámos otro ejemplo ilustrativo:

Campo 1 : Z1 (Número mayor)	Campo 2 : Z2 (Número menor)	División	Resto
78	42	1	36
42	36	1	6
36	6	6	0

El M.C.D. de 78 y 42 es 6.

El esquema de cálculo puede transformarse fácilmente en un diagrama de flujo tal como muestra la tabla de arriba.

Pero antes pasemos al mínimo común múltiplo (m.c.m.):

El m.c.m. de dos números Z1 y Z2 se obtiene dividiendo el producto de ambos entre el M.C.D.


Así, para los números Z1=78 y Z2=42 se obtiene:

$$\text{m.c.m.} = \frac{Z1 * Z2}{\text{M.C.D.}} = \frac{78 * 42}{6} = 546$$

Para efectuar este cálculo dentro de un programa BASIC según la tabla de arriba cambiando continuamente el valor

que ocupa cada campo, calcular el M.C.D. e introducir los valores Z1 y Z2 en dos campos auxiliares H1 y H2 de "memoria intermedia" para que estén disponibles para el cálculo del m.c.m.

3. PASO : Diagrama de flujo



4. Paso : Programa

```

10 REM      MATES2 -MCD Y MCM-
20 COLOR 1,15,15:CLS
30 PRINT "PROGRAMA PARA DETERMINAR EL M.C.D. Y"
40 ? "EL M.C.M. DE LOS NUMEROS INTRODUCIDOS.":
   ??:?
50 ? "      PROF. DR. W. VOSS, 1984":?:?:?
60 INPUT "NUMERO MAYOR : ";Z1
70 PRINT:PRINT
80 INPUT "NUMERO MENOR : ";Z2
90 H1=Z1:H2=Z2
100 Q=Z1/Z2
110 R=Z1-Z2*INT(Q)
120 IF R>0 THEN Z1=Z2:Z2=R:GOTO 100
130 MD=Z2
140 MM=(H1*H2)/(MD)
150 ??:?:? "M.C.D. = ";MD
160 ??:?:? "M.C.M. = ";MM
170 ??:?:? "FINAL DEL CALCULO":END
    
```

5. Paso : Lista de variables

MD : Máximo común divisor  
 H1 : Campo auxiliar 1  
 H2 : Campo auxiliar 2  
 MM : Mínimo común múltiplo  
 Q : Cociente de los números Z1 y Z2  
 R : Resto de la división  
 Z1 : Número mayor  
 Z2 : Número menor



6. PASO : Descripción del programa

Línea 10-50 : Título y explicaciones

Línea 60 : Introducción de un número

Línea 70 : Dos líneas en blanco

Línea 80 : Introducción del otro número

Línea 90 : Ocupación de los dos campos auxiliares H1 y H2 con los valores introducidos, para más tarde seguir disponiendo de los mismos (ver 140)

Línea 100 : Determinación del cociente de los dos números introducidos

Línea 110 : Determinación del resto de la división utilizando la función INT

Línea 120 : Si este resto es mayor que 0 se ocupa el campo Z1 con el segundo número Z2 y el campo Z2 con el resto R, volviendo el programa a la línea 100

Línea 130 : Si en cambio el resto es igual a cero, en el campo Z2 tendremos el M.C.D., que será transferido al campo M.C.D.

Línea 140 : De acuerdo con nuestra reflexión, el mcm resulta del producto de ambos números (que aún se encuentran en los campos H1 y H2), dividido entre el MCD.

Línea 150-160 : Impresión de los resultados con una separación de algunas líneas en blanco

Línea 170 : Finalización del programa

7. PASO : Resultados del programa

Después del comando RUN, el programa indica, en una "pantalla limpia", lo siguiente :

```
PROGRAMA PARA DETERMINAR EL M.C.D.
Y EL M.C.M. DE LOS DOS NUMEROS
INTRODUCIDOS
```

PROF. DR.W.VOSS, 1984

NUMERO MAYOR : ?

Cuando aparezca el interrogante, que como ya se sabe, indica una interrupción del programa, introducir por

ejemplo el valor 78; a continuación el programa pedirá el segundo número:

NUMERO MENOR : ?

Introduzcamos por ejemplo el número 42 y el ordenador llevará a pantalla los resultados siguientes:

M.C.D. = 6

M.C.M. = 546

FINAL DEL CALCULO

## 2.4 Comprobación de números primos

---

### 1. PASO : Presentación del programa

Se trata de desarrollar un programa que compruebe si es primo o no un número entero positivo cualquiera que hemos introducido.

Como ya es sabido, los números primos sólo son divisibles por 1 y por sí mismos.

Voss	Capítulo	2	: Matemáticas	Página
	Punto	4	: Comprobación de	60
Colegio			números primos	

## 2. PASO : Análisis del problema

Para comprobar si un número dado es primo, hay que dividirlo entre todos los divisores mayores que 1, pero menores que él mismo. Mientras aparezca un resto en estas divisiones o el cociente no sea entero, habrá que seguir dividiendo por otro divisor.

De obtener en una de estas divisiones un resultado sin resto, el número a examen no será primo.

Si, en cambio, se obtiene un resto en todas las divisiones posibles, se llegará a la conclusión de que el número inicial es primo.

Los divisores que deben ser probados comienzan por el 2 y pueden finalizar a la mitad del número inicial; los divisores mayores que éste no es necesario probarlos, porque en ningún caso darán un cociente entero.

Ejemplo : Número inicial  $X = 25$

1. Divisor : 2     $25/2 = 12$  resto 1
  2. Divisor : 3     $25/3 = 8$  resto 1
  3. Divisor : 4     $25/4 = 6$  resto 1
  4. Divisor : 5     $25/5 = 5$  resto 0
- ===

25 no es ningún número primo.

Voss	Capítulo	2	: Matemáticas	Página
	Punto	4	: Comprobación de	61
Colegio			números primos	

Ejemplo : Número de partida  $X = 7$

1. Divisor : 2     $7/2 = 3$  resto 1
2. Divisor : 3     $7/3 = 2$  resto 1


No es necesario probar más divisores, porque el resultado de la división ya no puede ser entero.

Como en este ejemplo no aparece ningún resto 0, podemos decir que:

7 es un número primo.

Estos ejemplos muestran que hay que aumentar paso a paso el divisor comenzando por el valor 2 hasta alcanzar como máximo el valor  $X/2$ .

3. PASO : Diagrama de flujo



Este ejemplo muestra como puede simplificarse de modo considerable - y con ello acelerarse - el programa si analizamos cuidadosamente el problema:

Un número par nunca puede ser primo pues, en todo caso, éste ya es divisible por 2. Por ello pueden excluirse los números pares como "números no - primos" desde el comienzo del programa.

Nos quedan los números impares. Para un número cualquiera X no es necesario que el divisor llegue a ser X/2; basta con que probemos divisiones hasta máximo  $\sqrt{X}$ . Si hasta este punto no es posible ninguna división sin resto, tampoco lo será con divisores mayores.

Invitamos al lector a que intente demostrar esta afirmación.

Voss	Capítulo	2	: Matemáticas	Página
	Punto	4	: Comprobación de	64
Colegio			números primos	

4. Paso : Programa.

```

10 REM      "MATES3" -NUMERO PRIMO"
20 COLOR 1,15,15:CLS
30 ? " PROGRAMA PARA COMPROBAR SI EL NUMERO"
40 ? " INTRODUCIDO ES PRIMO O NO."
45 ??:?:?
47 ? "      PROF. DR. W. VOSS, 1984"
50 ??:?:?:?
60 INPUT "UN NUMERO ENTERO, POR FAVOR : ";X
70 T=2
75 IF X=0 THEN 20
80 IF X/2=INT(X/2) THEN 140
90 Q=X/T
100 IF Q=INT(Q) THEN 140
110 T=T+1
120 IF T<=SQR(X) THEN 90
130 ??:?:?:? X;"ES UN NUMERO PRIMO":GOTO 150
140 ??:?:?:? X;"NO ES UN NUMERO PRIMO, SINO"
145 ? " QUE ES DIVISIBLE POR ";T
150 ??:?:?:?:? "FIN DEL CALCULO":END

```

5. Paso : Lista de variables

Q = Cociente entre el número X y el divisor T  
T = Divisor  
X = Número a Introducir para comprobar si es primo o no.

Voss	Capítulo	2	: Matemáticas	Página
	Punto	4	: Comprobación de	65
Colegio			números primos	

6. PASO : Descripción del programa

- Línea 10-50 : Título, limpieza de la pantalla e impresión del título.
- Línea 60 : Demanda del número a comprobar.
- Línea 70 : Ocupación del campo divisor con el 2.
- Línea 80 : SI X es divisible por 2 (entonces  $X/2 = INT(X/2)$ ), pasar a la línea 140; si no es así,proseguir en la línea 90.
- Línea 90 : División de X entre T (se obtiene Q).
- Línea 100 : SI en una de las divisiones el resto es 0 ( $Q = INT(Q)$ ), seguir en la línea 140, en caso contrario seguir en la línea 110.
- Línea 110 : Aumentar el divisor T en 1.
- Línea 120 : Mientras T siga siendo más pequeño que  $\sqrt{X}$ , volver a la línea 90, de lo contrario proseguir en 130.
- Línea 130 : SI se alcanza esta línea es porque ninguna de las divisiones ha dado resto 0. Esto significa que X debe ser número primo.

Se imprime el resultado y tiene lugar un salto a la línea 150.

Línea 140 : La línea 140 se alcanza únicamente en y 145 el caso de haber obtenido resto 0 en una de las divisiones.

Esto supone que X no es primo, y que es (como mínimo) divisible por el valor que hay en el campo T.

Esto se visualiza.

Línea 150 : Finalización del programa.

7. PASO : Resultados

En este y en próximos ejemplos prescindiremos de repetir la impresión del título.

Después de aparecer el título, el programa pide el número a comprobar y lo almacena en el campo X (también prescindiremos en este y en próximos ejemplos de reproducir los correspondientes textos de demanda).

Introduzcamos p.e. el número 13 y el ordenador contestará:

13 ES UN NUMERO PRIMO

FIN DEL CALCULO

Si en cambio introducimos el número 27, el ordenador nos contestará:

27 NO ES UN NUMERO PRIMO, SINO  
QUE ES DIVISIBLE POR 3

FIN DEL CALCULO

Voss	Capítulo	2 : Matemáticas	Página
	Punto	5 : Ecuación de	68
Colegio		segundo grado	

## 2.5 Ecuación de segundo grado

### 1. PASO : Presentación del problema

La resolución de ecuaciones de segundo grado se cuenta entre los problemas fundamentales de la matemática en los últimos niveles de EGB y primeros del BUP y prepara para problemas de mayor complejidad.

La forma general de una ecuación de segundo grado en la cual hay que despejar la incógnita  $x$  es:

$$(1) \quad ax^2 + bx + c = 0$$

Dividiendo esta igualdad por el coeficiente  $a$ , se obtiene:

$$(2) \quad x^2 + px + q = 0$$

Se obtienen dos soluciones  $x_1$  y  $x_2$  a partir de la fórmula siguiente:

$$(3) \quad x = -p/2 \pm \sqrt{p^2/4 - q}$$

Voss	Capítulo	2 : Matemáticas	Página
	Punto	5 : Ecuación de	69
Colegio		segundo grado	

Según la constelación de coeficientes  $a$ ,  $b$  y  $c$ , la fórmula (3) nos llevará a resultados distintos.

Por ejemplo, cuando el valor bajo la raíz es negativo, obtendremos resultados fuera del campo de los números reales. Este valor se llama discriminante ( $D$ ).

Se obtienen también soluciones especiales cuando el coeficiente  $a$  es igual a 0; en este caso ya no tenemos una ecuación de segundo grado.

Por ello, de ahora en adelante contemplaremos sólo los casos "auténticos" ( $a \neq 0$ ) y al mismo tiempo con soluciones reales ( $D > 0$ ). En todos los demás casos el programa deberá generar un mensaje de interrupción.


Para resolver el problema utilizaremos la fórmula (3) en su variante (4):

$$(3) \quad x = -p/2 \pm \sqrt{p^2/4 - q}$$

$$(4) \quad x = H \pm \sqrt{D} \quad \begin{array}{l} H \text{ es un valor} \\ \text{auxiliar que} \\ \text{equivale a } -p/2 \end{array}$$

El correspondiente programa BASIC debe determinar primero los valores  $H$  y  $D$ , para luego imprimir las soluciones o un mensaje de interrupción según sea  $D > 0$  o  $D < 0$ .

3. PASO : Diagrama de flujo



4. PASO : Programa

```

10 REM "MATES4"
15 REM -ECUACION DE SEGUNDO GRADO-
20 COLOR 1,15,15:CLS
30 ? " PROGRAMA PARA RESOLVER UNA ECUACION"
40 ? " DE SEGUNDO GRADO."
45 ?:::
47 ? " PROF. DR. W. VOSS, 1984"
50 ?:::?:
60 INPUT "Coeficiente del término cuadrático A:"
  ";A:PRINT
70 INPUT "Coeficiente del término cuadrático B:"
  ";B:?
80 INPUT "Término independiente C:"
  ";C:?
90 P=B/A:Q=C/A
100 H=-P/2:D=P*P/4-Q
110 IF D<0 THEN ? "SOLUCION IMPOSIBLE EN EL CAMPO
REAL":GOTO 150
120 X1=H+SQR(D):X2=H-SQR(D)
130 ?:::?"X1 = ";X1
140 ?:::?"X2 = ";X2
150 ?:::?:?"FINAL DE LOS CALCULOS":END
  
```

5. Paso : Lista de variables

- A = Coeficiente de la parte cuadrática.
- B = Coeficiente de la parte lineal
- C = Constante de la ecuación cuadrática
- D = Discriminante
- P = B/A
- H = -P/2
- Q = C/A
- X1= Primera solución.
- X2= Segunda solución.



Voss	Capítulo	2 : Matemáticas	Página
	Punto	5 : Ecuación de	72
Colegio		segundo grado	

6. PASO : Descripción del programa

- Línea 10-50 : Comentario e Impresión del título
- Línea 60-80 : Introducción de los coeficientes de la igualdad cuadrática (en la forma (1))
- Línea 90 : Determinación de P y de Q según la igualdad (2)
- Línea 100 : Determinación del valor auxiliar H y del discriminante D
- Línea 110 : Si D es menor que 0, tiene lugar un mensaje de interrupción y se pasa a la línea 150
- Línea 120 : Si, en cambio, el discriminante no es negativo, se calculan las dos soluciones X1 y X2
- Línea 130-140: Ambas soluciones se visualizan en la pantalla
- Línea 150 : Final del programa

Voss	Capítulo	2 : Matemáticas	Página
	Punto	5 : Ecuación de	73
Colegio		segundo grado	

7. PASO : Resultados

Introduzcamos p.e.:

$$A = 4,$$

$$B = 4,$$

$$C = -24,$$

¡El lector debería probar también otros valores!

Se trata de la ecuación de segundo grado:

$$4x^2 + 4x - 24 = 0,$$

tras lo cual el programa visualiza los siguientes resultados:

$$X1 = 2$$

$$X2 = -3$$

FINAL DE LOS CALCULOS

## 2.6 Número e

### 1. PASO : Presentación del problema

El número de Euler, abreviado e, llamado también constante de crecimiento, juega un papel importante en el desarrollo matemático de series, que pueden ser contempladas como cadenas de crecimiento o de desarrollo.

Este sirve además como base de los llamados logaritmos neperianos, que volveremos a encontrarnos en el capítulo dedicado a problemas de economía.

Si nos hemos olvidado del valor del número e, podemos calcularlo con ayuda de un programa BASIC adecuado y siempre sabiendo cómo surge e.

Se trata del límite de la siguiente sucesión, cuando n tiende hacia infinito:

$$e = \left(1 + \frac{1}{n}\right)^n \quad \text{para } n \text{ hacia infinito.}$$


### 2. PASO : Análisis del problema

El análisis del problema es muy sencillo en este caso, dado que se trata únicamente de programar la fórmula anterior.

Este programa puede probarse con valores crecientes y alternativos de n. Como el valor e se alcanza sólo con n tendiendo hacia infinito, recomendamos hacer crecer rápidamente n, por ejemplo doblando su valor en cada cálculo.

Ello exige, no obstante, empezar con un valor de n mayor que 1 (es decir, p.e. n = 2), para que n no se quede estancado en 1 al duplicarlo.

3. PASO : Diagrama de flujo



4. Paso : Programa

```

10 REM "MATES-5" -EULER-
20 COLOR 1,15,15:CLS
30 ? "PROGRAMA PARA DETERMINAR EL NUMERO E."
40 ? " PROF. DR. W. VOSS, 1984"
50 ?
60 N=2
70 ? " n"," e":??:
80 FOR I=1 TO 12
90 E=(1+1/N)^N
100 ? N, E
110 N=N*2
120 NEXT I
130 ??: "FINAL DEL PROGRAMA":END
  
```

5. Paso : Lista de variables

E = Número e (valor aproximado)  
I = Índice contador de bucle  
N = Parámetro en la fórmula para el cálculo del número e

6. PASO : Descripción del programa

Línea 10-50 : Título

Línea 60 : Ocupación del parámetro N con el valor inicial 2

Línea 70 : Impresión del título de la tabla y de una línea en blanco

Línea 80-120 : Recorrer 12 veces el bucle. En cada vuelta se calcula E, se imprimen N y E y se duplica N

Línea 130 : Final del programa

7. PASO : Resultados

Este programa genera una tabla, en la que se enfrentan los valores crecientes de N con los valores resultantes de E.

En esta tabla puede observarse como e se acerca a un valor fijo.

Para acercarnos aún más al límite, deberíamos recorrer más vueltas de bucle.

Entonces habría que modificar la línea 80, por ejemplo del modo siguiente:

```
80 FOR I = 1 TO 30
```

En todo caso, hay que tener en cuenta que los valores demasiado altos de n pueden conducir a errores, debido a las imprecisiones de cálculo del ordenador.

La tabla resultante del programa anterior ofrece este aspecto:

n	e
2	2.25
4	2.44140625
8	2.5657845139503
16	2.6379284973669
32	2.6769901293775
64	2.6973449525641
128	2.707739019692
256	2.7129916242772
512	2.71563200017
1024	2.7169557294491
2048	2.7176184823503
4096	2.7179500809023

FINAL DEL PROGRAMA

Voss	Capítulo	2	: Matemáticas	Página
	Punto	7	: Cálculo de	80
Colegio			porcentajes	

## 2.7 Cálculo de porcentajes

### 1. PASO : Presentación del problema

La regla de tres juega un papel fundamental en el cálculo de porcentajes. Por ejemplo responde a la pregunta de qué parte de una cantidad total G representa un determinado importe B. Podemos ilustrar este problema con un ejemplo típico:

El importe de una factura asciende a Ptas. 45000.--. El cliente debe pagar adicionalmente el llamado impuesto sobre el valor añadido, que supone un 10% del importe neto de la factura, de lo cual resulta:

$$\text{Importe bruto} = \text{Importe neto} + \text{Impuesto}$$

El problema que aquí se plantea consiste entonces en determinar, primero el importe del impuesto y después el importe bruto, para un importe neto y un porcentaje del impuesto dados.

Naturalmente podría pedirse el importe neto sabiendo el porcentaje y el importe bruto.

Voss	Capítulo	2	: Matemáticas	Página
	Punto	7	: Cálculo de	81
Colegio			porcentajes	

### 2. PASO : Análisis del problema

El ejemplo citado arriba puede solucionarse por regla de tres del modo siguiente (señalándonos así el modo de resolución):


$$\begin{aligned} \text{Importe neto} &= 100 \% \text{ equivalente a } 45000.- \text{ Ptas.} \\ \text{Impuesto} &= 14 \% \text{ equivalente a } x \text{ Ptas.} \end{aligned}$$

---


$$x = \frac{45000 \cdot 10}{100} = 4500.- \text{ (Ptas.)}$$

Con ello queda claro el camino a seguir: Se trata de introducir en un programa general el importe neto y el porcentaje del impuesto (o de forma más general, el porcentaje, que puede elegirse arbitrariamente para cualquier aplicación).

3. PASO : Diagrama de flujo



4. Paso : Programa

```

10 REM "MATES6" -PORCENTAJE-
20 COLOR 1,15,15:CLS
30 ? "PROGRAMA PARA LA ESTIMACION DE PORCEN"
40 ? "TAJES."?:?:?
50 ? " PROF. DR. W. VOSS, 1984"?:?:?:?
60 INPUT "Importe neto : ";NB
70 INPUT "Porcentaje : ";SS
80 SB=NB*SS/100
90 BB=NB+SB
95 ??:?:?:?
100 ? "IMPORTE NETO : ";NB:~
110 ? "VALOR PORCENTUAL : ";SS:~
120 ? "IMPORTE BRUTO : ";BB
130 ??:?:?:? "FIN DEL PROGRAMA":END
  
```

5. Paso : Lista de variables

- BB = Importe bruto
- NB = Importe neto
- SB = Valor porcentual
- SS = Porcentaje

Voss	Capítulo	2	: Matemáticas	Página
	Punto	7	: Cálculo de	84
Colegio			porcentajes	

6. PASO : Descripción del programa

Línea 10-50 : Título.

Línea 60-70 : Introducción del importe neto y del porcentaje.

Línea 80-90 : Cálculo del importe del porcentaje y del importe bruto.

Línea 95 : Cuatro líneas en blanco.

Línea 100-120 : Impresión en pantalla de los resultados

Línea 130 : Final del programa.

7. PASO : Resultados

Si se introduce por ejemplo el importe neto 200 y el porcentaje 15%, el programa contestará:

IMPORTE NETO : 200  
 VALOR PORCENTUAL : 30  
 IMPORTE BRUTO : 230  
 FIN

Voss	Capítulo	2	: Matemáticas	Página
	Punto	8	: Dado	85
Colegio				

2.8 Dado

1. PASO : Presentación del problema

Presentamos un programa que "echa" 100 veces los dados y que calcula el promedio de puntos obtenido.

Para este fin puede hacerse uso de la función RND (X); (RND = random = azar, y X = argumento), que genera números aleatorios entre el 0 y el 1.

Números aleatorios son aquellos que aparecen de forma irregular y no sistemática.

2. PASO : Análisis del problema

Para utilizar la función RND con un ordenador MSX basta con indicar el argumento 1.

El lector debería probar simplemente esta función, introduciendo por ejemplo:

```
PRINT RND(1)
```

y otra vez:

```
PRINT RND(1)
```

o todavía mejor:

```
FOR I = 1 TO 10 : PRINT RND(1) : NEXT I
```

y seguidamente la tecla RETURN.

En la pantalla aparecerán números aleatorios situados todos ellos entre 0 y 1.

No obstante, queremos obtener como resultado 100 valores enteros situados entre 1 y 6 - como cuando se echa un dado -.


El esquema siguiente ilustra cómo pueden generarse números enteros entre el 1 y el 6 a partir de números aleatorios tomados del ámbito de valores entre 0 y 1:

Elaboración	Resultado
Función RND	Entre 0 y menos de 1
* 6	entre 0 y menos de 6
+ 1	entre 1 y menos de 7
Función INT	entero entre 1 y 6

Con ello se ha trazado ya el camino para resolver el problema. Como finalmente se desea obtener el promedio, habrá que sumar todos los resultados y dividir esta suma entre 100.



3. PASO : Diagrama de flujo



4. Paso : Programa

```

10 REM "MATES7" -DADO-
20 COLOR 1,15,15:CLS
30 ? "PROGRAMA PARA GENERAR 100 TIRADAS DE"
40 ? "DADO Y CALCULAR EL PROMEDIO DE PUNTOS"
50 ??
60 ? " PROF. DR. W. VOSS, 1984"
70 ??
80 I=1
90 X=RND(1)
100 X=X*6:X=X+1:X=INT(X)
110 ? X;" ";
120 S=S+X
130 I=I+1
140 IF I<101 THEN 90
150 AM=S/100
160 ???
170 ? "VALOR MEDIO = ";AM
180 ??? "FINAL DE LA IMPRESION": END
  
```

5. Paso : Lista de variables

- AM = Valor medio
- I = Contador de bucle
- S = Suma de todas las tiradas
- X = Valor aleatorio

Voss	Capítulo	2 : Matemáticas	Página
	Punto	8 : Dado	90
Colegio			

6. PASO : Descripción del programa

- Línea 10-70 : Título
- Línea 80 : Ajuste del contador de vueltas
- Línea 90-100 : Generación de un número aleatorio y transformación en un número entero situado entre 1 y 6
- Línea 110 : Impresión
- Línea 120 : Incremento de la suma S en el resultado de la tirada X
- Línea 130 : Salto de uno del contador de vueltas
- Línea 140 : Mientras I sea menor que 101 vuelta a la línea 90
- Línea 150-170 : Cálculo del promedio e Impresión
- Línea 180 : Final del programa

Voss	Capítulo	2 : Matemáticas	Página
	Punto	8 : Dado	91
Colegio			

7. PASO : Resultados

Después de iniciar el programa, éste genera tras el título 100 sucesos aleatorios de tirada de un dado y los imprime línea a línea (pensar por qué tiene lugar una impresión línea a línea).

A continuación el ordenador calcula, a partir de los 100 sucesos, el promedio y lo visualiza. Si el generador de números aleatorios funciona correctamente, el resultado deberá acercarse a 3.5.

Como los resultados dependen del azar, lógicamente no los podemos representar aquí.

Voss	Capítulo	2 : Matemáticas	Página
	Punto	9 : Exactitud de los cálculos	92
Colegio			

## 2.9 Exactitud de los cálculos

Un ordenador MSX reserva un número limitado de bytes para cada número real con el que va a calcular. Esto significa que la precisión con la que trabaja está limitada. Puede verse esto claramente si introducimos valores "grandes" en el programa que sigue, haciendo que se impriman a continuación :

```

10 REM EXACTITUD CALCULO
20 A=12345678#
30 B=123456789#
40 C=1234567891#
50 D=12345678912#
60 E=123456789123#
70 F=1234567891234#
80 G=12345678912345#
90 H=1234567891234567#
95 ? "A = ";A: ? "B = ";B: ? "C = ";C:
? "D = ";D: ? "E = ";E: ? "F = ";F:
? "G = ";G: ? "H = ";H: ? "H+1 = ";H+1:
? "H+99= ";H+99

```

Obtenemos los siguientes resultados:

```

A = 12345678
B = 123456789
C = 1234567891
D = 12345678912
E = 123456789123
F = 1234567891234
G = 12345678912345
H = 1.2345678912346E+15
H+1 = 1.2345678912346E+15
H+99= 1.2345678912356E+15

```

Voss	Capítulo	2 : Matemáticas	Página
	Punto	9 : Exactitud de los cálculos	93
Colegio			

Es decir, si el ordenador recibe números de más de 14 cifras (ejemplo H) a la hora de imprimir, los pasa a la llamada representación exponencial, como por ejemplo para H:

1.23456789123456E+15

Esto hay que leerlo como:

1.23456789123456 \* 1 000 000 000 000 000

Este producto da como resultado:

1234567891234560

Por consiguiente, el ordenador ha redondeado por debajo la última posición del valor de H, que ya no puede considerar.

En todo caso, como puede mostrarse por medio de otros ejemplos, este redondeo se efectúa no "recortando" simplemente la posición.

El principal problema que surge aquí es el siguiente:

Si, por ejemplo, sumamos 1 al valor H en el programa anterior, y hacemos que se imprima también el resultado en pantalla, obtendremos:

1.23456789123456E+15

no advertiremos ninguna diferencia entre los valores de H y de H+1 - ¡el ordenador calcula mal! -

Este problema deberá ser considerado siempre que queramos operar números muy grandes con números muy pequeños (o viceversa).

# 3

## Capítulo 3 : Química

---

### 3.1 Otras Instrucciones BASIC


---

Para confeccionar programas en el campo de la química pueden aprovecharse magníficamente las posibilidades gráficas y de color de los ordenadores MSX - por ejemplo para representar gráficamente moléculas o enlaces químicos -.

Tomemos el caso de la molécula de agua:



que, como ya es sabido, está compuesta de dos átomos de hidrógeno (H) y de un átomo de oxígeno (O), representables de la forma siguiente:



Para generar representaciones como ésta se necesitan algunas instrucciones adicionales. Estas se refieren a la utilización de caracteres especiales (en un próximo capítulo hablaremos de las posibilidades de los llamados gráficos de alta resolución).

Para confeccionar este tipo de dibujos resulta muy adecuado el juego de caracteres gráficos MSX (ver manual de instrucciones).

Estos caracteres, que pueden hallarse bajo determinados códigos ASCII (para ello resultan necesarios algunos caracteres especiales adicionales) pueden seleccionarse con la función:

CHR\$ (argumento)

El argumento de esta función es el número de código ASCII del símbolo deseado.

Para visualizar todos los símbolos del código ASCII puede utilizarse el programa siguiente:

```

10 SCREEN 1:COLOR 1,15,15:KEY OFF:CLS:?:?:?:?
20 ? "ESTE PROGRAMA VISUALIZA TODOS"
30 ? "LOS SIMBOLOS DEL CODIGO ASCII"
40 ? "UTILIZADOS POR ESTE ORDENADOR":?:?:?:?:?:?
50 GOSUB 250
60 FOR X=0 TO 14
70 FOR Y=1 TO Y+16
80 IF Y<32 THEN A$="NO USADO" ELSE A$="SIMBOLO  "
90 IF Y=10 THEN ? "ASCII = 10";,"PASA LINEA":GOTO 160
100 IF Y=11 THEN ? "ASCII = 11";,"PONE CURSOR":GOTO 160
110 IF Y=12 THEN ? "ASCII = 12";,"CLEAR SCREEN":GOTO 160
120 IF Y=27 THEN ? "ASCII = 27";,"DESPLAZA":GOTO 160
130 IF Y=31 THEN ? "ASCII = 31";,"PASA LINEA":GOTO 160
140 ? "ASCII =" ;Y,A$;CHR$(1);CHR$(Y)
150 IF Y=254 THEN 200
160 NEXT Y
170 I=I+17
180 GOSUB 250
190 NEXT X
200 ??:? ? " FIN DEL PROGRAMA"
210 END
220 REM
230 REM          -SUBROUTINA-
240 REM
250 ??:?:? ? " Pulse cualquier tecla":? ? " para con
tinuar"
260 B$=INPUT$(1)
270 CLS
280 RETURN

```

Si el usuario introduce por ejemplo:

```
PRINT CHR$(1);CHR$(73)
```

en la pantalla aparece un símbolo similar a una bolita.

De la misma manera, después de introducir

```
PRINT CHR(255)
```

se obtiene un cuadrado de color claro (el símbolo del cursor que tan bien conocemos).

En el caso de nuevos problemas disponemos posibilidades interesantes:

```

PRINT CHR$(1);CHR$(65) que da ☺
PRINT CHR$(1);CHR$(66) que da 🌐
PRINT CHR$(1);CHR$(67) que da ♥
PRINT CHR$(1);CHR$(79) que da ✱
PRINT CHR$(1);CHR$(93) que da /
PRINT CHR$(1);CHR$(94) que da \
PRINT CHR$(1);CHR$(188) que da ◆
PRINT CHR$(1);CHR$(216) que da △

```

Incorporando este tipo de instrucciones a los programas podemos generar gráficos (aún cuando sean bastante toscos).

Estas posibilidades se muestran en el siguiente pequeño programa:

```

10 REM --ANIMACION--
20 SCREEN 1:COLOR 15,1,1:KEY OFF
30 FOR Y=0 TO 23
40 FOR X=0 TO 28
50 LOCATE X,Y
60 ? CHR$(1);CHR$(66);
70 TIME=0
80 IF A=1 THEN 90 ELSE 70
90 CLS
100 NEXT X
110 NEXT Y
120 END

```

Además, en este capítulo, y en la mayoría de los que siguen, necesitaremos ciertas instrucciones que nos permitan introducir mayores cantidades de información de una forma más elegante que con las instrucciones PRINT e INPUT.

Para esto resulta particularmente adecuada la instrucción DATA, en unión de la instrucción READ.

Instrucción 13:

---

nn DATA Valor 1, Valor 2, Valor 3, ...

Instrucción 14:

---

nn READ variable 1, variable 2, variable 3,...

Ejemplo:

```

10 REM -INTRODUCCION DATOS-
20 COLOR 1,15,15:KEY OFF:CLS
30 DATA 4600,4630,8000,2000
40 READ P1,P2,P3,P4
50 LOCATE 13,10
60 PRINT ,P1:PRINT ,P2:PRINT ,P3:PRINT ,P4
70 END

```

Invitamos al lector a probar este programa.

Cuando la cantidad de datos sea elevada, recomendamos utilizar un único nombre de variable que, no obstante, tomará varios valores. Esto se consigue utilizando variables indexadas como las que ya conocemos de las matemáticas. Normalmente en términos matemáticos se escribe:

$x_1, x_2, x_3, \dots$  término general  $x_i$

Esto puede llevarse a cabo de la misma forma con el lenguaje BASIC:

$X(1), X(2), X(3), \dots$  término general  $X(i)$

(en general:  
nombre variable (nombre del índice))

Cuando utilizamos estas variables indexadas hay que comunicar al ordenador al comienzo del programa cuántas posiciones de memoria deben mantenerse libres para cada una de las variables en el principio del programa.

Para ello utilizaremos la Instrucción DIM:

#### Instrucción 13

nn DIM nombre var. 1 (num1), nombre var.2 (num2), ...

Así por ejemplo la Instrucción:

```
10 DIM X(20)
```

mantiene libres 21 posiciones de memoria para la variable X.

Aquí se trata de las posiciones  $X(0), X(1), X(2), X(3), \dots, X(20)$ .

Para completar esta explicación añadiremos que son también posibles indexados dobles:

La Instrucción:

```
10 DIM Z(2,3)
```

mantiene libres  $3 \times 4$  posiciones de memoria para la variable Z, que uno puede imaginarse divididas en tres líneas y cuatro columnas:

Z(0,0)	Z(0,1)	Z(0,2)	Z(0,3)
Z(1,0)	Z(1,1)	Z(1,2)	Z(1,3)
Z(2,0)	Z(2,1)	Z(2,2)	Z(2,3)

El formato general es en este caso:

Z(I,J)

El primer índice (I) es el llamado índice de línea y el segundo índice (J) es el llamado índice de columna.

Para finalizar queremos indicar que la Instrucción READ de un programa, toma valores de los DATAS en orden sucesivo. Ello significa que una segunda Instrucción READ proseguirá con la lectura de los DATAS, allí donde dejó de hacerlo la última Instrucción READ.

Aún así, a menudo es más razonable volver a leer desde el principio la lista de DATAS con una segunda (o siguiente) Instrucción READ.

Para hacer posible esto necesitamos una Instrucción nueva:

Instrucción 14:

---

nn RESTORE

Esta Instrucción provoca que el "cursor de lectura" sea emplazado nuevamente al comienzo de la primera Instrucción DATA, de modo que los datos sean leídos nuevamente desde el principio con la siguiente Instrucción READ.

### 3.2 La molécula de agua

---

#### 1. PASO : Presentación del problema

Vamos a generar, con ayuda de un programa BASIC, la representación gráfica de la molécula de agua ( $H_2O$ ); (véase también el punto 3.1). Los átomos de hidrógeno se representarán p.e. en forma de pequeños círculos y el átomo de oxígeno como una bolita blanca que nos sonríe.

Así, este programa puede usarse como base de programas más complejos de este tipo de representación gráfica.

#### 2. PASO : Análisis del problema

Una vez tenemos ante nosotros el gráfico que debe generarse con este programa, el análisis del problema resulta muy sencillo:

Se trata de producir exactamente el mismo dibujo que ya realizamos al comienzo del punto 3.1.

Observaremos que este dibujo se compone de cinco líneas:




1. Línea : un pequeño círculo.
2. Línea : una línea oblicua de arriba a la izquierda a abajo a la derecha.
3. Línea : una bolita blanca sonriente
4. Línea : una línea oblicua de abajo a la izquierda a arriba a la derecha.
5. Línea : un pequeño círculo.

El pequeño círculo podemos generarlo, como ya hicimos antes, mediante el uso de los códigos ASCII 1-73; escribiendo `CHR$(1);CHR$(73)`, y la bolita sonriente mediante `CHR$(1);CHR$(66)`.

Las dos líneas oblicuas pueden generarse haciendo uso de los operadores aritméticos BASIC `"/`, (división) y `"\"`, (división de enteros).

### 3. PASO : Diagrama de flujo



4. Paso : Programa.

```

10 REM      "QUIMICA -,1"
20 COLOR 15,1,1:KEY OFF:CLS:SCREEN 1
30 ? "REPRESENTACION DE LA MOLECU-"
40 ?:" " "LA DE AGUA H2-O"
50 ?:" "  "PROF. DR. W. VOSS, 1984"
60 LOCATE 16,10:? CHR$(1);CHR$(73)
70 LOCATE 18,11:? "\ "
80 LOCATE 20,12:? CHR$(1);CHR$(66)
90 LOCATE 16,14:? CHR$(1);CHR$(73)
100 LOCATE 18 13:? "/"
110 LOCATE 3,12:? "  H -O"
120 LOCATE 7,13:? "2"
130 LOCATE 5,20:? "FIN DEL PROGRAMA"
140 END

```

5. Paso : Lista de variables

En este programa no se utiliza ninguna variable

6. PASO : Descripción del programa

Línea 10-50 : Título del programa.

Línea 60 : Impresión de un círculo en la columna 16, línea 10.

Línea 70 : Impresión de una raya oblicua en la columna 18, línea 11.

Línea 80 : Impresión de una bolita sonriente en la columna 20, línea 12.

Línea 90 : Impresión de un círculo en la columna 16, línea 14.

Línea 100 : Impresión de una raya oblicua en la columna 18, línea 13.

Línea 110 : Impresión de H<sub>2</sub>-O en la columna 3, línea 12.

Línea 120 : Impresión del subíndice 2

Línea 130 : Impresión de "Final del programa"

Línea 140 : Fin del programa

### 7. PASO : Resultados

La impresión en pantalla de este programa es similar a la que presentamos en el punto 3.1.

### 3.3 Reacción química


---

#### 1. PASO : Presentación del problema

Las reacciones químicas también pueden representarse gráficamente de una forma clara, como se demuestra en el ejemplo del proceso de fabricación del acero.


La fabricación del acero aparece en los países industrializados con la producción de hierro bruto en los altos hornos. Esta producción se desarrolla en los siguientes pasos:

1. Paso : Se quema aire caliente con carbón para obtener monóxido de carbono:



carbón oxígeno monóxido  
de carbono

2. Paso : El monóxido de carbono (CO) caliente asciende por el horno y reacciona allí con el óxido férrico (Fe<sub>2</sub>O<sub>3</sub>):



Oxido	Monóxido	Hierro	Dióxido
férrico	de carbono		de carbono

De este modo se obtiene hierro bruto para su posterior utilización.

Presentamos ahora un programa BASIC que sirve para mostrar estas reacciones químicas.

Para la representación de los átomos y de las moléculas nos serviremos de las mismas posibilidades gráficas que en el ejemplo precedente.

## 2. PASO : Análisis del problema

En realidad, con la descripción anterior y considerando lo dicho en el ejemplo anterior, el análisis del problema ya queda resuelto, de modo que podemos ahorrarnos cualquier otra explicación.

3. Paso : Diagrama de flujo

Tampoco aquí resulta necesario el diagrama de flujo, por los mismos motivos que en el punto anterior.

4. Paso : Programa

```

10 REM Q2-REAC. QUIMICA
20 COLOR 15,1,1:KEY OFF:SCREEN 1:CLS
30 ? "PROGRAMA PARA REPRESENTAR UNA"
40 ? "REACION QUIMICA."
50 ?:" " PROF. DR. W. VOSS, 1984"
60 LOCATE 6,9: ? CHR$(1);CHR$(73)
70 LOCATE 21,10: ? CHR$(1);CHR$(73);CHR$(1);CHR$(66)
80 LOCATE 11,11: ? CHR$(1);CHR$(66);CHR$(1);CHR$(66)
90 LOCATE 17,11: ? "="
100 LOCATE 21,12: ? CHR$(1);CHR$(73);CHR$(1);CHR$(66)
110 LOCATE 6,13: ? CHR$(1);CHR$(73)
120 LOCATE 5,17: ? "2 C + 0      = 2 CO"
130 LOCATE 12,18: ? "2"
140 ?:"
150 ? "Pulse una tecla para seguir"
160 A$=INPUT$(1)
170 CLS
180 LOCATE 23,2: ? CHR$(1);CHR$(66)
190 LOCATE 17,3: ? CHR$(1);CHR$(79);"      ";CHR$(1);
CHR$(73)
200 LOCATE 23,4: ? CHR$(1);CHR$(66)
210 LOCATE 3,6: ? CHR$(1);CHR$(66);"      ";CHR$(1);
CHR$(73);CHR$(1);CHR$(66)
220 LOCATE 2,7: ? CHR$(1);CHR$(79)
230 LOCATE 23,7: ? CHR$(1);CHR$(66)

```

```

240 LOCATE 3,8:? CHR$(1);CHR$(66);" + ";CHR$(1);
CHR$(73)
250 LOCATE 10,8:? CHR$(1);CHR$(66)
260 LOCATE 20,8:? "+ ";CHR$(1);CHR$(73)
270 LOCATE 2,9:? CHR$(1);CHR$(79)
280 LOCATE 23,9:? CHR$(1);CHR$(66)
290 LOCATE 3,10:? CHR$(1);CHR$(66)
300 LOCATE 9,10:? CHR$(1);CHR$(73);CHR$(1);CHR$(66)
310 LOCATE 23,12:? CHR$(1);CHR$(66)
320 LOCATE 17,13:? CHR$(1);CHR$(79)
330 LOCATE 23,13:? CHR$(1);CHR$(73)
340 LOCATE 23,14:? CHR$(1);CHR$(66)
345 LOCATE 13,8:? "="
350 LOCATE 0,17:? " Fe 0 + 3 CO = 2 Fe + 3 CO"
360 LOCATE 2,18:? " 2 3"
370 LOCATE 26,18:? "2"
380 ?::? " FIN DEL PROGRAMA":END

```

5. PASO : Lista de variables

No se utiliza ninguna variable.

6. PASO : Descripción del programa

Línea 10-50 : Título

Línea 60 : Impresión de un círculo, columna 6,  
línea 9

Línea 70 : Impresión de un círculo y de una bolita,  
en las columnas 21/22, línea 10

Línea 80 : Impresión de dos bolitas en las columnas  
11 y 12, línea 11

Línea 90 : Impresión de un signo = en la columna  
17, línea 11

Línea 100 : Igual que en la línea 70, pero en la  
línea 12

Línea 110 : Igual que en la línea 60, pero en la  
línea 13

Línea 120-140 : Impresión del texto:  
 $2 C + O_2 = 2 CO$

Línea 150-170 : Espera y borrado de la pantalla

Voss	Capítulo	3 : Química	Página
	Punto	3 : Ecuación química	114
Colegio			

Línea 180-345 : Impresión del gráfico de la segunda  
reacción

Línea 350-370 : Impresión de la segunda reacción

Línea 380 : Final del programa

Voss	Capítulo	3 : Química	Página
	Punto	3 : Ecuación química	115
Colegio			

7. PASO : Resultados

El programa presentado genera consecutivamente dos gráficos en la pantalla; el segundo dibujo aparece cuando el usuario pulsa cualquier tecla.

Podemos prescindir de presentar aquí ambos dibujos, pues no nos es posible efectuar una representación en colores. Nos parece más razonable que el lector pruebe directamente el programa en un televisor o monitor.

### 3.4 Cálculo estequiométrico

---

#### 1. PASO : Presentación del problema


Los cálculos estequiométricos consisten en cuantificar las cantidades de las diferentes sustancias que participan en una reacción química. Permiten contestar por ejemplo a la siguiente cuestión:

¿Cuántos gramos de azufre son necesarios para convertir completamente en sulfuro de zinc 100 gramos de zinc? (Azufre = S; Zinc = Zn; Sulfuro de zinc = ZnS).

Peso atómico del azufre : 32.1  
 Peso atómico del zinc : 65.4

#### 2. PASO : Análisis del problema

Para resolver el problema deberemos partir de la correspondiente reacción:



Si relacionamos las llamadas masas molares de la sustancia buscada y de las sustancias dadas, obtendremos que:

son necesarios 32.1 g de azufre para transformar 65.4 g de zinc (ver arriba).

Se cumple que:

$$x/100 = 32.1/65.4$$


o

$$x = 3210/65.4$$

x es el valor buscado.

3. PASO : Diagrama de flujo.

En este problema el diagrama de flujo vuelve a ser muy sencillo:



4. Paso : Programa

```

10 REM <QUIM-3>-CALCULOS QUIMICOS
20 KEY OFF:SCREEN 0:COLOR 1,15,15:CLS:WIDTH 40
30 ? "PROGRAMA PARA CALCULOS ESTEQUIOMETRICOS."
40 ? " PROF. DR. W. VOSS, 1984"
45 ?:" ";
46 FOR X=2 TO 37
47 ? "-";
48 NEXT X
50 DIM A$(10),B$(10)
60 ?:" " Ejemplo :":?:?
70 ? " ZINC + AZUFRE = SULFURO DE ZINC : "
80 ?:" Zn + S = Zn S":?
90 ?:" "Cuántos gramos de azufre son necesarios"
100 ? "'para transformar completamente 100 g de'"
110 ? "'Zinc en Sulfuro de Zinc?':?
120 INPUT "1. ELEMENTO : ";A$
130 INPUT "2. ELEMENTO : ";B$
140 ?:"PESO ATOMICO DEL ";A$:LOCATE 30,21:? "= ";:
    INPUT A
150 ? "PESO ATOMICO DEL ";B$:LOCATE 30,22:? "= ";:
    INPUT B
160 X=B*100/A
170 ?:"SE NECESITAN ";CSNG(X);" gr. de ";B$
180 ?:" " FINAL DEL CALCULO":END
    
```

5. PASO : Lista de variables

A = Peso atómico elemento A  
 A\$ = Nombre del elemento A  
 B = Peso atómico elemento B  
 B\$ = Nombre del elemento B  
 X = Cantidad buscada



Voss	Capítulo	3 : Química	Página
	Punto	4 : Estequiometría	120
Colegio			

## 6. PASO : Descripción del programa

- Línea 10-110 : Título y visualización del ejemplo del que se ocupa el programa
- Línea 50 : Dimensionado de las cadenas
- Línea 120-130 : Introducción del nombre de ambas sustancias
- Línea 45-48 : Bucle de impresión de una línea
- Línea 140-150 : Impresión de los nombres de las sustancias y demanda de los pesos atómicos de las mismas
- Línea 160 : Cálculo de la cantidad buscada
- Línea 170 : Impresión del resultado
- Línea 180 : Final del programa

Voss	Capítulo	3 : Química	Página
	Punto	4 : Estequiometría	121
Colegio			

## 7. PASO : Resultados

Después de las aclaraciones, el programa pide el nombre del primer elemento.

Si introducimos p.e. "azufre", entonces nos pide el nombre del segundo elemento que deberá considerarse en el cálculo.

Si introducimos p.e. "zinc", el programa nos obliga a introducir los pesos atómicos del azufre y del zinc.

Si en respuesta introducimos los valores 32.1 y 65.4, el programa nos ofrece el siguiente resultado:

SE NECESITAN 203.738 gr. de Zinc

FINAL DEL CALCULO

Se entiende que el programa debe ser modificado profundamente en determinados casos, cuando se trata de procesos químicos distintos a este (p.e. también cuando participan más de dos sustancias en las reacciones del problema).

En este caso hemos procurado únicamente presentar a modo de ejemplo un programa de este tipo.

### 3.5 El sistema periódico de los elementos

---

#### 1. PASO : Presentación del problema

No sólo podemos utilizar un ordenador para efectuar cálculos como el del ejemplo precedente o para generar gráficos ilustrativos (como en el punto anterior), sino que puede utilizarse además -en el ámbito escolar- precisamente como "fichero de consulta".

El ejemplo del sistema periódico de los elementos ilustra el posible aspecto que cubre un "programa de consulta" de este tipo, que también puede entenderse como programa de entreno.

El usuario puede desear, por ejemplo, que el ordenador le dé informaciones básicas acerca de un elemento previamente introducido (p.e. el azufre).

Estas informaciones pueden ser por ejemplo:

1. Símbolo químico,
2. Número de orden en el sistema periódico,
3. Peso atómico,
4. Peso específico
5. Punto de fusión,
6. Punto de evaporación,
7. Grupo (I-VIII,0),
8. Subgrupo (a,b, ),
9. Capas de electrones (K-Q)

#### 2. PASO : Análisis del problema


El desarrollo de un programa que nos pueda proporcionar estas informaciones no supone dificultades especiales.

El usuario indicará a través del INPUT el elemento del que desea obtener informaciones y el programa pondrá a su disposición el registro completo asignado a este elemento, extrayéndolo de los DATAS mediante la Instrucción READ.

Finalmente puede consultar si se desean obtener las mismas informaciones de otro elemento químico. Si es así, puede accederse nuevamente al fichero mediante RESTORE.

3. PASO : Diagrama de flujo

El diagrama de flujo es igual de sencillo que el análisis del problema efectuado en el párrafo anterior:



4. PASO : Programa

```

10 REM <QUIM-4> -ELEMENTOS-
20 COLOR 1,15,15:SCREEN 0:WIDTH 40:CLS
30 PRINT " PROGRAMA PARA INDICAR CARACTERISTICAS"
40 PRINT " FUNDAMENTALES DE LOS ELEMENTOS QUIMICOS."
50 PRINT "          PROF. DR. W. VOSS, 1984"
55 REM NUMERO DE ELEMENTOS REGISTRADOS
60 N=16
70 DIM ES(20),ABS(2),GS(4),NS(1),SS(1)
75 DIM AS(1),BS(20)
76 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT:PRINT "PON EL NOMBR
E EN MAYUSCULAS"
77 PRINT:PRINT:PRINT:PRINT
80 INPUT "Nombre del elemento ? : ";BS
90 RESTORE
100 FOR I=1 TO N
110 READ ES,ABS,G,W,S1,S2,GS,NS,SS
120 IF ES=BS THEN 150
130 NEXT I
140 PRINT:PRINT:PRINT "ELEMENTO NO REGISTRADO":GOTO 280
150 CLS
160 PRINT "          ";ES:LOCATE 27,0:PRINT ABS
170 PRINT "          -----":PRINT
180 PRINT "          Número atómico   = ";G
190 PRINT "          Grupo              = ";GS
200 PRINT "          Subgrupo           = ";NS
210 PRINT "          Capa                = ";SS
220 PRINT:PRINT
230 PRINT "          Peso atómico       = ";G
240 PRINT "          Peso específico    = ";W
250 PRINT:PRINT
260 PRINT "          Punto fusión       = ";S1
270 PRINT "          Punto evaporación = ";S2
280 PRINT:PRINT:PRINT:INPUT "          OTRO ELEMENTO? (S/N) ";A
$
290 IF AS="S" OR AS="s" THEN CLS:GOTO 80
300 PRINT:PRINT:PRINT "          FIN DEL PROGRAMA":END
310 DATA ALUMINIO,A1,13,26.97,2.7,658,2270,III,B,M
320 DATA BARIO,Ba,56,137.36,3.5,704,1537,II,A,P
330 DATA PLOMO,Pb,82,207.21,11.34,327,1690,IV,B,P
340 DATA CLORO,Cl,17,35.457,1.557,-100,-34,VII,B,M
    
```

Voss	Capítulo	3 : Química	Página
	Punto	5 : Sistema periódico	126
Colegio			

```

350 DATA HIERRO, Fe, 26, 55.85, 7.86, 1525, 2450, VIII, O, N
360 DATA FLUOR, F, 9, 19, .0017, -218, -187, VII, B, L
380 DATA ORO, Au, 79, 197, 19.25, 1063, 2677, I, B, P
390 DATA HELIO, He, 2, 4.003, .000018, -272, -269, O, O, K
400 DATA IODO, I, 53, 126.92, 4.942, 114, 184, VII, B, O
410 DATA CADMIO, Cd, 48, 112.41, 8.64, 321, 770, II, B, O
420 DATA SODIO, Na, 11, 22.991, .971, 98, 880, I, A, M
430 DATA FOSFORO, P, 15, 30.98, 1.83, 44, 280, V, B, M
440 DATA OXIGENO, O, 8, 16, .0014, -218.7, -182.97, VI, B, L
450 DATA AZUFRE, S, 16, 32.066, 2.07, 112, 444, VI, B, M
460 DATA HIDROGENO, H, 1, 1.008, .00009, -262, -252.78, I, A, K
470 DATA ZINC, Zn, 30, 65.38, 7.14, 419, 906, II, B, N

```

Voss	Capítulo	3 : Química	Página
	Punto	5 : Sistema periódico	127
Colegio			

Debe recordarse que el fichero de datos de este programa contiene únicamente 16 elementos. No obstante, puede descubrirse con facilidad el modo de variar el programa cuando quieran incorporarse más de 16, o quizá todos los elementos químicos conocidos.

#### 5. PASO : Lista de variables

A\$ = Cadena de respuesta (si/no)  
AB\$ = Abreviación del elemento  
B\$ = Campo para acoger el elemento buscado por el usuario  
E\$ = Nombre del elemento  
G = Peso atómico  
G\$ = Grupo a que pertenece el elemento  
I = Índice variable  
N = Número de elementos del registro  
N\$ = Subgrupo  
O = Número de orden  
S1 = Punto de fusión  
S2 = Punto de evaporación  
W = Peso específico

Voss	Capítulo	3 : Química	Página
	Punto	5 : Sistema periódico	128
Colegio			

6. PASO : Descripción del programa

Línea 10-50 : Título

Línea 60 : Indicación del número de elementos (este programa no incluye todos los elementos químicos, sino sólo 16. Dado el caso habrá que modificar la línea 60 si se quieren incorporar más DATAS)

Línea 70-75 : Dimensionado

Línea 76-80 : Demanda del elemento del que se desea Información

Línea 90 : Restauración del registro (para eventualmente repetir el proceso ver línea 280)

Línea 100-130: Lectura de los registros

Línea 120 : Si se ha hallado el elemento, salto hacia 150

Línea 140 : La línea 140 se alcanza únicamente cuando el elemento buscado no se encuentra en el fichero; después se salta a la línea 280

Línea 150-270: Impresión de las informaciones una vez borrada la pantalla

Voss	Capítulo	3 : Química	Página
	Punto	5 : Sistema periódico	129
Colegio			

Línea 280-290 : Consulta si se desea consultar otro fichero. En caso afirmativo borrar la pantalla y saltar a la línea 80. En caso negativo seguir en 300.

Línea 300 : Finalización del programa

Líneas 310 y siguientes : Fichero en forma de DATAS

7. PASO : Resultados

Introduzcamos por ejemplo el elemento azufre, a lo cual el ordenador contestará:

	AZUFRE	S
Número atómico	= 16	
Grupo	= VI	
Subgrupo	= B	
Capa	= M	
Peso atómico	= 32.066	
Peso específico	= 2.07	
Punto de fusión	= 112	
Punto evaporación	= 444	
OTRO ELEMENTO (S/N)		?

# 4

## Capítulo 4 : Física

---

### 4.1 Consideración previa

---

El campo de la física a nivel escolar, ofrece problemas similares a los que ya conocemos del capítulo dedicado a las matemáticas (ver capítulo 2). En todo caso, aquí deben considerarse ciertas particularidades que nos obligan a hacer uso nuevamente de las posibilidades gráficas del ordenador.

Por esta razón - y ampliando lo dicho en el capítulo anterior con respecto a los métodos de programación de gráficos - en esta introducción profundizaremos otra vez en las instrucciones gráficas del BASIC.

#### 4.2 Programación de gráficos

---

Un ordenador MSX ofrece dos opciones distintas para la creación de gráficos:

1. Gráficos normales (llamados también gráficos de bloque
2. Gráficos de alta resolución

Para poder utilizar uno u otro tipo de representación gráfica es necesario cambiar de modo la pantalla, que hasta ahora era el llamado modo texto, utilizado para imprimir caracteres en pantalla.

Se hace necesaria entonces la siguiente instrucción (que también puede utilizarse como comando en el modo directo) :

Instrucción 17:

---

nn SCREEN (modo),(tamaño sprite),(sonido teclas),  
(velocidad en baudios),(tipo impresora)

En el submandato "modo" de esta instrucción puede aparecer un valor situado entre 0 y 3, pudiéndose observar resultados muy diversos según el valor que seleccionemos.

Para nuestros fines basta con que distingamos entre las cuatro posibilidades siguientes:

1. SCREEN 0 : Texto con 40x24 caracteres.
2. SCREEN 1 : Texto con 32x24 caracteres.
3. SCREEN 2 : Conmuta al modo gráfico de alta resolución.
4. SCREEN 3 : Conmuta al modo de gráficos multicolores

En el modo SCREEN 0 hay que imaginarse la pantalla dividida en 40 columnas (0 a 39) y 24 líneas (0 a 23).

La línea 0 está en la parte superior de la pantalla y, la columna 0, en el extremo izquierdo de la misma.

Voss	Capítulo	4 : Física	Página
	Punto	2 : Gráficos	134
Colegio			

En la modalidad SCREEN 1, los caracteres tienen una anchura de 8 pixels, quedando dividida de este modo la pantalla en 32 columnas y 24 líneas. Esto permite una más cómoda visualización del texto.

En la modalidad SCREEN 2, podremos hacer uso de los mandatos BASIC de trazado de gráficos, a estos efectos, la pantalla quedará dividida en una matriz de 256x192 pixels, numerados del 0 al 255, y del 0 al 191, siendo la posición del pixel (0,0), la esquina superior izquierda de la pantalla o monitor.

En modo SCREEN 3, disponemos de más libertad a la hora de dar color de forma individual a cada pixel, aunque a costa de una menor resolución, ya que en esta modalidad, la unidad para el trazado de gráficos es un bloque de 4x4 puntos. La ubicación de los mismos se especifica utilizando los puntos 0 a 255 para las coordenadas horizontales y 0 a 191 para las verticales.

El color de la pantalla o monitor queda definido mediante la siguiente instrucción:

Instrucción 18:

nn COLOR (tinta),(papel),(Borde)

Los colores se seleccionan mediante el uso de códigos numéricos de acuerdo con la siguiente tabla:

Código	Color	Código	Color
0	Transparente	8	Rojo
1	Negro	9	Rojo claro
2	Verde	10	Amarillo oscuro

Voss	Capítulo	4 : Física	Página
	Punto	2 : Gráficos	135
Colegio			

Código	Color	Código	Color
3	Verde claro	11	Amarillo claro
4	Azul oscuro	12	Verde oscuro
5	Azul claro	13	Magenta
6	Rojo oscuro	14	Gris
7	Azul celeste	15	Bianco

Tanto los caracteres como los gráficos se visualizarán con el color especificado para la tinta. En modo SCREEN 0, el color del borde será siempre igual al del papel.

En el modo gráfico de gran definición el color podrá especificarse individualmente para cada pixel, sin embargo, solo podrá darse un color por cada bloque horizontal de ocho pixels.(cada fila de pixels de la pantalla está dividida en 32 bloques horizontales de 8 pixels)

En el modo de gráficos multicolores no tenemos la anterior limitación, podemos dar un color diferente a cada bloque de 4x4 pixels, aunque a costa de una menor definición en pantalla.

Las instrucciones para el trazado de gráficos son las siguientes:

PSET, PRESET, LINE, CIRCLE, PAINT y DRAW

Además, disponemos de un macrolenguaje específico para el trazado de gráficos, cuyas instrucciones se asemejan a las del lenguaje LOGO.



Instrucción 19:

nn PSET (coordenada X, coordenada Y), (color)

Marca un píxel en las coordenadas y color especificado

Instrucción 20:

nn PRESET (coordenada X, coordenada Y), (color)

Marca o borra un píxel

Instrucción 21:

nn LINE (coordenada X, coordenada Y)-(coordenada X',  
coordenada Y'), (color), (B ó BF)

Traza una línea entre el punto de partida (X,Y) y el (X',Y'). Además en caso de especificarse el submandato B, se trazará un cuadrado que conectará diagonalmente los dos puntos especificados.

Cuando se especifique BF, se trazará un rectángulo con una línea recta que conectará diagonalmente los dos puntos especificados, y coloreará el área limítrofe.

Instrucción 22:

nn CIRCLE (X,Y), (radio), (color), (ángulo inicial),  
(ángulo final), (excentricidad)

Traza un círculo, un sector circular o una elipse, de acuerdo con los submandatos especificados.

Instrucción 21

nn PAINT (X,Y), (color), (color de la línea limítrofe)

Colorea el área encerrada por la línea limítrofe.

Instrucción 22

nn DRAW ''submandatos de gráficos''

Traza gráficos de acuerdo con los submandatos especificados. Refiérase a su manual de programación para consultar los submandatos.

4.3 Programas BASIC

```

10 REM
20 REM      -CUADRADO-
30 REM
40 SCREEN 2;COLOR 15,1,1:CLS
50 LINE (100,60)-(150,130),15,BF
60 A$=INPUT$(1)
70 REM
80 REM      -PIXEL EN SCREEN 3-
90 REM
100 SCREEN 3:CLS
110 PSET (128,96),10
120 A$=INPUT$(1)

```

```

130 REM
140 REM      -RECTAS AL AZAR-
150 REM
160 SCREEN 2:CLS
170 X=INT(RND(1)*255);Y=INT(RND(1)*191);C=INT(RND(1)*15)
180 LINE (128,96)-(X,Y),C
190 GOTO 170
200 REM.
210 REM      -Ejes de coordenadas-
220 REM
230 SCREEN 2:COLOR 15,1,1:CLS
240 DRAW "BMO,95R255BM127,0D191"
250 S$=INPUT$(1)
260 REM - "BMO,95R255BM127,0D191" - es el argumento de
DRAW, y en este caso, tiene el siguiente significado:
Pon el cursor de píxel en la posición X=0, Y=95; a
partir de la misma, traza hacia la derecha una línea
recta hasta la coordenada 255, es decir, traza hacia la
derecha una línea recta de longitud = 256 píxels. Pon
de nuevo al cursor de píxel en la posición X=127,
Y=0; a partir de esta posición, traza hacia abajo una
línea recta hasta la coordenada 191. No olvide que la
pantalla en el modo de gran definición está dividida
en un total de 256x192 = 49152 píxels, o unidades ele
mentales de imagen.

```

Para conseguir que en el modo gráfico la imagen se mantenga estable deberemos de hacer entrar al ordenador en un bucle sin fin. Para salir del mismo deberemos pulsar a la vez CTRL y STOP.

Este programa dibuja píxel a píxel y en diferentes colores un cuadrado. Fíjate que aunque se intenta dibujar cada píxel en un color diferente, esto no es posible en el modo de gran resolución, lo único que se logra es colorear diferenciadamente bloques horizontales de 8 píxels.

```

10 SCREEN 2:COLOR 15,1,1:CLS
20 FOR S=80 TO 173
30 FOR Z=47 TO 155
40 F=INT(RND(1)*15)
50 PSET (S,Z),F
60 NEXT Z
70 NEXT S
80 A$=INPUT$(1)

```

El siguiente programa traza una trama reticular en la pantalla del televisor:

```

10 SCREEN 2:COLOR 15,1,1:CLS
20 FOR X=0 TO 255 STEP 5
30 PSET (X,0)
40 LINE -(X,190)
50 NEXT X
60 FOR Y=0 TO 191 STEP 5
70 PSET (0,Y)
80 LINE -(255,Y)
90 NEXT Y
100 A$=INPUT$(1)

```

El próximo programa traza aleatoriamente una trama de rectángulos de distintos tamaños.

```

10 SCREEN 2:COLOR 15,1,1:CLS
20 FOR I=1 TO 140
30 S=RND(1)*215
40 S1=S
50 Z=RND(1)*160
60 Z1=Z
70 L=RND(1)*70
80 B=RND(1)*55
90 PSET (S,Z)
100 S=S+L
110 LINE -(S,Z)
120 Z=Z+B
130 LINE -(S,Z)
140 S=S-L
150 LINE -(S,Z)
160 Z=Z-B
170 LINE -(S1,Z1)
180 NEXT I
190 A$=INPUT$(1)

```

El siguiente programa traza tres columnas horizontales de distintos colores:

```

10 SCREEN 2:COLOR 15,1,1:CLS
20 DRAW "BM10,0C10R50D192L50U192"
30 DRAW "BM107,0C8R50D192L50U192"
40 DRAW "BM200,0C12R50D192L50U192"
50 PAINT (35,96),10:PAINT (132,96),8:PAINT (225,96),12
60 A$=INPUT$(1)

```

El siguiente programa dibuja un rectángulo y sus diagonales:

```

10 SCREEN 2:COLOR 15,1,1:CLS:KEY OFF
20 DRAW "BMO,OR255D191L255U191M255,191BM255,OMO,191"
30 A$=INPUT$(1)

```

Este otro sirve para ilustrar el uso de la sentencia CIRCLE, produciendo una serie de cónicas de forma aleatoria, y conformando en pantalla, un bonito efecto artístico.

```

10 REM EJEMPLO DE LA SENTENCIA CIRCLE
20 SCREEN 0:WIDTH 40:CLS
30 COLOR 15,1,1:KEY OFF
40 LOCATE 11,12
50 PRINT "PULSE UNA TECLA"
60 A$=INPUT$(1)
70 SCREEN 2:CLS
80 FOR I=1 TO 1000
90 X=RND(TIME)*255
100 Y=RND(TIME)*191
110 N=RND(TIME)*15
120 CIRCLE (X,Y),RND(TIME)*40,N,,RND(TIME)*2
130 PAINT (X,Y),N
140 NEXT I
150 LOCATE 11,12:?"FIN DEL PROGRAMA"

```

El programa siguiente imprime en pantalla y con distintos colores, 1000 puntos. Se consigue de este modo un efecto semejante al que producen las estrellas en la esfera celeste.

```

10 REM      -ESTRELLAS-
20 SCREEN 2:CLS
30 FOR I=1 TO 1000:
40 X=255*RND(1)
50 Y=191*RND(1)
60 C=15*RND(1)
70 PSET (X,Y),C
80 NEXT I
90 A$=INPUT$(1)

```

Podemos conseguir distintos efectos gráficos en pantalla mediante el programa siguiente:

```

10 A=15:B=1
20 GOSUB 110
30 A=2:B=10
40 GOSUB 110
50 A=7:B=11
60 GOSUB 110
70 A=13:B=10
80 GOSUB 110
90 END
100 A=6:B=4
110 SCREEN 2:CLS
120 FOR Y=60 TO 75
130 FOR X=81 TO 173
140 IF C=A THEN C=B ELSE C=A
150 PSET (X,Y),C
160 NEXT X
170 NEXT Y
180 RETURN

```

Habrás observado sin duda alguna, que en muchos de los programas que ilustran este capítulo, aparece la instrucción A\$=INPUT\$(1).

Esta instrucción tiene como misión retener el formato de gráficos de gran resolución, induciendo al ordenador a permanecer en estado de espera, o mejor dicho, en un bucle sin salida, mientras que no se aprete una tecla cualquiera. Esta instrucción pues, espera a que se introduzca un carácter a través del teclado y mientras esto no ocurre, retiene la ejecución del programa en el ordenador.

Si no induciésemos de algún modo al ordenador a entrar en estado de espera o en un bucle sin salida, no lograríamos llegar a visualizar nada en cualquiera de los dos modos gráficos, ya que el ordenador pasa automáticamente a modo texto después de la ejecución de un programa.

Se podría conseguir también el mismo efecto incluyendo en el programa una línea del tipo:

```
nn GOTO nn
```

Lo que haría que el ordenador cayese en un bucle sin fin. Para salir del mismo será necesario provocar un BREAK mediante el uso de las teclas CTRL y STOP.

#### 4.4 El dinamómetro

##### 1. PASO : Presentación del problema

Si colgamos diferentes objetos de un dinamómetro, éste se estira en mayor o menor grado, según sea el peso de los cuerpos.

Así se cumple una ley de proporcionalidad que dice: si doblamos el peso, el alargamiento del dinamómetro se duplica.


Existe también una llamada constante elástica, que es igual al cociente entre el peso  $F$  y la deformación del muelle  $s$ , es decir:

$$D = F/s = \text{constante}$$

Si, por ejemplo,  $D = 20$ , una fuerza de 100 (cN) producirá un alargamiento de 5 cm, y una fuerza de 200 (cN) un alargamiento del muelle de 10 cm etc.

##### 2. PASO : Análisis del problema


La relación antes descrita puede ser representada gráficamente (p.e. para  $D = 20$ ):



El programa BASIC a desarrollar, debe imprimir el gráfico correspondiente a diversos valores de la constante de muelle  $D$  en la pantalla, siendo necesario estructurarlo del modo siguiente:

- Trazado de unos ejes de coordenadas,
- introducción de un valor arbitrario para  $D$ ,
- impresión de una recta de constancia (ver arriba),
- contestación a la pregunta referente al alargamiento del dinamómetro tras introducir un peso cualquiera.

3. PASO : Diagrama de flujo



4. Paso : Programa

```

10 REM PHYSICS/1 -DINAMOMETRO-
20 SCREEN 0:COLOR 1,15,15:WIDTH 40:KEY OFF:CLS
30 PRINT:PRINT " PROGRAMA PARA ILUSTRAR GRAFICAMENTE LA"
40 PRINT:PRINT " LEY DE HOOKE."
50 PRINT:PRINT:PRINT " -----"
60 PRINT " PROF. DR. W. VOSS, 1984"
70 PRINT " -----"
80 CLOSE:A-D=0
90 LOCATE 0,16:INPUT " Valor de la constante elástica : ";
D
100 IF D<1 THEN CLS:LOCATE 4,11:PRINT "Constante elástica i
ncorrecta":GOSUB S10
110 LOCATE 0,19:INPUT " Fuerza aplicada al muelle : ";
F
120 IF F>999 THEN CLS:LOCATE 6,11:PRINT "Es malo hacer tant
a fuerza":GOSUB S40
130 IF F<=0 THEN 110
140 SCREEN 2:COLOR 15,1,1:CLS
150 DRAW "BM40,140R165BM40,140U140BM40,141"
160 DRAW "R165BM41,141U141"
170 FOR X=41 TO 185 STEP 5
180 FOR Y=20 TO 142 STEP 5
190 PSET (X,Y),11
200 NEXT Y
210 NEXT X
220 FOR X=41 TO 185 STEP 20
230 FOR Y=142 TO 146
240 PSET (X,Y),15
250 NEXT Y
260 NEXT X
270 FOR X=36 TO 41
280 FOR Y=140 TO 10 STEP -20
290 PSET (X,Y),15
300 NEXT Y
310 NEXT X
320 OPEN "GRP:" FOR OUTPUT AS#1
330 PSET (13,149),1:PRINT #1,"0,0"
340 S=(F/D)
350 LINE (41,141)-(181,20),8
360 LINE (41,140)-(180,20),8
  
```

Voss	Capítulo	4 : Física	Página
	Punto	4 : Dinamómetro	148
Colegio			

```

370 LINE (42,141)-(182,20),8
380 PSET (164,149),1:PRINT #1,USING "##.##";S
390 PSET (206,149),1:PRINT #1,"cm"
400 PSET (0,17),1:PRINT #1,F
410 PSET (13,0),1:PRINT #1,"cN"
420 PSET (181,2),1:PRINT #1,"D="
430 PSET (190,2),1:PRINT #1,D
440 PSET (45,182),1:PRINT #1,"Otra gráfica (S/N) ?"
450 AS=INKEYS
460 IF AS="" THEN 450
470 IF AS="S" OR AS="s" THEN 20
480 SCREEN 0:WIDTH 40:COLOR 15,1,1
490 LOCATE 11,10:PRINT "FIN DEL PROGRAMA"
500 END
510 TIME=0
520 A=INT(TIME/50)
530 IF A=2 THEN 20 ELSE 520
540 TIME=0
550 A=INT(TIME/50)
560 IF A=2 THEN CLS:GOTO 110 ELSE 550

```

5. PASO : Lista de variables

A = Registro del reloj  
A\$ = Cadena de respuesta (sí/no)  
D = Constante elástica  
TIME = Reloj del sistema  
F = Fuerza aplicada al muelle  
S = Recorrido del muelle  
X,Y = Índices variables

Voss	Capítulo	4 : Física	Página
	Punto	4 : Dinamómetro	149
Colegio			

6. PASO : Descripción del programa

Línea 10-70 : Título

Línea 80 : Cerrar canales e Inicializar variables

Línea 90 : Introducción de la constante elástica D

Línea 100 : Se comprueba que D es correcta

Línea 110 : Demanda de la fuerza F.

Línea 120-130: Se comprueba si F es correcta y si sobrepasa el formato de Impresión.

Línea 140 : Establece el modo de gráficos de gran definición.

Línea 150-310: Dibujo de los ejes y escalas del gráfico.

Línea 320 : Apertura del canal de Impresión de caracteres en modo gráfico.

Línea 340 : Cálculo del recorrido del muelle.

Línea 350-370: Trazado del gráfico.

Línea 380-440: Impresión de caracteres complementarios.

Línea 450 : Se espera la entrada de un caracter.

Voss	Capítulo	4	: Física	Página
	Punto	4	: Dinamómetro	150
Colegio				

Línea 460-470: Se establece si se ha pulsado una opción u otra. Si se ha pulsado otra tecla o no se ha pulsado ninguna se vuelve a la línea 450.

Línea 480 : Se vuelve al modo texto (40 columnas).

Línea 490-500: Fin del programa.

Línea 510 : Pone a cero el reloj del sistema.

Línea 520-530: Bucle temporizador.

Línea 540-560: Otro bucle temporizador (Se repite el mismo proceso que en las líneas anteriores)

#### 7. PASO : Resultados

Después de introducir la Información INPUT, el programa genera un gráfico tal como el que ya presentamos esquemáticamente en una de las páginas precedentes.

Después de una pausa, el programa pide la fuerza  $F$  y calcula el correspondiente recorrido de muelle. Sobra aquí cualquier ejemplo de cálculo (comparar con el 1. PASO, que contiene ya un ejemplo).

Voss	Capítulo	4	: Física	Página
	Punto	5	: Principio de Arquímedes	151
Colegio				

#### 4.5 El principio de Arquímedes

---

##### 1. PASO : Presentación del problema

El principio de Arquímedes dice lo siguiente:

El empuje que recibe un cuerpo sumergido en un fluido es igual al peso del líquido que desplaza.

Este principio describe el hecho conocido de que la fuerza necesaria para apretar hacia abajo un cubo vacío dentro de un líquido aumenta progresivamente.

A medida que un cuerpo se sumerge en un líquido, éste se ve sometido a una pérdida de peso aparente cada vez mayor, debida al creciente empuje.

Presentamos a continuación un programa que, basándose en este principio, calcula el volumen y la densidad (peso específico) de un cuerpo desconocido sumergido en un líquido conocido (p.e. agua), una vez se conoce su peso en el aire.



## 2. PASO : Análisis del problema

Como en todos los demás problemas, aquí se procederá de acuerdo con el probado esquema:

E = Entrada

P = Proceso

S = Salida

Es decir, en primer lugar estudiaremos qué informaciones de entrada necesita el ordenador si debe resolver el problema antes formulado. Necesita lo siguiente:


- E1. El peso del cuerpo desconocido en el aire;
- E2. el peso aparente del cuerpo dentro del fluido, en el que se sumerge;
- E3. el peso específico del líquido utilizado para la comprobación.

Cuando ya disponemos de estas informaciones, puede determinarse mediante el proceso lo siguiente:

- V1. El empuje y el peso de la cantidad de líquido desplazada;
- V2. el volumen de esta cantidad de líquido;
- V3. el volumen del cuerpo desplazante de prueba;
- V4. El peso específico del cuerpo de prueba sumergido.

Finalmente deberán imprimirse como resultados los obtenidos en los últimos puntos V3 y V4.

## 3. PASO : Diagrama de flujo



#### 4. PASO : Programa

```

10 REM <PHYSICS-2> -ARQUIMEDES-
20 SCREEN 0:COLOR 15,1,1:WIDTH 40:KEY OFF:CLS
30 ? " PRINCIPIO DE ARQUIMEDES"
40 ? " _____"
50 ? " PROF. DR. W. VOSS, 1984"
60 ?::? "Con el principio de Arquímedes pueden deter-
minarse el volumen y la densidad de un cuerpo desconoci-
do. Introduzca las siguientes informaciones:"
70 ?::?INPUT "PESO DEL CUERPO EN EL AIRE : ";PA
80 ?::?INPUT "PESO APARENTE DEL CUERPO EN EL FLUI-
DO : ";PF
90 INPUT "DENSIDAD DEL FLUIDO : ";DF
100 PP=PA-PF
110 VD=PP/DF
120 VC=VD
130 D=PA/VC
140 ?::? "LOS RESULTADOS DEL ESTUDIO SON : "
150 ?::? "Volumen del cuerpo = ";VD
160 ?::? "Densidad del cuerpo = ";D
170 ?::?::? " OTRA VEZ (S/N) ?"
180 A$=INKEY$
190 IF A$="S" OR A$="s" THEN 20
200 IF A$="N" OR A$="n" THEN 210 ELSE 180
210 ?::?::? " FIN DEL PROGRAMA":END

```

#### 5. PASO : Lista de variables

A\$ = Variable de cadena para las respuestas sí/no  
PA = Peso del cuerpo en el aire  
PF = Peso aparente del cuerpo en el fluido  
PP = Pérdida aparente de peso  
VC = Volumen del cuerpo  
VD = Volumen del líquido desplazado  
DF = Peso específico del líquido  
DC = Peso específico del cuerpo

#### 6. PASO : Descripción del programa

Línea 10-60 : Título y anotaciones aclaratorias

Línea 70-90 : Demanda de las informaciones INPUT

Línea 100-130 : Cálculos del resultado

Línea 140-160 : Impresión de los resultados

Línea 170-200 : Consulta si se quiere repetir el problema  
Si la respuesta es positiva va a 20, si no  
va la 210 (END)

Línea 210 : Final del programa

Voss	Capítulo	4 : Física	Página
	Punto	5 : Principio de	156
Colegio		Arquímedes	

7. PASO : Resultados

Si introducimos por ejemplo las siguientes informaciones  
INPUT:

E1: Peso del cuerpo = 70 gr  
 E2: Peso aparente del cuerpo en el fluido de prueba (a medir con un dinamómetro) = 40 gr  
 E3: Peso específico del fluido de prueba (agua) = 1

El programa genera estos resultados:

LOS RESULTADOS DEL ESTUDIO SON:

Volumen del cuerpo = 30

Densidad del cuerpo = 2.333333333333

OTRA VEZ (S/N) ?


Voss	Capítulo	4 : Física	Página
	Punto	6 : Péndulo	157
Colegio			

4.6 Movimiento pendular

1. PASO : Presentación del problema

El movimiento pendular es un fenómeno físico básico que juega un papel importante en la acústica, la óptica y en la electricidad.

Los movimientos pendulares siguen una oscilación sinusoidal, como se muestra en el esquema contiguo.



Presentamos ahora un programa, que representa gráficamente una oscilación sinusoidal para diversas amplitudes (A) y periodos (P), que a su vez determinan la llamada frecuencia (F) de cualquier oscilación.

2. PASO : Análisis del problema

Para el programa previsto deben introducirse la amplitud y el periodo de la oscilación como informaciones INPUT.


El programa puede ocuparse entonces de que la oscilación sinusoidal "pase", recorriendo la pantalla de derecha a izquierda.

Fundamentalmente, el contenido del programa debe centrarse en la normalización de la amplitud y del período para aprovechar óptimamente la representación gráfica de la oscilación sobre la pantalla.

Esto se consigue colocando unos ejes coordenados que dividen la pantalla en dos zonas del mismo tamaño. El cálculo de los parámetros de la oscilación está optimizado para que no se sobrepasen los límites de la pantalla.

Como puede deducirse de la distribución que acabamos de realizar, seleccionamos el modo gráfico de alta resolución (ver Cap. 9), lo que nos permitiría operar con un margen de 256\*192 posibles parámetros.

3. PASO : Diagrama de flujo



Voss	Capítulo	4 : Física	Página
	Punto	6 : Péndulo	160
Colegio			

4. PASO : Programa

```

10 REM "PHYS/3" -PENDULO-
20 COLOR 1,15,15:SCREEN 0:WIDTH 40:KEY OFF:CLS:CLOSE
30 ? "PROGRAMA PARA LA SIMULACION DE UN MOVI-"
40 ? "MIENTO VIBRATORIO ARMONICO SIMPLE."
50 ??:? " PROF. DR. W. VOSS, 1984"
60 ??:?:?:?:?
70 ? "Para salir del programa pulse CTRL+STOP":?:?:?
80 LOCATE 0,17:INPUT " AMPLITUD DE LA OSCILACION : ";A
90 IF A>75 OR A=0 THEN 80
100 LOCATE 0,20:INPUT " PERIODO DE LA OSCILACION : ";P
110 IF P=0 OR P>200 THEN 100
120 IF P<20 THEN PS=.314159265# ELSE PS=1
130 SCREEN 2:COLOR 7,1,1:CLS
140 DRAW "BM24,95R200BM24,0D191"
150 OPEN "GRP:" FOR OUTPUT AS #1
160 PSET (10,93),1:COLOR 15:? #1, "0"
170 PSET (10,0),1:? #1, "A"
180 PSET (10,183),1:? #1, "A"
190 PSET (239,93),1:? #1, "+"
200 FOR T=0 TO 199 STEP PS
210 Y=A*COS((2*3.141592654#/P)*T)
220 PSET (T+24,95-Y),9
230 NEXT T
240 PSET (62,183),1:COLOR 15:? #1, "Otra vez (s/n) ?"
250 A$=INKEY$
260 IF A$=" " THEN 250
270 IF A$="S" OR A$="s" THEN 20
280 IF A$="N" OR A$="n" THEN SCREEN 0:LOCATE 10,12:? "F
IN DEL PROGRAMA" ELSE 250

```

Voss	Capítulo	4 : Física	Página
	Punto	6 : Péndulo	161
Colegio			

6. PASO : Descripción del programa

Línea 10-70 : Título y aclaraciones

Línea 80-100 : Demanda de Informaciones INPUT

Línea 120 : Optimización de los parámetros de P (período).

Línea 130 : Modo gráfico de gran definición y borrado de la pantalla.

Línea 140 : Impresión de los ejes del gráfico.

Línea 150 : Apertura del canal de Impresión de caracteres en el modo de gran resolución.

Línea 160-190 : Impresión de caracteres en pantalla.

Línea 200-230 : Bucle para determinar e imprimir la posición de cada punto de la onda. El cálculo matemático se realiza dentro de la línea 210.

Línea 240 : Impresión de "Otra vez...." en la parte inferior de la pantalla.

Línea 250 : Espera a que se introduzca un carácter.

Línea 260-280 : Se determina que carácter se ha entrado desde el teclado. Si la respuesta es positiva se vuelve a reinicializar el programa y se va a la línea 20. Si la respuesta es negativa se borra la pantalla y se imprime "FIN DEL PROGRAMA"

#### 7. PASO : Resultados

Los resultados que ofrece este programa son análogos al croquis del 1. PASO, con la diferencia de que este gráfico no es apaisado.


Después de pedir los valores de la amplitud y el periodo de la oscilación, se produce un borrado de la pantalla, y aparecen en los ejes coordenados sobre los que se dibuja la sinusolde, que va desarrollándose paulatinamente.

Cuando esta ha aparecido en su totalidad, el programa pregunta si se quiere repetir de nuevo el proceso. Si la respuesta es negativa, se borra el gráfico y aparece impresa en la mitad de la pantalla la frase: "FIN DEL PROGRAMA". Si la respuesta ha sido positiva se vuelve a reinicializar el programa.

#### 4.7 Reproducción óptica

##### 1. PASO : Presentación del problema

Una lente convexa nos permite obtener la imagen de cualquier objeto - tal como se muestra en la ilustración siguiente:



La cámara fotográfica funciona según este principio, así como también el ojo humano.

Presentamos a continuación un programa que calcula el tamaño de la imagen y la escala de reproducción de una lente convexa para un tamaño del objeto, una distancia del mismo y una distancia focal de la lente dados.

2. PASO : Análisis del problema

El programa a realizar necesita las siguientes informaciones INPUT:

1. Tamaño del objeto,
2. distancia del objeto,
3. distancia focal de la lente.

Para resolver el problema necesitamos la llamada ecuación óptica:

$$1/f = 1/g + 1/b$$

donde: f = distancia focal  
 g = distancia del objeto  
 b = distancia de la imagen

Despejando b en esta ecuación se obtiene:

$$b = \frac{1}{1/f - 1/g}$$


Se cumple también que la escala de reproducción A es igual a:

$$A = B/G = b/g$$

Si el objeto a reproducir tiene por ejemplo un tamaño G = 20 cm, puede determinarse también el tamaño de la imagen con ayuda de las igualdades citadas:

$$B = (b \cdot G)/g, \text{ porque } B/G = b/g$$

3. PASO : Diagrama de flujo



```

10 REM "PHYS/4" -OPTICA-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA CALCULAR LA DISTANCIA, EL"
40 PRINT "TAMA";CHR$(165);"O Y LA ESCALA DE REPRODUCCION D
E LA"
50 PRINT "IMAGEN PRODUCIDA POR UNA LENTE CONVEXA."
60 PRINT CHR$(127):PRINT:PRINT " PROF. DR. W. VOSS,
1984"
70 PRINT:PRINT:PRINT:PRINT " Introduce los datos
siguientes : "
80 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT " -TAMA";CHR$(1
65);"O DEL OBJETO = ":LOCATE 30,17:INPUT G
90 PRINT:INPUT " -DISTANCIA AL OBJETO = ";GW
100 PRINT:INPUT " -DISTANCIA FOCAL = ";F
110 CLS
120 BW=1/(1/F-1/GW)
130 A=BW/GW
140 B=(G*BW)/GW
145 BW=INT(BW*100+.5)/100
146 A=INT(A*100+.5)/100
147 B=INT(B*100+.5)/100
150 COLOR 1,15,15:PRINT " DATOS DE PARTIDA : "
160 PRINT:PRINT:PRINT " -Distancia focal = ";F
170 PRINT:PRINT " -Tama";CHR$(164);"o del objeto =
";G
180 PRINT:PRINT " -Distancia al objeto = ";GW
190 PRINT:PRINT:PRINT " RESULTADOS : "
200 PRINT:PRINT:PRINT " -Escala de reproduccion = ";A
210 PRINT:PRINT " -Tama";CHR$(164);"o de la imagen =
";B
220 PRINT:PRINT " -Distancia a la imagen = ";BW
225 PRINT:PRINT
230 PRINT:PRINT " OTRO CALCULO (S/N)"
240 AS=INKEY$
250 IF AS="" THEN 240
260 IF AS="S" OR AS="s" THEN 10 ELSE PRINT:PRINT:PRINT "
FIN DEL PROGRAMA"

```

## 6. PASO : Descripción del programa

- Línea 10-70 : Título y aclaraciones
- Línea 80-105 : Demanda de las informaciones INPUT
- Línea 110-140 : Borrado de la pantalla y cálculos
- Línea 145-147 : Redondear a dos posiciones decimales
- Línea 150-225 : Impresión de los resultados
- Línea 230-240 : Consulta si se desean repetir los cálculos y espera una respuesta desde el teclado.
- Línea 250-260 : Identifica la entrada que se ha producido desde el teclado y en su caso finaliza el programa.

## 7. PASO : Resultados

Si introducimos p.e. : G = 20 cm, GW = 300 cm, F = 15 cm, obtendremos:

```

DATOS DE PARTIDA :
-Distancia focal           = 15
-Tamaño del objeto         = 20
-Distancia al objeto       = 300
RESULTADOS :
-Escala de reproducción    = .05
-Tamaño de la Imagen       = 1.05
-Distancia a la Imagen     = 15.79
OTRO CALCULOS (S/N)?


```



#### 4.8 Paralelogramo de fuerzas

##### 1. PASO : Presentación del problema

Cuando se aplican dos fuerzas sobre un punto común, se obtiene una fuerza resultante, que es la diagonal del llamado paralelogramo de fuerzas, como muestra el esquema siguiente:



En el programa que sigue se efectúa la representación gráfica del paralelogramo de fuerzas para cualquier constelación posible de dos fuerzas y se calcula la fuerza resultante  $F_3$ .

##### 2. PASO : Análisis del problema

El punto de aplicación de ambas fuerzas debe emplazarse en el centro del sistema de coordenadas definido (95,95).

Se introducen las fuerzas  $F_1$  y  $F_2$ , que en primer lugar se optimizarán de modo que el espacio de la pantalla resulte suficiente incluso en una situación poco favorable.

Deben introducirse además los ángulos  $W_1$  y  $W_2$  con los que se aplican las fuerzas  $F_1$  y  $F_2$  en el punto de aplicación.

Con ayuda del teorema de Pitágoras y de las relaciones entre ángulos del triángulo rectángulo pueden calcularse los extremos de las componentes vectoriales de las dos fuerzas.

Si se conoce la posición de C puede calcularse asimismo la fuerza  $R$  resultante de  $F_1$  y  $F_2$  (teorema del coseno).

##### 3. PASO : Diagrama de flujo

Aquí puede prescindirse del diagrama de flujo.

```

10 REM "PHYS/5"--REGLA DEL PARALELOGRAMO-
20 SCREEN 0:WIDTH 40:COLOR 1,15,15:KEY OFF:CLS
30 PRINT "PROGRAMA PARA ILUSTRAR LA REGLA DEL PARALELOGRAM
O.":PRINT
40 PRINT "          -----"
50 PRINT "          PROF. DR. W. VOSS, 1984
60 PRINT "          -----":PRINT:PRINT
70 PRINT:PRINT
80 INPUT " -Primera fuerza F1      = ";F1
90 PRINT:INPUT " -Segunda fuerza F2      = ";F2
100 PRINT:PRINT:PRINT "PONGA LOS ANGULOS DE APLICACION DE A
MBASFUERZAS EN GRADOS (0-360) :":PRINT:PRINT:PRINT
110 INPUT " -Angulo de aplicación de F1 = ";W1
120 PRINT:INPUT " -Angulo de aplicación de F2 = ";W2
130 SCREEN 2:COLOR 15,1,1:CLS
140 REM
150 REM          TRAZADO DE LOS EJES
160 REM
170 FOR X=0 TO 191 STEP 2
180 PSET (X,95),9
190:NEXT X
200 FOR Y=0 TO 191 STEP 2
210 PSET (95,Y),9
220 NEXT Y
230 REM
240 REM          OPTIMIZACION DE LA POSICION
250 REM
260 A=(F1/(F1+F2))*S1
270 B=(F2/(F2+F1))*S1
280 REM
290 REM          CALCULO DE LOS EXTREMOS
300 REM
310 AX=FIX(A*COS(-(W1*3.1416)/180))
320 AY=FIX(A*SIN(-(W1*3.1416)/180))
330 BX=FIX(B*COS(-(W2*3.1416)/180))
340 BY=FIX(B*SIN(-(W2*3.1416)/180))
350 REM
360 REM          REPRESENTACION FUERZAS
370 REM
380 LINE (95,95)-(95+AX,95+AY),13

```

```

390 LINE (95,95)-(95+BX,95+BY),12
400 RX=AX+BX
410 RY=AY+BY
420 LINE (95,95)-(95+RX,95+RY),15
430 LINE (95+RX,95+RY)-(95+AX,95+AY),6
440 LINE (95+RX,95+RY)-(95+BX,95+BY),6
450 OPEN "GRP:" FOR OUTPUT AS #1
460 PSET (208,0),1:PRINT #1, "-----"
470 PSET (208,10),1:PRINT #1, "COLOR"
480 PSET (208,20),1:PRINT #1, "-----"
490 PSET (200,60),1:PRINT #1, "F1=C-13"
500 PSET (200,90),1:PRINT #1, "F2=C-12"
510 PSET (200,120),1:PRINT #1, "R =C-15"
520 PSET (208,168),1:PRINT #1, "PULSA"
530 REM
540 REM          TEMPORIZADOR
550 REM
560 TIME=0
570 A=TIME
580 IF A<200 THEN 570
590 AS=INPUT$(1)
600 REM
610 REM          CALCULO DE LA RESULTANTE
620 REM
630 SCREEN 0:WIDTH 40:COLOR 1,11,11:CLS
640 W=W1-W2
650 W=180-W:W=(W/180)*(3.141592654#)
660 R=SQR(F1*F1+F2*F2-2*F1*F2*COS(W))
670 LOCATE 15,3:PRINT "RESULTADOS:"
680 LOCATE 0,8:PRINT "Primera fuerza = ";F1;" Angulo = ";
W1
690 LOCATE 0,10:PRINT "Segunda fuerza = ";F2;" Angulo = "
;W2
700 LOCATE 0,14:PRINT "RESULTANTE      = ";R
710 LOCATE 13,20:CLOSE:PRINT "OTRA VEZ (S/N)"
720 AS=INKEY$
730 IF AS=" " THEN 720
740 IF AS="S" OR AS="s" THEN 10
750 IF AS="N" OR AS="n" THEN 760 ELSE 720
760 COLOR 15,1,1:CLS:LOCATE 10,12:PRINT "FIN DEL PROGRAMA":
END

```

Voss	Capítulo	4 : Física	Página
	Punto	8 : Paralelogramo	172
Colegio			

5. PASO : Lista de las principales variables

- F1 = Primera fuerza
- F2 = Segunda fuerza
- R = Resultante
- X = Se usa en el cálculo de ésta coordenada
- Y = Se usa en el cálculo de ésta coordenada
- A = Magnitud auxiliar de optimización
- B = Magnitud auxiliar de optimización
- W = W1-W2
- W1 = Primer ángulo de aplicación
- W2 = Segundo ángulo de aplicación

6. PASO : Descripción del programa

- Línea 10-70 : Título
- Línea 80-120 : Introducción de los datos de partida
- Línea 130-220 : Trazado de los ejes coordenados
- Línea 230-270 : Optimización de las fuerzas para que no excedan los límites de la pantalla.

Voss	Capítulo	4 : Física	Página
	Punto	8 : Paralelogramo	173
Colegio			

Línea 280-340 : cálculo de las coordenadas de los extremos de las fuerzas.

Línea 350-390 : Trazado en pantalla de las dos fuerzas.

Línea 400-410 : Cálculo de las componentes de la resultante.

Línea 420-440 : Trazado de la resultante y líneas auxiliares.

Línea 450 : Apertura del canal de impresión de caracteres en modo gráfico de alta resolución.

Línea 460-520 : Impresión de etiquetas en pantalla

Línea 530-580 : Bucle temporizador (Reloj)

Línea 590 : Espera la entrada de un carácter

Línea 630 : Vuelta al modo texto y borrado de pantalla

Línea 640-660 : Cálculo de la fuerza resultante.

Línea 670-700 : Impresión de los resultados del problema

Línea 710 : Pregunta si se desea repetir el programa

Línea 720-750 : Se espera e identifica que caracter se ha tecleado

Línea 760 : Fin del programa

7. PASO : Resultados

En vista del esquema del 1. Paso, prescindiremos aquí de ofrecer resultados.

#### 4.9 La Ley de Ohm

##### 1. PASO : Presentación del problema

La Ley de Ohm se cumple en aquellos conductores eléctricos en los que la corriente crece proporcionalmente a la diferencia de potencial.

Esta Ley dice lo siguiente:

$$\frac{\text{dif. de pot.}}{\text{Intensidad}} = \text{constante} = R$$


La magnitud  $R$  se denomina resistencia eléctrica del conductor.

En el programa siguiente se calcula para cada par de valores (dif. de pot., Intensidad, resistencia), la tercera magnitud.

##### 2. PASO : Análisis del problema

El programa que nos ocupa no ofrece ninguna dificultad desde el punto de vista matemático.

##### 3. PASO : Diagrama de flujo



```

10 REM "PHYS/6" -LEY DE OHM-
20 COLOR 15,1,1:KEY OFF:SCREEN 0:WIDTH 40:CLS
30 PRINT "PROGRAMA PARA MOSTRAR LAS RELACIONES EN-TRE LA R
ESISTENCIA, LA DIFERENCIA DE PO-TENCIAL Y LA INTENSIDAD."
40 PRINT:PRINT:PRINT " PROF. DR. W. VOSS, 1984":PR
INT
50 PRINT:PRINT:PRINT " SELECCIONE LOS DATOS DE TRABAJO
:"
60 PRINT:PRINT:PRINT
70 PRINT "--DIF. DE POT. E INTENSIDAD.....(1)"
80 PRINT:PRINT "--DIF. DE POT. Y RESISTENCIA.....(2)"
90 PRINT
100 PRINT "---INTENSIDAD Y RESISTENCIA.....(3)"
105 PRINT:PRINT:PRINT
110 LOCATE 8,22:PRINT "Introduzca un número":AS=INKEY$
120 IF AS="1" OR AS="2" OR AS="3" THEN 130 ELSE 110
130 IF AS>"1" THEN 190 ELSE CLS
140 LOCATE 4,4:INPUT "DIF. DE POTENCIAL = ";U:LOCATE 35,4:
PRINT "v"
150 LOCATE 4,8:INPUT "INTENSIDAD = ";I:LOCATE 35,8:
PRINT "A"
160 R=U/I:R=FIX(R*100+.5)/100
170 LOCATE 4,14:PRINT "RESISTENCIA = ";R:LOCATE 35,1
4:PRINT CHR$(234)
180 GOTO 300
190 IF AS="3" THEN 250 ELSE CLS
200 LOCATE 4,4:INPUT "DIF. DE POTENCIAL = ";U:LOCATE 35,4
:PRINT "v"
210 LOCATE 4,8:INPUT "RESISTENCIA = ";R:LOCATE 35,8
:PRINT CHR$(234)
220 I=U/R:I=FIX(I*100+.5)/100
230 LOCATE 4,14:PRINT "INTENSIDAD = ";I:LOCATE 35,
14:PRINT "A"
240 GOTO 300
250 CLS:LOCATE 6,4:INPUT "RESISTENCIA = ";R:LOCATE 35
,4:PRINT CHR$(234)
260 LOCATE 6,8:INPUT "INTENSIDAD = ";I:LOCATE 35,8:P
RINT "A"
270 U=R*I:U=FIX(U*100+.5)/100
280 LOCATE 6,14:PRINT "DIF. DE POTENCIAL = ";U:LOCATE 35,14

```

```

:PRINT "v"
300 LOCATE 11,19:PRINT "MAS CALCULOS (S/N) ":AS=INKEY$
310 IF AS=" " THEN 300
320 IF AS="S" OR AS="s" THEN 10
330 IF AS="N" OR AS="n" THEN 340 ELSE 300
340 CLS:LOCATE 10,12:PRINT "FIN DE LOS CALCULOS":END

```

5. PASO : Lista de variables

AS = Variable de selección (menú)

I = Intensidad

R = Resistencia

U = Diferencia de potencial

6. PASO : Descripción del programa

Línea 10-100 : Título y menú

Línea 110 : Selección en el menú

Voss	Capítulo	4 : Física	Página
	Punto	9 : Ley de Ohm	178
Colegio			

- Línea 120 : Solo acepta las opciones 1, 2 y 3
- Línea 130 : Si A\$ es mayor que 1, proseguir en la línea 190, sino borrar pantalla
- Línea 140-150 : Introducciones de acuerdo con A\$=1
- Línea 160-170 : Cálculos e Impresión del resultado
- Línea 180 : Salto a la línea 300
- Línea 190-240 : Cálculos iguales a los anteriores, pero para A\$ = 2
- Línea 250-290 : Cálculo análogo, pero para A\$ = 3
- Línea 300-330 : Consulta si se desean más cálculos  
En caso afirmativo borrar la pantalla y volver a la línea 10
- Línea 340 : Final del programa

#### 7. PASO : Resultados

Si seleccionamos el punto 1 del menú, el programa nos pedirá la diferencia de potencial y la Intensidad. Si en respuesta introducimos, por ejemplo, el valor 220 para la diferencia de potencial y el valor 11 para la Intensidad, obtendremos el siguiente resultado:

RESISTENCIA : 20

MAS CALCULOS (S/N)

Voss	Capítulo	5 : Lenguas	Página
	Punto	1 : BASIC	179
Colegio			



#### Capítulo 5 : Lenguas

---

##### 5.1 Consideración previa y más BASIC

---

Los ordenadores pueden cumplir también útiles servicios para el aprendizaje y el repaso de idiomas extranjeros. No se trata aquí de presentar cursos completos de idiomas (por ejemplo, para aprender Inglés), porque esto superaría ampliamente el marco y el nivel de este libro - además, estos "programas completos" pueden adquirirse ya en los comercios a precios relativamente asequibles. Nos interesa mucho más el confeccionar programas sencillos, que puedan servirle de ayuda al escolar a la hora de aprender idiomas.

Se consideran aquí en primera línea aquellos programas que facilitan el aprendizaje de vocablos o que proporcionan traducciones, ahorrándonos así la consulta de diccionarios.

Voss	Capítulo	5 : Lenguas	Página
	Punto	1 : BASIC	180
Colegio			

No obstante, para este tipo de programas resultan necesarias algunas Instrucciones BASIC nuevas.

A menudo no será útil poder separar determinados segmentos del programa principal para poder saltar a los mismos más de una vez.

Estos programas parciales se denominan subrutinas. El salto del programa principal a la subrutina requiere la siguiente Instrucción:

Instrucción 21:

nn GOSUB mm

Esta Instrucción provoca un salto a la subrutina que comienza en la línea mm.

Voss	Capítulo	5 : Lenguas	Página
	Punto	1 : BASIC	181
Colegio			

Una vez recorrida la subrutina debe retornarse al programa principal y cerrar la subrutina mediante la Instrucción siguiente:

Instrucción 22:

nn RETURN

Esta Instrucción provoca un salto de vuelta a aquel punto del programa principal situado justo después de la correspondiente Instrucción GOSUB.

El funcionamiento de ambas Instrucciones queda patente en los siguientes programas, de modo que prescindiremos de cualquier ejemplo.

## 5.2 Verbos Irregulares Ingleses

### 1. PASO : Presentación del problema

En este primer ejemplo presentamos un programa que, tras introducir un verbo en castellano, ofrece su traducción al inglés y las correspondientes formas irregulares.

Así, por ejemplo tras introducir en el ordenador:

Ir,

éste nos contesta

go    went    gone    (Ir).

Para que el programa no resulte demasiado largo, nos limitaremos a 20 verbos irregulares. El lector descubrirá de inmediato cómo debe completar el programa para obtener una panorámica sobre todos los verbos irregulares Ingleses.

### 2. PASO : Análisis del problema

En este programa deben prepararse verbos castellanos, sus traducciones al inglés y las formas irregulares de estos verbos.

Cuando la cantidad de información es muy amplia, resulta aconsejable no leer las informaciones de partida tras cada comienzo del programa, sino efectuar la lectura interactiva de la información requerida a partir de un fichero grabado en una diskette convenientemente preparada.

Esta posibilidad, que seguramente utilizarán los programadores más adelantados - por lo menos si disponen de una unidad de diskettes - no vamos a tratarla aquí.

Así, tras cada inicio del programa haremos que el usuario indique el verbo acerca del cual quiere información. A renglón seguido, el programa buscará en el fichero de datos. Una vez hallado el verbo, podrá visualizarse el correspondiente registro.

Si el programa no hallase en su fichero el verbo introducido, sería debido a :


- o bien el verbo no existe aún en el archivo de datos (en principio solo hemos registrado 20 verbos)
- o bien el verbo introducido no es un verbo Irregular.



Estos dos casos no pueden separarse el uno del otro, debido a que la Información de entrada no es completa. En ambos debe visualizarse un mensaje.

Además, el programa debe ofrecerle al usuario la posibilidad de consultar otro verbo.

3. PASO : Diagrama de flujo



```

10 REM "LENG/1" --VERBOS IRREG. ING.-
20 COLOR 15,1,1:KEY OFF:SCREEN 0:WIDTH 40:CLS
30 PRINT " PROGRAMA PARA LA IMPRESION DE VERBOS I-"
40 PRINT " RREGULARES INGLESSES."
50 PRINT:PRINT:PRINT " PROF. DR. W. VOSS, 1984"
60 LOCATE 9,7:PRINT "-----"
70 LOCATE 0,12:PRINT " El número de verbos existentes en
este programa está limitado. Pueden efectuarse ampliaciones
alargando la lista de datos";
80 PRINT " (línea 500 y siguientes), y aumentando en la lí
nea 160 el valor de N."
90 LOCATE 5,21:PRINT "PULSE UNA TECLA PARA CONTINUAR"
100 AS=INPUT$(1)
110 DIM AS$(1),DS(15),E1$(15),E2$(15),E3$(15),US(15)
120 CLS
130 N=20
140 PRINT:PRINT:PRINT " INTRODUCIR EL VERBO EN ESPA";CHR$(1
65);"OL DEL CUAL"
150 PRINT:PRINT "SE DESEAN SABER LAS FORMAS INGLESAS (USE"
160 PRINT:PRINT "MAYUSCULAS POR FAVOR)"
170 LOCATE 11,15:INPUT "VERBO";US:CLS
180 FOR I=1 TO N
190 READ DS,E1$,E2$,E3$
200 IF DS=US THEN 230
210 NEXT I
220 LOCATE 0,8:PRINT " VERBO NO EXISTENTE EN EL FICHERO O
NO ":LOCATE 15,10:PRINT "IRREGULAR":GOTO 280
230 LOCATE 10,2:PRINT "VERBO :";US
240 PRINT:LOCATE 10,6:PRINT "FORMAS INGLESAS : "
250 PRINT:LOCATE 12,10:PRINT E1$
260 LOCATE 12,11:PRINT E2$
270 LOCATE 12,12:PRINT E3$
280 LOCATE 10,18:PRINT "OTRA VEZ (S/N)"
290 AS=INKEY$
300 IF AS="S" OR AS="s" THEN CLS:GOTO 140
310 IF AS="N" OR AS="n" THEN CLS:LOCATE 11,12:PRINT "FIN DE
L PROGRAMA" ELSE 290
500 DATA SER, BE, WAS, BEEN
510 DATA HABER, HAVE, HAD, HAD
520 DATA COMENZAR, BEGIN, BEGAN, BEGUN
    
```

530 DATA TRAER, BRING, BROUGHT, BROUGHT  
540 DATA VENIR, COME, CAME, COME  
550 DATA HACER, DO, DID, DONE  
560 DATA COMER, EAT, ATE, EATEN  
570 DATA COMPRAR, BUY, BOUGHT, BOUGHT  
580 DATA SENTIR, FEEL, FELT, FELT  
590 DATA ENCONTRAR, FIND, FOUND, FOUND  
600 DATA IR, GO, WENT, GONE  
610 DATA PERDER, LOOSE, LOST, LOST  
620 DATA HACER, MAKE, MADE, MADE  
630 DATA LEER, READ, READ, READ  
640 DATA CORRER, RUN, RAN, RUN  
650 DATA DECIR, SAY, SAID, SAID  
660 DATA VER, SEE, SAW, SEEN  
670 DATA SENTARSE, SIT, SAT, SAT  
680 DATA HABLAR, SPEAK, SPOKE, SPOKEN  
690 DATA COGER, TAKE, TOOK, TAKEN

#### 5. PASO : Lista de variables

A\$ : Variable de cadena para Introducciones  
D\$ : Verbo castellano  
E1\$ : Verbo Inglés, forma 1  
E2\$ : Verbo Inglés, forma 2  
E3\$ : Verbo Inglés, forma 3  
I : Variable índice  
N : Número de registros  
V\$ : Verbo elegido

#### 6. PASO : Descripción del programa

Línea 10-90 : Título, aclaraciones, indicaciones acerca de cómo debe actualizarse el archivo de datos

Línea 100 : Programa en "posición de espera". La realización del programa prosigue únicamente tras pulsar una tecla

Línea 110 : Dimensionado

Línea 120 : Borrado de la pantalla

Línea 130 : Indicación del número de verbos

Línea 140-170 : Demanda del verbo elegido

Línea 180-210 : Busca el fichero de datos correcto, en caso de hallarlo sigue en 230; en caso de no hallarlo sigue en 220

Línea 220 : Mensaje para indicar que el verbo no existe en el fichero de datos; seguir en la línea 280

Línea 230-270 : Impresión de las informaciones buscadas

Línea 280-310 : Consulta si se desea otra Impresión; en caso afirmativo, borrado de la pantalla y vuelta a la línea 140

Línea 310 : En caso negativo final del programa

Línea 500-690 : Datos

Voss	Capítulo	5 : Lenguas	Página
	Punto	1 : BASIC	188
Colegio			

## 7. PASO : Resultados

SI introducimos por ejemplo en respuesta a la demanda del ordenador en la línea 170:

IR,

El ordenador contesta:

VERBO : IR

FORMAS INGLESAS :

GO  
WENT  
GONE

OTRA VEZ (S/N) ?

Puede verse claramente cómo hay que modificar el programa para completar la lista de verbos irregulares : En la línea 500 y siguientes habrá que añadir más instrucciones DATA y en la línea 130 habrá que apuntar el nuevo número de verbos.

Voss	Capítulo	5 : Lenguas	Página
	Punto	3 : Vocablos	189
Colegio			

## 5.3 Vocablos franceses

=====

### 1. PASO : Presentación del problema

En este ejemplo vamos a mostrar como puede encontrar aplicación el ordenador para sustituir la libreta de vocablos. Este contesta a la palabra francesa que le introducimos con el equivalente castellano o viceversa (rige aquí la misma observación del punto 5.2 acerca de la utilización interactiva de una memoria de diskettes).

Para que el programa no crezca demasiado, lo limitaremos a diez vocablos. Es decir, que sirve sólo a modo de ejemplo. Para aprovecharlo realmente, el lector deberá tener la paciencia de introducir previamente en el ordenador todo su archivo castellano-francés.

En los principios de funcionamiento del programa no hay ningún cambio.

## 2. PASO : Análisis del problema

Con este problema podemos ser escuetos:


Cuando el usuario introduce una palabra castellana, el ordenador debe buscar entre su archivo de datos el correspondiente par de palabras castellano-francesas mediante la comparación con todos los vocablos castellanos (y sus correspondientes parejas en francés) de que dispone para, después, visualizarlas.

Para que ambos "sentidos" de la traducción puedan ser cubiertos por un solo programa, primero hay que decirle al ordenador si se desea la traducción del castellano al francés o del francés al castellano.

En caso de no poder traducir la palabra introducida por no estar incluida (aún) en la (muy) corta lista de vocablos, deberá visualizarse el correspondiente mensaje.

Además, debe ofrecérsele al usuario la posibilidad de obtener otra traducción en la ejecución del mismo programa, si así lo desea.

## 3. PASO : Diagrama de flujo



```

10 REM "LENG/2" #Castellano-Francés#
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT "DICCIONARIO CASTELLANO-FRANCES Y FRANCES"
40 PRINT "-CASTELLANO."
50 PRINT:PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
60 PRINT "          -----"
70 LOCATE 0,11:PRINT "Este programa contiene Únicamente 10
vo-cablos en su fichero."
80 PRINT:PRINT "Para ampliar el programa a";CHR$(164);"ada
más datos";
90 PRINT "a partir de la línea 500 y cambie el va-lor de N
en la línea 130."
95 LOCATE 6,19:PRINT "ESCRIBA SOLO CON MAYUSCULAS"
100 LOCATE 4,22:PRINT "#Pulse una tecla para continuar"
110 AS=INPUT$(1)
120 CLS
130 N=10
140 DIM AS(1),WS(15),DS(15),FS(15)
150 LOCATE 9,3:PRINT "QUE TRADUCCION DESEA ?"
160 PRINT "          -----"
170 LOCATE 5,9:PRINT "-CASTELLANO-FRANCES.....(1)"
180 LOCATE 5,14:PRINT "-FRANCES-CASTELLANO.....(2)"
190 LOCATE 8,21:PRINT "Pulse la opción elegida"
200 RESTORE
210 AS=INKEY$
220 IF AS="1" OR AS="2" THEN 230 ELSE 210
230 IF AS="2" THEN 350
240 CLS
250 LOCATE 2,3:INPUT "DEME LA PALABRA CASTELLANA ";WS
260 FOR I=1 TO N
270 READ DS,FS
280 IF DS=WS THEN 320
290 NEXT I
300 LOCATE 17,11:PRINT WS:LOCATE 4,13:PRINT"No consta en mi
banco de datos."
310 GOTO 450
320 LOCATE 10,9:PRINT "** CASTELLANO: ";DS
330 LOCATE 10,14:PRINT "** FRANCES : ";FS
340 GOTO 450
350 CLS

```

```

360 LOCATE 5,3:INPUT "DEME LA PALABRA FRANCESA ";WS
370 FOR I=1 TO N
380 READ DS,FS
390 IF FS=WS THEN 430
400 NEXT I
410 LOCATE 17,11:PRINT WS:LOCATE 4,13:PRINT "No consta en m
i banco de datos."
420 GOTO 450
430 LOCATE 10,9:PRINT "** FRANCES : ";FS
440 LOCATE 10,14:PRINT "** CASTELLANO: ";DS
450 LOCATE 12,21:PRINT "OTRA VEZ (S/N)"
460 AS=INKEY$
470 IF AS="S" OR AS="s" OR AS="N" OR AS="n" THEN 480 ELSE 4
60
480 IF AS="S" OR AS="s" THEN CLS:GOTO 150
490 CLS:LOCATE 12,11:PRINT "FIN DEL PROGRAMA":END
500 DATA COMPRAR,ACHETER
510 DATA HABER,AVOIR
520 DATA IR,ALLER
530 DATA UER,VOIR
540 DATA BEBER,BOIRE
550 DATA HACER,FAIRE
560 DATA LEER,LIRE
570 DATA COMER,MANGER
580 DATA TOMAR,PRENDRE
590 DATA FUMAR,FUMER

```

#### 5. PASO : Lista de variables

AS = Variable de cadena para Introducciones  
DS = Vocablos castellanos  
FS = Vocablos franceses  
I = Variable Índice  
N = Número de registros  
WS = Palabra elegida

6. PASO : Descripción del programa

Línea 10-100 : Título, aclaraciones etc.

Línea 110-120 : Espera y borrado de la pantalla

Línea 130-140 : Indicación del número de líneas de datos y dimensionado

Línea 150-190 : Borrado de la pantalla y visualización del menú

Línea 200-220 : Selección y restauración

Línea 230 : Si se introduce 2 (francés-castellano), seguir en la línea 350

Línea 250 : Introducción de un vocablo castellano

Línea 260-290 : Busca el correspondiente juego de datos

Línea 300-310 : Si la palabra buscada no se halla (todavía) en el archivo de datos y seguir en 450

Línea 320-330 : Impresión de los resultados

Línea 340 : Salto a la línea 450

Línea 350-440 : Procedimiento análogo para la introducción de la palabra francesa

Línea 450-470 : Finalización del programa con posibilidad de repetir

Línea 500-590 : Datos

7. PASO : Resultados

Si pulsamos el número 1 en el menú, decidiéndonos por la traducción del castellano al francés, el ordenador nos pedirá una palabra castellana.

Si en respuesta introducimos la palabra, el ordenador imprimirá:

CASTELLANO : TENER

FRANCES : AVOIR

OTRA VEZ (S/N) ?

#### 5.4 Test de vocablos ingleses

=====

##### 1. PASO : Presentación del problema

De forma similar al punto anterior, aquí deben ponerse a nuestra disposición traducciones a partir de un determinado fichero de vocablos. Pero, en contraposición al programa precedente, el usuario es requerido para contestar a una palabra castellana elegida al azar por el ordenador con la palabra Inglesa correcta.

Es decir, que presentamos un programa que funciona del mismo modo que el aprendizaje de vocablos con la columna de palabras en el idioma extranjero tapada de antemano.

Puede observarse rápidamente que este programa es también aplicable a otros Idiomas.

Con el fin de reducir el tiempo de programación nos limitaremos a ofrecer solo diez vocablos a modo de ejemplo. El ordenador selecciona en cada caso un subconjunto aleatorio del mismo, siendo informado el usuario acerca del número de aciertos por serie.


##### 2. PASO : Análisis del problema.

Partiendo de un fichero de datos dado, el programa debe realizar una selección al azar.

El usuario responde a cada vocablo visualizado. El número de respuestas correctas es comparado con el número de preguntas.

El análisis del problema no ofrece ninguna otra dificultad.

3. PASO : Diagrama de flujo



```

10 REM "LENG/3" -TEST DE VOCABLOS-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT "TEST DE VOCABLOS CASTELLANO/INGLES O VI-"
40 PRINT "CEVERSA."
50 PRINT:PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
60 PRINT "-----"
70 LOCATE 0,10:PRINT " Este programa contiene solo 10 voca-
blosen su fichero."
80 PRINT " Para ampliar el programa a";CHR$(164);"adir más
sentencias DATA a partir de la línea 500, y modificar en la
línea 120 el valor de N."
90 LOCATE 7,18:PRINT "ESCRIBA SOLO EN MAYUSCULAS"
100 LOCATE 6,22:PRINT "Pulse una tecla para comenzar"
110 AS=INPUT$(1)
120 N=10
130 DIM AS$(1),DS$(15),ES$(15),WS$(15),HS$(15)
140 CLS
150 LOCATE 15,3:PRINT "OPCIONES"
160 LOCATE 15,4:PRINT "-----"
170 LOCATE 5,8:PRINT "-CASTELLANO/INGLES.....(1)"
180 LOCATE 5,11:PRINT "-INGLES/CASTELLANO.....(2)"
190 LOCATE 13,20:PRINT "PULSE 1 ó 2"
200 AS=INKEY$
210 IF AS="1" OR AS="2" THEN 220 ELSE 200
220 CLS
230 LOCATE 1,11:INPUT "CUANTOS VOCABLOS QUIERE REPASAR ";A
240 CLS
250 RESTORE
260 K=0
270 I=1
280 R=FIX(RND(1)*N+1)
290 FOR J=1 TO R
300 READ DS,ES
310 NEXT J
320 IF AS="2" THEN HS=DS:DS=ES:ES=HS
330 CLS
340 LOCATE 7,5:PRINT DS;" EQUIVALE A:"
350 LOCATE 7,6
360 FOR X=1 TO LEN(DS)
370 PRINT "-";
    
```



```

380 NEXT X
390 PRINT "-----"
400 LOCATE 7,10:INPUT "RESPUESTA : ";WS
410 IF WS=ES THEN K=K+1:LOCATE 7,15:PRINT "MUY BIEN !":GOTO
440
420 CLS:LOCATE 8,5:PRINT "HA COMETIDO UN ERROR"
430 LOCATE 3,14:PRINT "LA RESPUESTA CORRECTA ES: ";ES
440 I=I+1
450 IF I<=A THEN RESTORE:GOTO 490
460 KA=(K/A)*100:KA=FIX(KA*100+.5)/100
470 LOCATE 5,21:PRINT "RESPUESTAS CORRECTAS: ";KA;" %"
480 END
490 TIME=0
495 T=TIME
498 IF T>125 THEN 280 ELSE 510
500 DATA CORRER,RUN,ESCRIBIR,WRITE
510 DATA IMPRIMIR,PRINT,IR,GO
520 DATA CORRER,RUN,ESCRIBIR,WRITE
530 DATA IMPRIMIR,PRINT,IR,GO
540 DATA SI,IF,SUBROUTINA,SUBROUTINE
550 DATA ORDENADOR,COMPUTER,CARACTER,CHARACTER
560 DATA PANTALLA,SCREEN,TECLA,KEY
570 DATA SI,IF,SUBROUTINA,SUBROUTINE
580 DATA ORDENADOR,COMPUTER,CARACTER,CHARACTER
590 DATA PANTALLA,SCREEN,TECLA,KEY

```

5. PASO : Lista de variables

A = Número de vocablos a comprobar  
A\$ = Cadena de respuesta  
D\$ = Vocablos castellanos  
E\$ = Vocablos Ingleses  
I = Variable índice  
J = Variable índice  
K = Número de respuestas correctas  
KA = Proporción de respuestas correctas  
N = Número de vocablos en el fichero  
R = Número aleatorio  
W\$ = Respuesta  
TIME = Reloj del sistema

6. PASO : Descripción del programa

Línea 10-100 : Título, aclaraciones y recordatorio de la eventual ampliación del programa  
Línea 110 : Espera una entrada desde el teclado  
Línea 120 : Indicación del número de vocablos incluidos en el fichero de datos, dimensionado  
Línea 130 : Dimensionado de las variables.  
Línea 140-220 : Borrado de la pantalla, Impresión del menú y demanda de la selección  
Línea 230 : Demanda del número de vocablos a repasar  
Línea 240-250 : Restauración y borrado de la pantalla  
Línea 260-270 : Contadores K e I a las posiciones iniciales  
Línea 280 : Generación de un número aleatorio  
Línea 290-310 : Lectura de datos  
Línea 320-330 : Eventual cambio de idioma

Voss	Capítulo	5 : Lenguas	Página
Colegio	Punto	4 : Test de vocablos	202

Línea 340-390 : Presentación de la pregunta

Línea 400 : Petición de una respuesta

Línea 410-430 : Evaluación de la respuesta. Si es correcta se imprime mensaje de felicitación, si no la es, se sigue en la línea 420 y se imprime la respuesta correcta.

Línea 440-450 : Vocablo siguiente

Línea 460-470 : Cálculo e impresión de la cuota de aciertos

Línea 480 : Fin del programa

Línea 490-498 : Subrutina temporizadora

Línea 500-590 : Datos

## 7. PASO : Resultados

No hace falta ofrecer aquí los resultados. Aquellos que son visualizados durante el transcurso del programa resultan directamente de la descripción del mismo

Voss	Capítulo	5 : Lenguas	Página
Colegio	Punto	4 : Escritura en clave	203

## 5.5 Escritura en clave

---

### 1. PASO : Presentación del problema

Después de los ejemplos anteriores, que eran típicamente "escolares", para variar un poco presentamos un programa de carácter más bien lúdico - la creación de una escritura en clave.

Este programa convierte tantas frases como queramos a través de un código, en frases Incomprensibles.

Se trata simplemente de convertir los caracteres utilizados normalmente para escribir, en otros caracteres siguiendo una regla que debe fijarse. El usuario podrá manipular la conversión de la forma que desee.

## 2. PASO : Análisis del problema

El análisis del problema no es excesivamente complicado: Se trata únicamente de enfrentar el juego de caracteres usual (A,B,C..., inclusive los símbolos especiales que queramos utilizar) al "juego de caracteres secreto". Ambas cosas pueden hacerse con variables de cadena adecuadas.

Después de esto, la frase introducida debe ser codificada de carácter a carácter.

Para hacer factible esto deberemos utilizar dos funciones, hasta ahora no mencionadas: La primera función sirve para determinar la longitud de una cadena, es decir, el número de caracteres de que consta (incluidos los espacios en blanco).

La segunda función sirve para "arrancar" caracteres sueltos de una cadena dada.

La función citada en primer lugar requiere el comando LEN (LEN = length = longitud).

Por ejemplo, si el usuario introduce estando en la modalidad directa:

```
PRINT LEN("COMPUTER")
```

el ordenador contestará:

8

Esta función nos ha servido para determinar que la cadena "computer" consta de ocho caracteres.

La segunda función se utiliza del modo siguiente:

Cadena parcial = MID\$(cadena de origen, número 1, número 2)

En la cadena parcial se almacenan N caracteres (número 2) de la cadena de origen contados hacia la izquierda del carácter número 1

Ejemplo :

```
10 A$ = "COMPUTER"
20 B$ = MID$(A$,8,4)
30 PRINT B$
```

Este programa imprime el contenido del campo B\$, compuesto por la serie de caracteres

UTER.


De la misma forma podemos extraer también caracteres sueltos, usando como número 1 = posición del carácter que se quiere extraer; y como número 2 = 1

```
20 B$ = MID$(A$,4,1)
```

Resultado : P

Le aconsejamos consultar en su manual de programación el uso de las funciones LEFT\$, RIGHT\$, STRING\$, y otras sobre la manipulación de cadenas alfanuméricas exclusivas del BASIC-MSX.

3. PASO : Diagrama de flujo



```

10 REM #LENG/4# -ESCRITURA EN CLAVE-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA CIFRAR CUALQUIER TEXTO."
40 PRINT:PRINT " PROF. DR. W. VOSS,1984"
50 PRINT " -----"
60 LOCATE 0,9:PRINT "Para modificar la clave cambiar la li
nea";
70 PRINT "180."
80 PRINT "Para mantener en secreto el código defini-r la c
adena CS";
90 PRINT " mediante una sentenciaINPUT, cuyo contenido no
debe";
100 PRINT " ser dado aconocer."
110 PRINT "La longitud del texto no podrá exceder los 255
caracteres."
120 LOCATE 7,21
130 PRINT "PULSE UNA TECLA, POR FAVOR"
140 AS=INPUT$(1)
150 CLS
160 DIM BS(30),CS(30)
170 BS="ABCDEFGHIJKLMNPOQRSTUVWXYZ ,."
180 CS="QWERTYUIOPASDFGHJKL ZX,C.VBNM"
190 LOCATE 4,4
200 INPUT "INTRODUZCA EL TEXTO A CODIFICAR
";AS:LOCATE 6,13

210 L=LEN(AS)
220 FOR I=1 TO L
230 FOR J=1 TO LEN(BS)
240 IF MIDS(AS,I,1)=MIDS(BS,J,1) THEN 250 ELSE 270
250 PRINT MIDS(CS,J,1);
260 GOTO 280
270 NEXT J
280 NEXT I
290 LOCATE 12,22:PRINT "OTRA VEZ (S/N)"
300 AS=INKEY$
310 IF AS="S" OR AS="s" OR AS="n" OR AS="N" THEN 320 ELSE 3
00
320 IF AS="S" OR AS="s" THEN CLS:GOTO 190
330 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
    
```

5. PASO : Lista de variables

A\$ = Texto a codificar  
B\$ = Juego de caracteres  
C\$ = Carácter cifrado  
I = Índice variable  
J = Índice variable  
L = Longitud del texto a codificar  
Z\$ = Cadena de respuesta (S/N)

6. PASO : Descripción del programa

Línea 10-110 : Impresión del título y aclaraciones

Línea 120-150 : Interrupción del programa y borrado de la pantalla

Línea 160 : Dimensionado

Línea 170 : Definición del juego de caracteres

Línea 180 : Definición clave

Línea 190-200 : Demanda del texto a codificar

Línea 210 : Determinación de la longitud del texto

Línea 220-280 : Codificación

220 : Comienzo del bucle I (para todos los caracteres del texto)

230 : Comienzo del bucle J (para el juego de caracteres)

240 : Verificación de que cada carácter del texto se corresponde con un carácter del juego;  
en caso negativo, seguir en 270  
en caso afirmativo, impresión del correspondiente carácter cifrado

Voss	Capítulo	5	: Lenguas	Página	
	Punto	5	: Escritura en clave	210	
Colegio					

- 260 : Salto hacia la línea 280, es decir, proceso del siguiente carácter del texto
- 270 : Comprobación del siguiente carácter del juego de caracteres
- 280 : Siguiendo carácter del texto

Línea 290-320 : Consulta si se desea una nueva codificación

Línea 330 : Final del programa


#### 7. PASO : Resultados

Los resultados de este programa dependen de la directriz seguida para codificar (ver línea 180 : C\$) y del texto a codificar (ver línea 200 : A\$).

Si introducimos p.e. en el programa el texto WERNER VOSS obtendremos:

,TKFTKBXGLL

Voss	Capítulo	5	: Biología/Ecología	Página	
	Punto	5	: Aspectos previos	211	
Colegio					



## Capítulo 6 : Biología/Ecología

---

### 6.1 Aspectos previos

---

La aplicación del ordenador para problemas biológico-ecológicos se refiere especialmente a la simulación de procesos de desarrollo y de crecimiento (naturalmente pueden confeccionarse también programas de consulta siguiendo el patrón de los del capítulo anterior). Estos desarrollos pueden visualizarse en la pantalla gráficamente o en forma de tabla.

## 6.2 Crecimiento exponencial

### 1. PASO : Presentación del problema

Una población inicial de  $X_0$  individuos tiene una tasa de crecimiento anual del  $R$  por ciento ( $R$  debe ser positivo). Si este crecimiento no choca con limitaciones externas - cosa muy improbable a largo plazo - se produce un llamado crecimiento exponencial.

El desarrollo de un proceso de crecimiento tan sencillo como éste, se simulará con un programa BASIC.

### 2. PASO : Análisis del problema

Para este programa hay que introducir la población inicial y la tasa de crecimiento anual.

Después de un año se obtiene la población  $X_1$ :

$$X_1 = X_0 + X_0 * R / 100 = X_0 * (1 + R / 100)$$

Después de dos años se obtiene una  $X_2$ :

$$X_2 = X_1 * (1 + R / 100) = X_0 * (1 + R / 100) * (1 + R / 100)$$

De lo que resulta:


$$X_2 = X_0 * (1 + R / 100)^2$$

Generalizando, después de  $T$  años se cumple que:

$$X_T = X_0 * (1 + R / 100)^T$$

Esta fórmula indica el algoritmo de cálculo para el correspondiente programa BASIC.

### 3. PASO : Diagrama de flujo



```

10 REM #ECOL/1# -CRECIMIENTO-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT "PROGRAMA PARA REPRESENTAR EN FORMA DE TABLA"
40 PRINT "BLA EL CRECIMIENTO EXPONENCIAL."
50 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
60 LOCATE 8,6:PRINT "-----"
70 LOCATE 2,12:INPUT "POBLACION INICIAL..... ";X0
80 LOCATE 2,16:INPUT "TASA DE CRECIMIENTO % ..... ";R
90 LOCATE 2,20:PRINT "NUMERO DE A";CHR$(165);"OS".....
... ":LOCATE 31,20:INPUT T
100 CLS
110 LOCATE 8,0:PRINT "A";CHR$(165);"OS";" POBLACION"
N"
120 LOCATE 8,1:PRINT "----";" -----"
130 I=1
140 XT=X0*(1+R/100)^I
150 XT=FIX(XT*100+.5)/100
160 PRINT
170 PRINT " ";I;" ";XT
180 I=I+1
190 IF I/10=FIX(I/10) THEN LOCATE 5,22:PRINT "PULSE UNA TECLA"
PARA CONTINUAR":AS=INPUT$(1):CLS
200 IF I<=T THEN 140
210 LOCATE 13,22:PRINT "DIRA VEZ (S/N)"
220 AS=INKEY$
230 IF AS="S" OR AS="s" THEN 20
240 IF AS="n" OR AS="N" THEN 250 ELSE 220
250 CLS:LOCATE 12,12:PRINT "FIN DEL PROGRAMA"

```

#### 5. PASO : Lista de variables

A\$ = Cadena de respuestas (si/no)  
I = Variable índice  
R = Tasa de crecimiento en %  
X0 = Población inicial  
XT = Población en el momento T

#### 6. PASO : Descripción del programa

Línea 10-60 : Título  
Línea 70-90 : Visualización de las informaciones INPUT que necesita el programa  
Línea 100-120 : Impresión del título de la tabla  
Línea 130 : Preselección del primer período  
Línea 140-150 : Cálculo de la población  
Línea 160-170 : Impresión  
Línea 180 : Paso al siguiente período  
Línea 190 : Después de cada diez períodos se interrumpe el programa (es decir, cuando I es divisible entre 10); la continuación sólo es posible si se pulsa una tecla.  
Línea 200 : Mientras I sea menor o igual a T, el programa debe continuar (línea 140)  
Línea 210-250 : Final del programa, a menos que se pida un proceso más (entonces vuelta a la línea 20 después de borrar la pantalla)



## 7. PASO : Resultados

Si por ejemplo introducimos el valor 500 para la población inicial y una tasa de crecimiento del 4(%), además de un período de valor 8 (p.e. años), el programa genera la siguiente impresión en pantalla:

AÑOS	POBLACION
---	-----
1	520
2	540.8
3	562.43
4	584.93
5	608.33
6	632.66
7	657.97
8	684.28

OTRA VEZ (S/N) ?

## 6.3 Crecimiento limitado

---

### 1. PASO : Presentación del problema

El crecimiento exponencial que ya presentamos en el punto anterior, no se da en la práctica, y si se da será por espacios de tiempo muy cortos. Deben considerarse ante todo efectos Inhibidores, que serán más acentuados cuanto mayores sean los valores de las variables en juego.

Pensemos por ejemplo que una población con crecimiento exponencial choca muy pronto con limitaciones de nutrición, que dificultan más y más su posterior desarrollo.

Este tipo de evolución "amortiguada" va a ser simulada en el próximo programa BASIC.

2. PASO : Análisis del problema

El análisis del problema lleva consigo una forma de resolución similar a la del punto anterior:

También aquí se requieren valores de partida:

$X_0$  = Población de partida

$R_0$  = Tasa de crecimiento anual (en %) al comienzo del proceso de desarrollo

En contraposición al ejemplo precedente, no partimos de la base de que la tasa de crecimiento es constante, sino que provocamos su reducción a medida que aumenta la población.

Como la población aumenta con el paso del tiempo, mediante un esquema mental muy sencillo podemos deducir que la tasa de crecimiento se reduce progresivamente. Por ejemplo, podemos dividir entre dos la tasa cuando el tiempo se ha duplicado, dividirla entre cuatro cuando el tiempo se ha cuatriplicado, etc.


Claro que podemos establecer otros modelos matemáticos para simular este proceso de frenado, aunque esto no es de vital importancia dentro de este contexto.

El algoritmo que acabamos de describir puede expresarse matemáticamente de la forma siguiente:

$$X_1 = X_0 * (1 + R/I)$$

Es decir, en el período  $X_1$  la población se obtiene a partir de la del período precedente, como en el ejemplo anterior; en cambio, la tasa de crecimiento se divide entre el índice variable  $I$ , de manera que aquella se reduce a medida que aumenta  $I$ , es decir, a medida que pasa el tiempo.

3. PASO : Diagrama de flujo



```

10 REM #ECOL/2# -CRECIMIENTO LIMITADO-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA SIMULAR EN UNA TABLA UN "
40 PRINT " PROCESO DE CRECIMIENTO FRENADO."
50 PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
60 LOCATE 8,6:PRINT "-----"
70 LOCATE 2,12:INPUT "POBLACION INICIAL.....: ";X0
80 LOCATE 2,16:INPUT "TASA DE CRECIMIENTO % ....: ";R
90 LOCATE 2,20:PRINT "NUMERO DE A";CHR$(165);"OS.....
...:LOCATE 31,20:INPUT T
100 CLS
110 PRINT "A";CHR$(165);"OS";"          POBLACION";"          CRECI
MIENTO (%)"
120 LOCATE 0,1:PRINT "----";"          -----";"          ----
-----"
130 I=1
140 X1=X0+X0*(R/I)/100:X1=FIX(X1*100+.5)/100
150 Z=(X1-X0)/X0:Z=Z*100:Z=FIX(Z*100+.5)/100
160 PRINT
170 PRINT ;I;"          ";X1;"          ";Z
180 X0=X1:I=I+1
190 IF I/10=FIX(I/10) THEN LOCATE 5,22:PRINT "PULSE UNA TEC
LA PARA CONTINUAR":AS=INPUT$(1):CLS
200 IF I<=T THEN 140
210 LOCATE 13,22:PRINT "OTRA VEZ (S/N)"
220 AS=INKEY$
230 IF AS="S" OR AS="s" THEN 20
240 IF AS="n" OR AS="N" THEN 250 ELSE 220
250 CLS:LOCATE 12,12:PRINT "FIN DEL PROGRAMA"
    
```

5. PASO : Lista de variables

A\$ = Cadena de respuesta (sí/no)  
 I = Variable índice  
 R = Tasa de crecimiento al comienzo del proceso  
 T = Duración del proceso  
 X0 = Población Inicial  
 X1 = Población después de un periodo  
 Z = Modificación de la población

6. PASO : Descripción del programa

Línea 10-60 : Título y aclaraciones  
 Línea 70-100 : Demanda de los datos de partida y borrado de la pantalla  
 Línea 110-120 : Impresión del título de la tabla  
 Línea 130 : Fijación del contador de períodos  
 Línea 140-170 : Cálculo de la población, del crecimiento e impresión de los valores tras redondear a dos posiciones decimales  
 Línea 180 : X1 toma el valor de X0  
 Línea 190 : Se controla que el número de datos en pantalla no supere el número de 10  
 Línea 200 : Se evalúa si deben continuarse los cálculos (mientras l'siga siendo más pequeño que T después de sumarle 1)  
 Línea 210-250 : Consulta si todavía se desean más cálculos, (en caso afirmativo vuelta a 20) y fin del programa.

7. PASO : Resultados

Si introducimos como población inicial el valor 500, y una tasa de crecimiento en el primer año del 4%, además de indicar que el proceso debe durar 8 años, obtendremos:

AÑOS	POBLACION	CRECIMIENTO
1	520	4
2	530.4	2
3	537.47	1.33
4	542.84	1
5	547.18	.8
6	550.83	.67
7	553.98	.57
8	556.75	.5

OTRA VEZ (S/N) ?

6.4 Contaminación ambiental

1. PASO : Presentación del problema

Las poblaciones en crecimiento, por regla general frenan ellas mismas su desarrollo, debido a que deterioran sus condiciones de vida en grado creciente.

Un ejemplo típico de ello es la contaminación ambiental creciente a lo largo de las diversas fases de industrialización.

Es decir, aquí también se produce un "frenado del crecimiento" - similar al del punto anterior - que modifica el crecimiento exponencial que conocemos del punto 6.2.

En contraposición al punto anterior, este "frenado" no opera de acuerdo con el tamaño de la población propiamente dicho, sino de forma indirecta en dependencia de una variable que, a su vez, es proporcional a la población.

2. PASO : Análisis del problema

En el primer período, la población resultante del período 0 ya no se calcula

$$X_1 = X_0 + X_0 * R/100$$

síno:

$$X_1 = X_0 + X_0 * (R/100 - P*B_0)$$

B<sub>0</sub> es el factor contaminación ambiental en el momento 0, que frena el crecimiento de la población por medio de un parámetro P adecuado (p.e. dentro del ámbito de P = 0.01).

Lógicamente, necesitamos también una relación funcional que pueda expresar B<sub>1</sub> (contaminación ambiental en el período 1) en dependencia de B<sub>0</sub> y de X<sub>0</sub>.

Un posible ejemplo de relación sería:

$$B_1 = B_0 + A * X_0$$


Partiendo de un factor B<sub>0</sub> = 0, la contaminación aumenta a medida que crece la población, de acuerdo con un parámetro A, que debe dimensionarse adecuadamente (p.e. A = 0.015).

De este modo se cumple el siguiente algoritmo:

(1)  $B_1 = B_0 + A*X_0$

(2)  $X_1 = X_0 + X_0 * (R/100 - P*B_1)$

3. PASO : Diagrama de flujo



```

10 REM #ECOL/3# -CONT. AMBIENTAL-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA SIMULAR EN UNA TABLA UN "
40 PRINT " PROCESO DE CRECIMIENTO FRENADO POR SU "
50 PRINT " PROPIA DINAMICA."
60 PRINT:PRINT:PRINT " PROF. DR. W. VOSS, 1984"
70 LOCATE 8,8:PRINT "-----"
80 LOCATE 2,12:INPUT "POBLACION INICIAL.....: ";X0
90 LOCATE 2,15:INPUT "TASA DE CRECIMIENTO % ....: ";R
100 LOCATE 2,18:PRINT "NUMERO DE A";CHR$(165);"OS.....
... ":LOCATE 31,18:INPUT T
110 LOCATE 2,21:INPUT"FACTOR DE CONT. AMBIENTAL.: ";P
120 CLS
130 PRINT "A";CHR$(165);"OS";" POBLACION";" ";CHR
$(216);" (%);" CONT."
140 LOCATE 0,1:PRINT "----";" -----";" ----"
;"
;"
150 I=1:A=.015:B0=0
160 B1=B0+A*X0:X1=X0+X0*(R/100-P*B0)
170 Z=(X1-X0)/X0:Z=Z*100:X1=FIX(X1*100+.5)/100:B1=FIX(B1*10
0+.5)/100
180 Z=FIX(Z*100+.5)/100
190 IF X1<=0 THEN LOCATE 10,20:PRINT "POBLACION ELIMINADA":
GOTO 250
200 PRINT
210 PRINT I;" ";X1;" ";Z;" ";B1
220 I=I+1
230 IF I/10=FIX(I/10) THEN LOCATE 5,22:PRINT "PULSE UNA TEC
LA PARA CONTINUAR":AS=INPUT$(1):CLS
240 IF I<=T THEN B0=B1:X0=X1:GOTO 160
250 LOCATE 13,22:PRINT "OTRA VEZ (S/N)"
260 AS=INKEY$
270 IF AS="S" OR AS="s" THEN 20
280 IF AS="n" OR AS="N" THEN 290 ELSE 260
290 CLS:LOCATE 12,12:PRINT "FIN DEL PROGRAMA"

```

### 5. PASO : Lista de variables

A = Factor contaminación ambiental dependiente de la población  
A\$ = Variable de cadena (sí,no)  
B0 = Contaminación ambiental de partida  
B1 = Contaminación ambiental en el siguiente período  
I = Índice variable  
P = Factor contaminación ambiental  
R = Tasa de crecimiento anual  
T = Tiempo  
X0 = Población de partida  
X1 = Población en el siguiente período  
Z = Crecimiento de la población

### 6. PASO : Descripción del programa

Línea 10-70 : Título y aclaraciones  
Línea 80-110 : Demanda de las Informaciones INPUT  
Línea 120-140 : Borrado de la pantalla e Impresión de los encabezamientos de la tabla  
Línea 150 : Definición de otros parámetros  
Línea 160 : Cálculos para el siguiente período  
Línea 170-180 : Más cálculos

Línea 190 : Si la población calculada es menor o igual a 0, imprimir mensaje y saltar a la línea 250

Línea 200-210 : Impresión de los resultados para los periodos

Línea 220 : Incremento del índice variable I

Línea 230 : Comprueba que no se impriman excesivos datos a la vez en pantalla.

Línea 240 : Si I es menor que T, X0 toma el valor de X1 y B0 el de B1, y el programa vuelve a la línea 160

Línea 250-290 : Consulta si se desea calcular más (en caso afirmativo volver a la línea 20) y final del programa

### 7. PASO : Resultados

Si a la población de partida le damos p.e. el valor 1000, al horizonte de tiempo el valor 8 años, a la tasa de crecimiento anual el 5% y como factor de contaminación ambiental tomamos el valor 0.01, obtendremos:

AÑOS	POBLACION	%	CONT.
1	1050	5	15
2	945	-10	30.75
3	701.66	-25.75	44.93
4	421.49	-39.93	55.45
5	208.85	-50.45	61.77
6	90.29	-56.77	64.9
7	36.21	-59.9	66.25
8	14.03	-61.25	66.79

OTRA VEZ (S/N) ?



## Capítulo 7 : Geografía/Historia

---

### 7.1 Consideración previa

---

En este capítulo pueden confeccionarse programas similares a los del capítulo "lenguas", es decir, ante todo programas de test y de consulta. También aquí nos limitaremos a utilizar ficheros de datos de muestra, explicando el montaje y funcionamiento del programa.

Para su utilización práctica, habrá que crear primero ficheros de datos completos y después introducirlos. Con este fin se completarán los DATAS, modificando el valor de N según el caso.



## 7.2 Fechas históricas

---

### 1. PASO : Presentación del problema

Presentamos un programa que puede utilizarse para aprender fechas históricas y para repasar estas fechas.

Con este fin, el ordenador le presenta al usuario acontecimientos históricos, requiriéndole para que introduzca la fecha correcta de estos acontecimientos históricos. El usuario puede decidir cuántas preguntas le debe formular el ordenador por cada serie de tests.


Si la respuesta del usuario es correcta, el ordenador le alaba; si, en cambio, la respuesta es incorrecta, el ordenador consulta si quiere probar otra vez. En caso de negativa por parte del usuario, el ordenador se ocupa de imprimir la respuesta correcta en pantalla.

### 2. PASO : Análisis del problema

La presentación del problema ya ha dado a entender claramente cómo debe procederse en este programa:

Una parte del programa está destinada a introducir información, también se debe preguntar al usuario cuántos tests desea hacer y el programa debe analizar cada respuesta. Por lo demás no se presentan dificultades especiales.

3. PASO : Diagrama de flujo



```

10 REM #GEO-H1# -FECHAS HISTORICAS-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT "PROGRAMA PARA REPASAR FECHAS HISTORICAS "
40 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
50 LOCATE 8,4:PRINT "-----"
60 LOCATE 0,6:PRINT "Este programa utiliza como fichero 30
da"
70 PRINT "tos históricos seleccionados (especifica"
80 PRINT "dos como sentencias DATA a partir de la"
90 PRINT "línea 500 del programa).
100 PRINT:PRINT "En caso de que se quisieran utilizar más"
110 PRINT "datos, deberán a";CHR$(164);"adirse más DATAS y
cam-"
120 PRINT "biar el valor de la variable N en la lí-"
130 PRINT "nea 160."
140 LOCATE 12,22:PRINT "PULSE UNA TECLA"
150 AS=INPUT$(1):CLS
160 N=30
170 DIM J$(5),S$(50),A$(1),K$(5)
180 LOCATE 2,4:PRINT "CUANTAS PREGUNTAS SE DESEAN REALIZAR
?"
190 LOCATE 4,12:INPUT "Por favor indicar un número : ";Z
200 LOCATE 10,19:PRINT "ESCRIBA EN MAYUSCULAS":GOSUB 440
210 FOR I=1 TO Z
220 CLS
230 R=FIX(RND(TIME)*N+1)
240 RESTORE
250 FOR J=1 TO R
260 READ J$,S$
270 NEXT J
280 LOCATE 14,1:PRINT "CUANDO FUE : "
290 LOCATE 0,5:PRINT S$
300 LOCATE 7,10:PRINT "DIGA EL A";CHR$(165);"O : ";LOCAT
E 25,10:INPUT K$
310 IF K$=J$ THEN LOCATE 11,17:PRINT "P E R F E C T O !":GO
SUB 440:GOTO 380
320 LOCATE 2,15:PRINT "ES UNA LASTIMA, PERO NO ES CORRECTO
!"
330 LOCATE 10,22:PRINT "OTRO INTENTIO (S/N)"
340 AS=INPUT$(1)
    
```

Voss	Capítulo	7	: Geografía/Historia	Página
	Punto	2	: Fechas históricas	236
Colegio				

```

350 IF AS="S" OR AS="s" THEN CLS:GOTO 280
360 LOCATE 2,18:PRINT "LA FECHA HISTORICA CORRECTA ES : ";J
$
370 GOSUB 440
380 NEXT I
390 LOCATE 9,22:PRINT." UN TEST MAS (S/N)"
400 AS=INKEY$
410 IF AS="S" OR AS="s" THEN CLS:GOTO 180
420 IF AS="n" OR AS="N" THEN 430 ELSE 400
430 CLS:LOCATE 12,12:PRINT "FIN DEL PROGRAMA":END
440 REM SUBROUTINA TEMPORIZADORA
450 TIME=0
460 A=TIME
470 IF A>150 THEN RETURN ELSE 460
500 DATA 1939,EL PRINCIPIO DE LA SEGUNDA GUERRA MUNDIAL
510 DATA 800,LA CORONACION DEL EMPERADOR CARLOMAGNO
520 DATA 1969,LA LLEGADA DEL HOMBRE A LA LUNA
530 DATA 1808,LA GUERRA DE LA INDEPENDENCIA ESPANOLA
540 DATA 1492,EL DESCUBRIMIENTO DE AMERICA
550 DATA 1936,EL COMIENZO DE LA GUERRA CIVIL ESPANOLA
560 DATA 1871,EL ESTABLECIMIENTO DEL IMPERIO ALEMAN
570 DATA 1812,LA LLEGADA DE NAPOLEON ANTE MOSCU
580 DATA 1975,LA MUERTE DEL GENERALISIMO FRANCO
590 DATA 1914,EL INICIO DE LA PRIMERA GUERRA MUNDIAL
600 DATA 1469,EL MATRIMONIO DE ISABEL Y FERNANDO
610 DATA 1494,EL TRATADO DE TORDESILLAS
620 DATA 1513,EL DESCUBRIMIENTO DEL OCEANO PACIFICO
630 DATA 1519,LA DISTINCION DE EMPERADOR A CARLOS I
640 DATA 1519,EL DESEMBARCO DE CORTES EN MEJICO
650 DATA 1525,LA BATALLA DE PAVIA
660 DATA 1598,LA MUERTE DE FELIPE II
670 DATA 1558,LA MUERTE DE CARLOS I
680 DATA 1588,EL FRACASO DE LA ARMADA INVENCIBLE
690 DATA 1618,EL COMIENZO DE LA GUERRA DE LOS 30 AÑOS
700 DATA 1648,LA PAZ DE WESTFALIA
710 DATA 1704,LA OCUPACION INGLESA DE GIBRALIAR
720 DATA 1789,LA REVOLUCION FRANCESA
730 DATA 1939,EL FIN DE LA GUERRA CIVIL
740 DATA 1776,LA INDEPENDENCIA DE LOS ESTADOS UNIDOS
750 DATA 1945,EXPLOSION DE LA PRIMERA BOMBA ATOMICA

```

Voss	Capítulo	7	: Geografía/Historia	Página
	Punto	2	: Fechas históricas	237
Colegio				

Línea 10-150 : Título y aclaraciones

Línea 160-170 : Número de líneas data y dimensionado

Línea 180-200 : Demanda el número de test

Línea 210-380 : Tests

--Línea 210-300 : Elección al azar de un acontecimiento histórico u demanda de la fecha como respuesta del usuario

--Línea 310-320 : Comprueba si la respuesta es correcta

--Línea 330-350 : Consulta si se desea otro intento y según la respuesta procede en con secuencia

--Línea 360 : Si la respuesta ha sido ne gativa se imprime la respuesta correcta.

--Línea 370-380 : Llamada a subrutina tempori zadora y vuelta a la línea 210.

Línea 390-430 : Consulta si se desea otra rueda de test; en caso afirmativo, borrado de la pantalla y vuelta a la línea 180. En caso negativo, final del programa.

Línea 440-470 : Subrutina temporizadora

Línea 500-790 : Datos

## 7. PASO : Resultados

No es necesario ofrecer aquí los resultados. Nos parece más razonable que el lector pruebe el programa considerando los contenidos y los DATAS modificándolo después de acuerdo con sus propios deseos.

### 7.3 Las capitales de las Comunidades Autónomas

#### 1. PASO : Presentación del problema

El programa que presentamos en este punto es también un llamado programa de entreno. El usuario debe responder a las preguntas que le formula el ordenador. Si el ordenador ofrece nombres de comunidades autónomas elegidos al azar, el usuario deberá introducir las capitales de estas comunidades; si en cambio el ordenador visualiza capitales, el usuario deberá citar las correspondientes comunidades.


Llegados a este punto no hay problemas nuevos, de modo que podremos ser breves. Este programa tiene carácter ejemplar :

El usuario puede registrar fácilmente nuevos datos en los DATAS, obteniendo así un programa de consulta de aplicación general, que puede utilizarse para las más diversas asignaturas escolares.

#### 2. PASO : Análisis del problema

Ya hemos mencionado que aquí no se presenta ningún problema adicional, de modo que podemos pasar rápidamente al siguiente paso. Todo lo referente a este programa ya ha sido suficientemente discutido en ejemplos precedentes.

#### 3. PASO : Diagrama de flujo



Voss	Capítulo	7	: Geografía/Historia	Página
	Punto	3	: Comunidades	240
Colegio			Autonómicas	

```

10 REM #GEO-H2# -COMUNIDADES AUTONOMAS-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA REPASAR GEOGRAFIA "
40 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
50 LOCATE 8,4:PRINT "-----"
60 LOCATE 0,6:PRINT " Este programa, tras indicar una comu
ni-"
70 PRINT " dad autónoma, nos pregunta su capital y"
80 PRINT " al contrario."
90 PRINT:PRINT " Este programa puede utilizarse para o-"
100 PRINT " tras fines si cambiamos los datos a par"
110 PRINT " tir de la sentencia 500 y N en la línea"
120 PRINT " 160."
130 LOCATE 12,22:PRINT "PULSE UNA TECLA"
140 AS=INPUT$(1):CLS
150 N=11
160 DIM L$(20),S$(20),A$(1),C$(10),X$(20)
170 LOCATE 2,2:PRINT "DEBEN INDICARSE AUTONOMIAS O CAPITALE
S"
180 LOCATE 4,9:INPUT "Por favor indique A ò C : ";A$
190 IF A$="C" OR A$="c" THEN C$="AUTONOMIA" ELSE C$="CAPITA
L"
200 LOCATE 4,15:INPUT "CUANTAS CONSULTAS : ";Z
210 LOCATE 10,22:PRINT "ESCRIBA EN MAYUSCULAS":GOSUB 460
220 FOR I=1 TO Z
230 CLS
240 R=FIX(RND(TIME)*N+1)
250 RESTORE
260 FOR J=1 TO R
270 READ L$,S$
280 NEXT J
290 IF A$="C" THEN SWAP L$,S$
300 LOCATE 15,3:PRINT L$:LOCATE 15,4:FOR X=1 TO LEN(L$):PRI
NT "-";:NEXT X
310 LOCATE 9,8:PRINT C$
320 LOCATE 20,8:INPUT X$
330 IF X$=S$ THEN LOCATE 11,17:PRINT "C O R R E C T O !":GO
SUB 460:GOTO 400
340 LOCATE 2,15:PRINT "ES UNA LASTIMA, PERO NO ES CORRECTO
!"

```

Voss	Capítulo	7	: Geografía/Historia	Página
	Punto	3	: Comunidades	241
Colegio			Autonómicas	

```

350 LOCATE 10,22:PRINT "OTRO INTENTIO (S/N)"
360 AS=INPUT$(1)
370 IF A$="S" OR A$="s" THEN CLS:GOTO 300
380 LOCATE 2,18:PRINT "LA RESPUESTA CORRECTA ES : ";S$
390 GOSUB 460
400 NEXT I
410 LOCATE 9,22:PRINT " UN TEST MAS (S/N)"
420 AS=INKEY$
430 IF A$="S" OR A$="s" THEN CLS:GOTO 170
440 IF A$="n" OR A$="N" THEN 450 ELSE 420
450 CLS:LOCATE 12,12:PRINT "FIN DEL PROGRAMA":END
460 REM SUBROUTINA TEMPORIZADORA
470 TIME=0
480 A=TIME
490 IF A>150 THEN RETURN ELSE 480
500 DATA CATALUÑA, BARCELONA, MADRID, MADRID
510 DATA RIOJA, LOGROÑO, BALEARES, PALMA DE MALLORCA
520 DATA PAIS VALENCIANO, VALENCIA, EUZCADI, GASTEIZ
530 DATA CANTABRIA, SANTANDER, ANDALUCIA, SEVILLA
540 DATA GALICIA, LA CORUÑA, ASTURIAS, GIJÓN
550 DATA EXTREMADURA, CACERES, CANARIAS, LAS PALMAS
560 DATA MURCIA, MURCIA, CASTILLA-LA MANCHA, ??????
570 DATA CASTILLA-LEÓN, VALLADOLID, ARAGON, ZARAGOZA
580 DATA ASTURIAS, OVIEDO, NAVARRA, PAMPLONAK

```

Voss	Capítulo	7	: Geografía/Historia	Página
	Punto	3	: Comunidades	242
Colegio			Autonómicas	

5. PASO : Lista de variables

A\$ = Variable de cadena  
 C\$ = Variable auxiliar que según se elija toma el valor "capital" o "autonomía"  
 H\$ = Variable auxiliar para el cambio  
 I = Índice variable  
 J = Índice variable  
 L\$ = Autonomías  
 N = Número de registros  
 S\$ = Capitales  
 X\$ = Introducción hecha por el usuario

6. PASO : Descripción del programa

Línea 10-130 : Título y aclaraciones  
 Línea 140 : Espera  
 Línea 150-160 : Dimensionado  
 Línea 170-180 : Consulta si deben visualizarse autonomías o capitales  
 Línea 190 : Si se eligen capitales, la variable auxiliar C\$ pasa a ser "autonomía"  
 Línea 200 : Demanda del número de consultas

Voss	Capítulo	7	: Geografía/Historia	Página
	Punto	3	: Comunidades	243
Colegio			Autonómicas	

Línea 210 : Se ruega se escriba en mayúsculas y se salta a la subrutina temporizadora

Línea 220-400 : Formulación del número de consultas elegido

240 : Elección al azar de un par de valores  
 250 : Restauración del fichero de datos  
 290 : Si se ha seleccionado C en el menú (ver 180-190), se intercambian S\$ y L\$  
 300 : Impresión de la pregunta  
 320 : Introducción de la respuesta por parte del usuario  
 330 : En caso correcto, seguir en 400  
 380 : Impresión de la respuesta correcta

Línea 410-490 : Final del programa y subrutina temporizadora

Línea 500-580 : Datos

## 7. PASO : Resultados

Tampoco en este programa resulta necesario ofrecer los resultados. El análisis del programa, o aún mejor, la ejecución del mismo muestra claramente lo que ocurre en cada momento.

## 7.4 El desarrollo de la población en diferentes países

---

### 1. PASO : Presentación del problema

Presentamos aquí un programa que pronostica el desarrollo de la población de diferentes países del mundo hasta el año 2000.

Los países o grupos de países seleccionados para el estudio son:

Mundo (comprendiendo a todos los países)

Países industrializados

Países en vías de desarrollo

RP China

India

Japón

USA

URSS

República Federal de Alemania

RDA.

2. PASO : Análisis del problema

Los pronósticos de población planteados requieren el decidirse por un determinado método de pronosticación.


Existe un gran número de métodos distintos, que aquí no pueden ser discutidos uno a uno.

Para este programa hemos extraído los valores obtenidos en pronósticos modelo como los publicados en el Informe GLOBAL 2000 (Washington 1980).

A partir de estos datos calculamos además las tasas de crecimiento medias anuales, para poder ofrecer a través del programa una información adicional.

Desde el punto de vista de la programación no aparecen dificultades importantes, si exceptuamos que, debido al volumen de resultados, en diversos puntos del programa es necesario efectuar interrupciones.

3. PASO : Diagrama de flujo





```

10 REM #GEO-M3# -TABLA DE POBLACION-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT:PRINT "PROGRAMA PARA REPRESENTAR EN UNA TABLA,":P
RINT:PRINT
40 PRINT "EL DESARROLLO DE LA POBLACION DE DIFE-":PRINT:PR
INT
50 PRINT "RENTES PAISES."
60 PRINT:PRINT:PRINT:PRINT "          PROF. DR. W. VOSS, 198
4"
70 LOCATE 8,12:PRINT "-----"
80 GOSUB 720:CLS
90 DIM B(10,6),J(6),WR(10),L$(20)
100 DIM L1$(5),L2$(5),L3$(5),L4$(5),L5$(5)
110 DIM L6$(5),L7$(5),L8$(5),L9$(5),L0$(5)
120 FOR I=1 TO 6
130 READ X
140 J(I)=X
150 NEXT I
160 FOR L=1 TO 10
170 FOR I=1 TO 6
180 READ X
190 B(L,I)=X
200 NEXT I
210 NEXT L
220 L1$="MUNDO":L2$="IND."
230 L3$="DES.":L4$="CHINA"
240 L5$="INDIA":L6$="JAPON"
250 L7$="USA":L8$="URSS"
260 L9$="RFA":L0$="RDA"
270 PRINT "      AND";"      ";L1$;"      ";L2$;"      ";L3$
280 PRINT:GOSUB 810
290 LOCATE 0,4
300 FOR I=1 TO 6
310 PRINT:PRINT "      ";J(I);"      ";B(1,I);"      ";B(2,I);"      "
;B(3,I)
320 NEXT I
330 PRINT
340 GOSUB 810
350 FOR L=1 TO 3
360 X=B(L,1):Y=B(L,6)
370 GOSUB 760
380 NEXT L
390 PRINT "      TASA %";"      ";WR(1);"      ";WR(2);"      ";WR(3)
400 GOSUB 720
410 CLS
420 PRINT "      AND";"      ";L4$;"      ";L5$;"      ";L6$;"      ";L7$;"
";L8$:PRINT
430 GOSUB 810:PRINT
440 T=0
450 FOR I=1 TO 6
460 LOCATE 0,3+I+T:PRINT J(I):LOCATE 7,I+3+T:PRINT B(4,I):L
OCATE 14,I+3+T:PRINT B(5,I):LOCATE 21,I+3+T:PRINT B(6,I):LO
CATE 28,I+3+T:PRINT B(7,I):LOCATE 35,I+3+T:PRINT B(8,I)

```

```

470 T=T+1
480 NEXT I
490 GOSUB 810
500 FOR L=4 TO 8
510 X=B(L,1):Y=B(L,6)
520 GOSUB 760
530 NEXT L
540 PRINT "      % ";"      ";WR(4);"      ";WR(5);"      ";WR(6);"      ";WR(7)
;"      ";WR(8)
550 GOSUB 720:CLS
560 PRINT "      AND ";"      ";L9$;"      ";L0$
570 PRINT:GOSUB 810
580 T=0
590 FOR I=1 TO 6
600 LOCATE 6,I+3+T:PRINT J(I):LOCATE 17,I+3+T:PRINT B(9,I):
LOCATE 28,I+3+T:PRINT B(10,I)
610 T=T+1
620 NEXT I
630 PRINT:GOSUB 810
640 FOR L=9 TO 10
650 X=B(L,1):Y=B(L,6)
660 GOSUB 760
670 NEXT L
680 PRINT
690 PRINT "      % ";"      ";WR(9);"      ";WR(10)
700 GOSUB 720
710 CLS:LOCATE 12,12:PRINT "FIN DEL PROGRAMA":END
720 REM      LECTURA DEL TECLADO
730 LOCATE 12,22:PRINT "PULSE UNA TECLA"
740 AS=INKEY$
750 IF AS="" THEN 740 ELSE RETURN
760 REM      TASA DE CRECIMIENTO
770 R=EXP(LOG(Y/X)/25)-1
780 R=R*100:R=FIX(R*100+.5)/100
790 WR(L)=R
800 RETURN
810 REM      RAYADO
820 FOR I=1 TO 40
830 PRINT "-";
840 NEXT I
850 PRINT
860 RETURN
870 DATA 1975,1980,1985,1990,1995,2000
880 REM      MUNDO
890 DATA 4134,4549,5013,5545,6143,6798
900 REM      PAISES INDUSTRIALIZADOS
910 DATA 1131,1174,1224,1276,1327,1377
920 DATA 3003,3375,3789,4269,4816,5420
930 REM      REPUBLICA POPULAR CHINA
940 DATA 978,1071,1151,1241,1348,1468
950 REM      INDIA
960 DATA 618,694,786,894,1013,1142
970 REM      JAPON
980 DATA 112,117,122,127,131,135
990 REM      ESTADOS UNIDOS
1000 DATA 214,222,235,248,260,270
1010 REM      UNION SOVIETICA
1020 DATA 254,268,282,296,310,323
1030 REM      REPUBLICA FEDERAL ALEMANA
1040 DATA 61.8,61.7,60,58.6,57.8,56.2
1050 REM      REPUBLICA DEMOCRATICA ALEMANA
1060 DATA 16.8,16.7,16.6,16.4,16.2,16.1

```

5. PASO : Lista de variables

B = Población  
 I = Índice variable  
 J = Año  
 L = Índice variable  
 L\$ = Países  
 WR = Tasa de crecimiento  
 X = Valor d partida (1975)  
 Y = Valor final (2000)

6. PASO : Descripción del programa

Línea 10-70 : Título y comentarios del programa  
 Línea 80-260 : Lectura de todos los datos de partida  
 Línea 270 : Encabezamiento de la tabla  
 Línea 280 : Salto a la subrutina de subrayado  
 Línea 290-340 : Impresión de los valores de la tabla  
 Línea 350-380 : Determinación de las tasas de crecimiento medias anuales para los países L=1, L=2 y L=3

Línea 360 : Toma del valor inicial y del valor final

Línea 370 : Salto a la subrutina de cálculo de la tasa de crecimiento

Línea 390 : Impresión de tasas de crecimiento  
 Línea 400-410 : Salto a subrutina de espera y borrado  
 Línea 420-540 : Mismo procedimiento que en la línea 270-410, pero para la segunda tabla  
 Línea 550-700 : Lo mismo para la tercera tabla  
 Línea 710 : Final del programa  
 Línea 720-750 : Subrutina de espera  
 Línea 760-800 : Subrutina Tasa de crecimiento media anual (se calcula con ayuda de logaritmos, ver línea 770)  
 Línea 810-860 : Subrutina Raya :  
 Mediante encadenamiento de signos menos se dibuja una raya en la pantalla  
 Línea 870-1060 : Datos

### 7. PASO : Resultados

Este programa genera en la pantalla tres tablas consecutivas :

#### 1. Tabla (referencias en millones)

AÑO	MUNDO	PAISES I	PAISES VD
1975	4134	1131	3003
1980	4549	1174	3375
1985	5013	1224	3789
1990	5545	1276	4269
1995	6143	1327	4816
2000	6798	1377	5420
TASA %	2.01	79	2.39

#### 2. Tabla (referencias en millones)


AÑO	CHINA	INDIA	JAPON	USA	USSR
1975	978	618	112	214	254
1980	1071	694	117	222	268
1985	1151	786	122	235	282
1990	1241	894	127	248	296
1995	1348	1013	131	260	310
2000	1468	1142	135	270	323
%	1.64	2.49	.75	.93	.97

#### 3. Tabla (referencias en millones)

AÑO	RFA	RDA
1975	61.8	16.8
1980	61.7	16.7
1985	60	16.6
1990	58.6	16.4
1995	57.8	16.2
2000	56.2	16.1
%	-.38	-.17

FINAL

Voss	Capítulo	8 : Economía	Página
	Punto	1 : Consideración	255
Colegio		previa	



Capítulo 8 : La economía

=====

8.1 Consideración previa

-----

Durante mucho tiempo, el campo de aplicación fundamental de los ordenadores fue el de la Economía, y en especial el empresarial y comercial. Puede decirse que hoy en día sigue siendo en estos campos, donde se emplean con más frecuencia los ordenadores.

Los problemas tratados en estos campos se prestan a ser resueltos por medio de ordenadores, hoy en día también con microordenadores.

Algunos de los problemas fundamentales de la Economía, por lo menos en lo referente a los cálculos, son tratados también en las escuelas; en escuelas empresariales de forma más intensiva, pero también en otros tipos de escuelas, donde tampoco se evitan este tipo de problemas.

También aquí, los problemas seleccionados sólo son representativos de otros muchos, que por motivos de espacio no pueden ser incluidos en esta obra.

## 8.2 El cálculo de intereses

### 1. PASO : Presentación del problema

Imaginemos que alguien lleva un determinado importe al banco, del que se deducirá el pago de un determinado interés por la imposición del mismo. Este interés puede ser fijado arbitrariamente.

Al final de cada año aumenta el capital en un determinado tanto por ciento. Hay que tener en cuenta, además, que a partir del segundo año se obtiene un interés sobre el capital más el interés del año anterior (interés compuesto).

Vamos a desarrollar ahora un programa que calculará - para un importe inicial arbitrario y para un rédito cualquiera - el importe final obtenido después de una cifra determinada de años.

Una programa como este puede servir además como patrón para la simulación de diversos procesos de crecimiento.

### 2. PASO : Análisis del problema

Si llamamos al capital de partida  $X_0$ , al rédito  $P$  (%) y al tiempo que se tiene el dinero colocado en el banco  $T$  (años).

Después del primer año, el capital resultante  $X_1$  es :

$$X_1 = X_0 + X_0 * P/100 = X_0 * (1 + P/100)$$

Tras el segundo año se obtiene del mismo modo:

$$X_2 = X_1 + X_1 * P/100 = X_1 * (1 + P/100)$$

$$= X_0 * (1 + P/100) * (1 + P/100)$$

$$= X_0 * (1 + P/100)^2$$


(Estas relaciones ya las conocemos de capítulos anteriores).

De esta forma, después de  $T$  años se obtiene:

$$X_T = X_0 * (1 + P/100)^T$$

Con ayuda de esta fórmula podemos confeccionar ahora un programa adecuado.

3. PASO : Diagrama de flujo



```

10 REM #ECON1# -CALCULO DE INTERESES-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA EL CALCULO DE INTERESES."
40 PRINT:PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
50 LOCATE 8,4:PRINT "-----"
60 LOCATE 5,10:INPUT "CAPITAL.....=" ;KO
70 LOCATE 5,15:INPUT "REDITO ANUAL % .....=" ;P
80 LOCATE 5,20:INPUT "TIEMPO DE IMPOSICION ...=" ;T
90 CLS
100 PRINT:PRINT "          AÑO          INTERESES          CAPITAL "
110 PRINT:PRINT "          ---          -----          -----"
120 M=6
130 FOR I=1 TO T
140 Z=KO*P/100
150 KT=KO+Z
160 Z=FIX(Z*100+.5)/100
170 KT=FIX(KT*100+.5)/100
180 LOCATE 5,M:PRINT I:LOCATE 13,M:PRINT Z:LOCATE 26,M:PRINT
I KT
190 KO=KT
200 IF I/7=FIX(I/7) THEN GOSUB 280
210 M=M+2:IF M>18 THEN M=6
220 NEXT I
230 LOCATE 10,22:PRINT "OTRO CALCULO (S/N)"
240 AS=INKEY$
250 IF AS="S" OR AS="s" THEN CLS:RESTORE:GOTO 10
260 IF AS="N" OR AS="n" THEN 270 ELSE 240
270 CLS:LOCATE 12,12:PRINT "FIN DEL PROGRAMA":END
280 REM          SUBROUTINA DE ESPERA
290 LOCATE 5,22:PRINT "Pulse una tecla para continuar"
300 AS=INPUT$(1)
310 CLS:PRINT:PRINT "          AÑO          INTERESES          CAPITAL "
320 PRINT:PRINT "          ---          -----          -----"
330 RETURN
    
```

5. PASO : Lista de variables

A\$ = Variable de cadena para las respuestas  
i = Índice variable  
K0 = Capital inicial  
KT = Capital en los períodos siguientes  
P = Rédito anual en tanto por ciento  
T = Tiempo de imposición en años  
Z = Interés anual

6. PASO : Descripción del programa

Línea 10-50 : Título

Línea 60-80 : Demanda de los datos de partida

Línea 90-110 : Limpieza de la pantalla e impresión de un encabezamiento de tabla

Línea 120-220 : Cálculos

140 : Interés  
150 : Capital tras el siguiente período  
160-170 : Redondeo  
180 : Impresión  
200 : K0 toma el valor KT, para poder entrar en el siguiente cálculo  
210 : Después de cada 7 períodos interrupción mediante el salto a la subrutina 280

Línea 230-270 : Final del programa, a menos que se desee seguir calculando (vuelta a la línea 10)

Línea 280-330 : Subrutina para esperar cuando la pantalla está llena:

290 : Indicación  
300 : Espera  
310-320 : Borrado de la pantalla y nueva impresión del encabezamiento  
330 : Salto de vuelta

7. PASO : Resultados

Si introducimos el capital inicial Ptas. 100, el rédito 5.5% y el tiempo 6 años, se obtiene :

AÑO	INTERES	CAPITAL
1	5.5	105.5
2	5.2	113.3
3	6.12	117.42
4	6.46	123.88
5	6.81	130.69
6	7.19	137.88

OTRO CALCULO (S/N)

### 8.3 Amortización de una hipoteca

---

#### 1. PASO : Presentación del problema

De entre los problemas de la Economía, aquéllos que están relacionados con el pago de créditos e hipotecas juegan también un papel importante. Hay programas de ordenador que pueden calcular el tiempo de amortización bajo diversas condiciones de partida, como p.e. valor de la deuda, el interés o las modalidades de pago.

El programa siguiente calcula el tiempo de amortización para diferentes importes e intereses o cuando se han acordado cuotas de amortización de distintas cuantías.

Puede caerse fácilmente en la cuenta de que este programa es aplicable de forma más general en problemas de este tipo.

#### 2. PASO : Análisis del problema

Para explicar la resolución del problema nos basaremos en el siguiente ejemplo:

Alguien solicita una hipoteca de S Ptas. y acuerda con el banco concesor un pago mensual de B Ptas. De este importe mensual B deben restarse primero los intereses por la deuda. El importe que queda después de restar los intereses puede aplicarse para saldar la deuda - suponiendo que quede un resto positivo.


El programa que vamos a confeccionar debe calcular mes a mes los intereses y el capital de amortización. Si este importe es igual o menor a 0, el programa debe imprimir un mensaje para avisar de que bajo esas condiciones ya no es posible amortizar la deuda.

Debe imprimirse año por año un estado de la deuda, y cuando la deuda sea cero o negativa, deberá indicarse la duración del proceso de amortización.



3. PASO : Diagrama de flujo

S = Deuda,  
 P = Interés,  
 B = Importe  
 mensual



```

10 REM #ECON2# -CALCULO DE HIPOTECAS-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA CALCULAR EL TIEMPO DE A-"
40 PRINT:PRINT " MORTIZACION DE UNA HIPOTECA."
50 PRINT:PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
60 LOCATE 8,6:PRINT "-----"
70 LOCATE 5,10:INPUT "CUOTA ANUAL....." ;B
80 LOCATE 5,15:INPUT "INTERES EN % ....." ;P
90 LOCATE 5,20:INPUT "DEUDA HIPOT. ACTUAL....." ;S
100 CLS
110 PRINT:PRINT "          AÑO          DEUDA "
120 PRINT:PRINT "          ---          ----"
130 H=5
140 J=1
150 M=1
160 Z=((P/100)*S)/12:T=B-Z
170 IF T<=0 THEN 230
180 S=FIX((S-T)*100+.5)/100
190 IF S<=0 THEN 290
200 M=M+1:IF M<13 THEN 160
210 LOCATE 10,H:PRINT J:LOCATE 23,H:PRINT S
220 J=J+1:H=H+2:GOTO 150
230 CLS:LOCATE 2,7:PRINT "LAS CONDICIONES NO PERMITEN AMORTI
    IZAR"
240 PRINT:PRINT " LA DEUDA. INTRODUZCA NUEVOS DATOS POR"
250 PRINT:PRINT " FAVOR."
260 LOCATE 11,20:PRINT "PULSE UNA TECLA"
270 AS=INPUT$(1)
280 CLS:GOTO 10
290 LOCATE 13,22:PRINT "PULSE UNA TECLA"
300 AS=INPUT$(1):CLS
310 LOCATE 9,10:PRINT "TIEMPO DE AMORTIZACION : "
320 LOCATE 10,13:PRINT J-1;" AÑOS ";M-1;" MESES "
330 LOCATE 11,22:PRINT "OTRO CALCULO (S/N)"
340 AS=INKEY$
350 IF AS="S" OR AS="s" THEN CLS:GOTO 10
360 IF AS="N" OR AS="n" THEN 370 ELSE 340
370 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
    
```

Voss	Capítulo	8 : Economía	Página
	Punto	3 : Hipoteca	266
Colegio			

5. PASO : Lista de variables

A\$ = Variable de cadena (si,no)  
 B = Cuota mensual  
 I = Índice variable  
 J = Contador de años  
 M = Contador de meses  
 P = Interés anual de la deuda  
 T = Importe mensual de amortización  
 Z = Pago mensual de intereses

6. PASO : Descripción del programa

Línea 10-60 : Título

Línea 70-90 : Petición de las informaciones de partida

Línea 100-120 : Impresión del encabezamiento de la tabla

Línea 130-150 : Fijación del contador de años (J) y del contador de meses M

Línea 160-180 : Cálculo del interés mensual, de la cuota mensual y de la nueva deuda

Línea 170 : En caso de no poder amortizar, seguir en 230

Línea 190 : Una vez saldada la deuda, seguir en 290

Voss	Capítulo	8 : Economía	Página
	Punto	3 : Hipoteca	267
Colegio			

Línea 200 : Mes siguiente  
 si N es menor que 13 seguir en 160, en caso contrario seguir en la línea 210

Línea 210 : Impresión de la deuda actual

Línea 220 : Año siguiente y seguir en línea 150

Línea 230-280 : Mensaje sobre condiciones no permitidas y vuelta a la línea 10

Línea 290-320 : Impresión del tiempo de amortización en años y en meses (piensa porque se le resta cada vez el valor 1 a J y a N)

Línea 330-370 : Final del programa, a menos que no se desee un nuevo cálculo (entonces retornar a la línea 10 tras borrar la pantalla)

7. PASO : Resultados

Introduzcamos estos datos cuando el ordenador nos lo pida:

CUOTA ANUAL : ? 500  
 INTERES EN % : ? 6.5  
 DEUDA HIPOT. ACTUAL.: ? 40000

y obtendremos en el transcurso del programa :

AÑO	DEUDA
1	36496.86
2	32759.09
3	28771.02
4	24515.85
5	19975.7
6	15131.48
7	9962.85
8	4448.07

TIEMPO DE AMORTIZACION :

8 AÑOS Y 9 MESES.

OTRO CALCULO (S/N) ?

8.4 Media aritmética

1. PASO : Presentación del problema

En algunos problemas de economía resulta necesario calcular el promedio de un número determinado de datos; normalmente se trata de la media aritmética.

El programa siguiente efectúa este cálculo para un número cualquiera de datos.

2. PASO : Análisis del problema


El análisis del problema no presenta especiales dificultades.

Primero hay que procurar que se introduzca un número no fijado de datos, por lo que se ha de informar al ordenador de cuántos valores se van a introducir.

Al introducir los valores puede efectuarse automáticamente la suma acumulativa de los mismos (la media aritmética no es más que la suma de todos los valores dividida entre el número de ellos). La división de la suma resultante del número de valores puede imprimirse como resultado del programa.

Para poder efectuar más cálculos, el programa se bifurca al principio. Si no se desea continuar finaliza él mismo.

### 3. PASO : Diagrama de flujo



### 4. PASO : Programa

```

10 REM #ECON3# -VALOR MEDIO-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS:S=0
30 PRINT " PROGRAMA PARA CALCULAR UNA MEDIA ARIT-"
40 PRINT:PRINT " METICA."
50 PRINT:PRINT:PRINT " PROF. DR. W. VOSS, 1984"
60 LOCATE 8,6:PRINT "-----"
70 LOCATE 5,12:INPUT "CUANTOS VALORES.....: ";N:IF N=
0 THEN 180
80 FOR I=1 TO N
90 LOCATE 5,17:PRINT "VALOR...(;I;)".....: ";LOCAT
E 32,17:PRINT " ";LOCATE 32,17:INPUT X
100 S=S+X
110 NEXT I
120 AM=S/N
130 LOCATE 5,17:PRINT "MEDIA ARITMETICA.....=" "; " ";LO
CATE 33,17:PRINT AM
140 LOCATE 10,22:PRINT "OTRO CALCULO (S/N)"
150 AS=INKEY$
160 IF AS="S" OR AS="s" THEN CLS:GOTO 10
170 IF AS="N" OR AS="n" THEN 180 ELSE 150
180 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
  
```

### 5. PASO : Lista de variables

AM = Media aritmética  
AS = Variable de cadena (sí/no)  
I = índice variable  
N = Número de valores a promediar  
S = Suma de los valores a promediar  
X = Valor que entra en el cálculo del valor medio

Voss	Capítulo	8 : Economía	Página
	Punto	4 : Promedio	272
Colegio			

## 6. PASO : Descripción del programa

Línea 10-60 : Título

Línea 70 : Indicación del número de valores a promediar

Línea 80-110 : Introducción de un valor tras otro y suma acumulativa de todos los valores

Línea 120-130 : Cálculo de la media aritmética e impresión del resultado

Línea 140-180 : Final del programa, a menos que no se desee continuar calculando

## 7. PASO : Resultados

Si introducimos p.e., que existen tres valores, el programa pedirá exactamente tres valores.

Si éstos son p.e. los valores 5, 7 y 9, obtendremos:

MEDIA ARITMETICA = 7

OTRO CALCULO (S/N) ?

Voss	Capítulo	8 : Economía	Página
	Punto	5 : Distribución	273
Colegio			

## 8.5 Distribución de frecuencias

### 1. PASO : Presentación del problema

A la hora de valorar volúmenes de datos grandes, que es cuando los ordenadores se muestran especialmente útiles, uno de los primeros pasos de la investigación estadística consiste en realizar una llamada distribución de frecuencias.

Se trata de asignar mediante una tabla o un gráfico la frecuencia con que aparecen los valores característicos (o a las clases de valores característicos) de la variable objeto de la investigación.

El programa siguiente realiza esta distribución de frecuencias tanto gráficamente como en forma de tabla. Como material de valoración utilizaremos la distribución por edades de la población de la RFA en el año 1980.

Esta distribución por edades de la población federal alemana es la siguiente:

Distribución de edades en la República Federal de Alemania en 1980:

Grupos de edades	Porcentaje
menos de 10	10.3
10 hasta 20	16.4
20 hasta 30	14.6
30 hasta 40	13.4
40 hasta 50	14.2
50 hasta 60	11.9
60 hasta 70	8.9
70 hasta 80	7.8
80 hasta 90	2.4
90 y más	0.2

2. PASO : Análisis del problema

Para el análisis del problema partiremos de la base de que disponemos de los datos tal como aparecen en la tabla, de forma que puedan ser utilizados como material de partida para el programa a confeccionar.

De no ser este el caso, y tuviéramos datos sueltos obtenidos en una encuesta, deberíamos procurar en una primera parte del programa la clasificación de estos datos sueltos por grupos de edades (esto puede conseguirse mediante una serie de instrucciones IF...THEN, que comprobarían cada uno de los valores de edad).

Una vez hecho esto, podemos relativizar los valores obtenidos para cada grupo de edades, comparándolos con la totalidad. De este modo, obtenemos los porcentajes que ya aparecen en la tabla anterior.

Esta parte del programa podría resumirse esquemáticamente de la forma siguiente:

```

100 INPUT "Edad siguiente : ";X
110 IF X<10 THEN Z1=Z1+i : GOTO 100
120 IF X<20 THEN Z2=Z2+i : GOTO 100
...
etc.
200 R1=(Z1/N)*100
210 R2=(Z2/N)*100
...
etc.

```


En este ejemplo se trata únicamente de representar en forma de tabla los valores introducidos, cosa que lógicamente no ofrece ninguna dificultad desde el o de vista de la programación.

La representación gráfica de esta distribución de frecuencias, en cambio, es más complicada.

Para poder aprovechar óptimamente la pantalla en la representación gráfica, determinaremos, en primer lugar, cuál es la frecuencia mayor y, a continuación, convertiremos todas las demás frecuencias de forma que la primera ocupe 15 líneas en vertical de la pantalla.

Las frecuencias en sí se representan en forma de barras verticales. Para que el tamaño de estas barras sea correcto, debe determinarse su correspondiente punto final. Junto con el punto de origen obtendremos dos posiciones de pantalla por barra y ya tan solo será necesario conectar con una línea.

### 3. PASO : Diagrama de flujo



```

10 REM #ECON4# -DISTRIBUCION-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS:S=0
30 PRINT "PROGRAMA PARA REPRESENTAR EN UNA TABLA Y"
40 PRINT "GRAFICAMENTE UNA DISTRIBUCION ESTADISTI-"
50 PRINT "CA."
60 PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
70 LOCATE 8,8:PRINT "-----"
80 LOCATE 0,12:PRINT " Se representa la distribución de ed
ades"
90 PRINT:PRINT " en la República Federal Alemana en 1980"
100 LOCATE 12,22:PRINT "PULSE UNA TECLA"
110 AS=INPUT$(1)
120 DIM F(10),FT(10)
130 CLS
140 FOR I=1 TO 10
150 READ X
160 F(I)=X
170 NEXT I:COLOR 1,15,15
180 PRINT "          EDAD";"          ":"PORCENTAJE"
190 FOR I=0 TO 39:LOCATE I,1:PRINT "-";:NEXT I
200 H=0
210 FOR I=1 TO 10
220 A=(I-1)*10:B=I*10
230 LOCATE 1,2+H:PRINT A:LOCATE 7,2+H:PRINT"HASTA MENOS DE
":LOCATE 22,2+H:PRINTB:LOCATE 30,2+H:PRINT F(I):LOCATE 36,2
+H:PRINT "%"
240 H=H+2
250 NEXT I
260 LOCATE 12,22:PRINT "PULSE UNA TECLA"
270 AS=INPUT$(1)
280 FM=0:CLS
290 FOR I=1 TO 10
300 IF F(I)>FM THEN FM=F(I)
310 NEXT I
320 FOR I=1 TO 10
330 FT(I)=F(I)*(15/FM)
340 NEXT I
350 B=21
360 FOR I=1 TO 10
370 A=21-FT(I):A=FIX(A+.5)
380 FOR Z=A TO B
390 LOCATE 4*I-2,Z
400 PRINT "*"
410 NEXT Z
420 NEXT I
430 LOCATE 0,22
440 FOR I=0 TO 39
450 PRINT"*";
460 NEXT I
470 FOR I=1 TO 10
480 PRINT(I*10);
490 NEXT I
500 END
510 DATA 10.3,16.4,14.6,13.4,14.2
520 DATA 11.9,8.9,7.8,2.4,.2

```

### 5. PASO : Lista de variables

A = Límite inferior del grupo de edades  
después : dirección de inicio de la columna

B = Límite superior del grupo de edades  
después : dirección final de la columna

F = Frecuencia

FM = Frecuencia máxima

FT = Frecuencia transformada

I = Índice variable

H = Línea de pantalla

### 6. PASO : Descripción del programa

Línea 10-90 : Título

Línea 100-110 : Espera

Línea 120-170 : Dimensionado y lectura de los datos de partida

Línea 180-190 : Impresión del encabezamiento de la tabla

Línea 210-250 : Impresión de la tabla:

220 : Determinación del punto de inicio del grupo

230 : Impresión



Voss	Capítulo	8 : Economía	Página
	Punto	5 : Distribución	280
Colegio			

- Línea 260-270 : Espera
- Línea 280-310 : Búsqueda de la frecuencia mayor y comparación de todas las frecuencias con cada frecuencia máxima alcanzada (empezando por el valor 0)
- Línea 320-340 : Determinación de la frecuencia transformada mediante adaptación de las frecuencias a la frecuencia máxima
- Línea 350-420 : Trazado de la gráfica:  
350 : Final de la columna  
370 : Determinación del comienzo de la columna medido con respecto al reparto de líneas  
380-410 : Trazado de la columna mediante el encadenamiento de asteriscos
- Línea 430-460 : Eje horizontal
- Línea 470-490 : Escalado del eje horizontal
- Línea 500 : Final del programa
- Línea 510-520 : Datos

Voss	Capítulo	8 : Economía	Página
	Punto	5 : Distribución	281
Colegio			

#### 7. PASO : Resultados

No vamos a ofrecer los resultados de este programa. Nos parece mucho más útil que el lector introduzca el programa en su ordenador y que después intente explicar paso a paso lo que sucede, comparando las impresiones en pantalla con la descripción que hacemos.

No obstante, creemos conveniente hacer la siguiente puntualización:

El gráfico que traza este programa no utiliza las instrucciones BASIC que hay disponibles para los gráficos de alta resolución, sino que se genera mediante instrucciones PRINT.

Esto tiene como consecuencia que el dibujo sea poco preciso. Las columnas de diagrama no representan exactamente las frecuencias, porque la unidad básica de trazado de las mismas es un cuadrado del tamaño del cursor, es decir con un lado de 7 mm. (en un monitor normal de ordenador).

Para obtener gráficos más exactos no queda más remedio que utilizar los gráficos de alta resolución, como haremos en el capítulo siguiente.

## 8.6 La economía de los países europeos

---

### 1. PASO : Presentación del problema

El programa siguiente es muy distinto a los programas precedentes. Aquí se trata nuevamente de confeccionar un "programa Informativo".

Vamos a desarrollar un programa que visualiza datos importantes sobre la economía nacional cuando el usuario lo requiere.

Estos datos se refieren a los siguientes campos:

1. Población (en millones)
2. Superficie (en miles de Km. cuadrados)
3. Producto interior bruto (en miles de millones de Ptas.)
4. Población activa (en millones)
5. Proporción de trabajadores extranjeros (en %)
6. Impuestos totales (en miles de millones de Ptas.)
7. Deudas (en miles de millones de Ptas.)
8. Porción de la producción industrial (en %)

El usuario puede alargar esta lista o utilizar otros conceptos en lugar de los países (p.e. Comunidades Autónomas).


Lógicamente, habrá que modificar el fichero de datos para adaptar el programa.

### 2. PASO : Análisis del problema

El lector habrá podido comprobar que a lo largo de estos capítulos hay ciertos problemas que se repiten o reaparecen de forma similar a la originaria. Aquí nos encontramos de nuevo ante este problema: En principio no se presentan nuevos problemas y el análisis resulta ciertamente sencillo.

Primero hay que introducir todas las informaciones relevantes. A continuación se pregunta al usuario el nombre del país acerca del que desea obtener informaciones. Tras introducir este nombre tiene lugar la impresión del juego de datos correspondiente.

3. PASO : Diagrama de flujo



```

10 REM #ECONOMIA# -ECONOMIA EUROPEA-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT:PRINT "PROGRAMA PARA IMPRIMIR INFORMACIONES SO-"
40 PRINT:PRINT "BRE EL PODER ECONOMICO DE DIFERENTES PAI"
50 PRINT:PRINT "SES EUROPEOS."
60 PRINT:PRINT " " PROF. DR. W. VOSS, 198
70 LOCATE 8,12:PRINT "-----"
80 LOCATE 11,22:PRINT "PULSE UNA TECLA"
90 AS=INPUT$(1)
100 CLS
110 DIM L$(26),U(8,18)
120 DIM B1$(26),B2$(26),B3$(26),B4$(26)
130 DIM B5$(26),B6$(26),B7$(26),B8$(26)
140 PRINT " DE QUE PAIS DESEA OBTENER INFORMACION ?"
150 FOR I=1 TO 18
160 READ L$
170 PRINT " ";I," ";L$
180 NEXT I
190 LOCATE 9,22:PRINT "INDIQUE UN NUMERO"
200 LOCATE 27,22:INPUT Z
210 IF Z<1 OR Z>18 THEN 190
220 CLS
230 LOCATE 10,12:PRINT "UN MOMENTO POR FAVOR"
240 FOR J=1 TO 8
250 FOR I=1 TO 18
260 READ X
270 U(J,I)=X
280 NEXT I
290 NEXT J
300 RESTORE
310 FOR I=1 TO Z
320 READ L$
330 NEXT I
340 B1$="POBLACION (mill.)"
350 B2$="SUPERFICIE (x1000 Km2)"
360 B3$="PROD. INTERIOR (x10ES)"
370 B4$="POBL. ACTIVA (mill.)"
380 B5$="PROP. EXTRANJEROS (%)"
390 B6$="IMP. TOTALES (x10ES)"
400 B7$="DEUDAS (x10ES)"
410 B8$="PROD. INDUSTRIAL (%)"
420 CLS
430 PRINT " ";L$
440 LOCATE 0,1:FOR Q=0 TO 39:PRINT "-";:NEXT Q
450 PRINT:PRINT B1$;" ";U(1,Z)
460 PRINT:PRINT B2$;" ";U(2,Z)
470 PRINT:PRINT B3$;" ";U(3,Z)
480 PRINT:PRINT B4$;" ";U(4,Z)
490 PRINT:PRINT B5$;" ";U(5,Z)
500 PRINT:PRINT B6$;" ";U(6,Z)
510 PRINT:PRINT B7$;" ";U(7,Z)
520 PRINT:PRINT B8$;" ";U(8,Z)
530 LOCATE 12,22:PRINT "PULSE UNA TECLA"
540 AS=INPUT$(1)
550 CLS
560 LOCATE 11,12:PRINT "OTRO PAIS MAS (S/N)"
570 AS=INKEY$
  
```

```

580 IF AS="S" OR AS="s" THEN CLS:RESTORE:GOTO 140
590 IF AS="N" OR AS="n" THEN 600 ELSE 570
600 CLS:LOCATE 12,11:PRINT "FIN DEL PROGRAMA":END
610 DATA "ESPAÑA","FRANCIA","ITALIA"
620 DATA "REINO UNIDO","RFA","IRLANDA"
630 DATA "GRECIA","BELGICA","HOLANDA"
640 DATA "SUECIA","DINAMARCA","PORTUGAL"
650 DATA "NORUEGA","FINLANDIA","SUIZA"
660 DATA "LUXEMBURGO","AUSTRIA","MALTA"
670 DATA 0,0,0,0,0,0,0,0,0,0
680 DATA 0,0,0,0,0,0,0,0,0,0
690 DATA 0,0,0,0,0,0,0,0,0,0
700 DATA 0,0,0,0,0,0,0,0,0,0
710 DATA 0,0,0,0,0,0,0,0,0,0
720 DATA 0,0,0,0,0,0,0,0,0,0
730 DATA 0,0,0,0,0,0,0,0,0,0
740 DATA 0,0,0,0,0,0,0,0,0,0
750 DATA 0,0,0,0,0,0,0,0,0,0
760 DATA 0,0,0,0,0,0,0,0,0,0
770 DATA 0,0,0,0,0,0,0,0,0,0
780 DATA 0,0,0,0,0,0,0,0,0,0
790 DATA 0,0,0,0,0,0,0,0,0,0
800 DATA 0,0,0,0,0,0,0,0,0,0
810 DATA 0,0,0,0,0,0,0,0,0,0

```

#### 5. PASO : Lista de variables

A\$ = Variable de cadena  
B\$ = Denominaciones de variables  
I = Variable índice  
J = Variable índice  
L\$ = Nombre de los países  
V = Variables  
X = Campo de datos  
Z = Número del país buscado

#### 6. PASO : Descripción del programa

Línea 10-90 : Título y espera

Línea 100-130 : Borrado y dimensionado  
Línea 140-210 : Requisirimiento del número del país del cual se necesitan informaciones después de ofrecer una lista de países en la pantalla.  
Línea 220-290 : Borrado de la pantalla y lectura de los datos  
Línea 300 : Inicializa el puntero de datos  
Línea 310-330 : Búsqueda del país seleccionado  
Línea 340-410 : Introducción de los nombres de las variables  
Línea 430 : Impresión del nombre del país  
Línea 440-520 : Impresión del registro  
Línea 530-540 : Espera  
Línea 550-600 : Consulta si desea otra información (en caso afirmativo, borrar la pantalla y volver a la línea 140, en caso negativo final del programa)  
Línea 610-660 : DATAS de los nombres de los países  
Línea 670-790 : DATAS de los valores de las variables: introducir los valores de tal manera, que para la primera variable se den todos los valores sobre el primer país, para la segunda variable todos los valores sobre el segundo país,... etc.

**ATENCIÓN : EN EL PROGRAMA APARECEN SOLO CEROS**  
(¡ver siguiente paso !)

Voss	Capítulo	8 : Economía	Página
	Punto	6 : Países	288
Colegio			

### 7. PASO : Resultados

El fichero de datos de este programa contiene únicamente ceros, de manera que si hacemos una pasada de prueba, sólo aparecerán ceros en la pantalla.

Para efectuar una pasada "auténtica", hay que introducir, primero, datos concretos, que el lector puede tomar de la tabla siguiente.

Datos económicos de diferentes países europeos :

País	Pobl. (mill.)	Superf. (1000 km2)	Produc. Interior (10E9)	Poblab. activa (mill.)
-----				
España				
Francia				
Italia				
Reino U.				
RFA				
Irlanda				
Grecia				
Bélgica				
Holanda				
Suecia				
Dinamarca				
Portugal				
Noruega				
Finlandia				
Suiza				
Luxemburgo				
Austria				
Malta				
-----				

Voss	Capítulo	8 : Economía	Página
	Punto	6 : Países	289
Colegio			

País	Prop. extran- jeros	Impues. totales (10E9)	Deudas. (10E9 Ptas.	Produc. Indust. (%)
-----				
España				
Francia				
Italia				
Reino U.				
RFA				
Irlanda				
Grecia				
Bélgica				
Holanda				
Suecia				
Dinamarca				
Portugal				
Noruega				
Finlandia				
Suiza				
Luxemburgo				
Austria				
Malta				
-----				



## Capítulo 9 : Matemáticas II

---

### 9.1 Consideración previa

---

Para finalizar incluimos un capítulo en el que se tratan nuevos problemas matemáticos, con un grado de dificultad algo mayor que los vistos hasta ahora. De esta forma, el usuario experimentado en la utilización del microordenador, comprobará que también se pueden abordar problemas de mayor complejidad con programas relativamente sencillos.

Llegados a este punto, podría parecer que el capítulo final se halla un poco por encima de las posibilidades del principiante (pero es precisamente esto lo que a veces le proporciona más interés).

Estos problemas de mayor complejidad, deben ser programados de tal forma que no requieran instrucciones BASIC adicionales. Esto significa, que los programas resultantes (al igual que todos los programas precedentes) no pueden satisfacer los niveles de elegancia de los programadores experimentados. Pero esto aquí no tiene importancia.

En contraposición a los anteriores capítulos, aquí seremos más breves, pues se van a tratar diecisiete nuevos problemas:

Después de una pequeña presentación del problema y de algunas indicaciones con referencia a su análisis, pasamos inmediatamente a presentar el programa y terminamos con una pequeña descripción del mismo. En adelante prescindiremos de los pasos "Diagrama de flujo" y "Resultados", no sólo por motivos de espacio, sino porque en cuanto al aspecto de la programación pura no introducen nuevos problemas de detalle.

## 9.2 Red

---

### Presentación del problema

---

El programa siguiente sirve para trazar en la pantalla una red de líneas con separaciones entre líneas y columnas variable.

El esquema mental utilizado como base de estos programas, sirve de forma parecida por ejemplo para estructurar tablas extensas con líneas horizontales y verticales.

### Análisis del problema

---

Podemos ahorrarnos cualquier explicación acerca del análisis del problema. El usuario únicamente debe introducir las distancias entre líneas y entre columnas. El trazado de las líneas se efectúa mediante dos bucles FOR-NEXT y la instrucción BASIC 'LINE'.

```

10 REM #MII/1# -RETICULA-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT:PRINT "PROGRAMA PARA TRAZAR UNA RED EN PANTALLA"
40 PRINT:PRINT:PRINT " PROF. DR. W. VOSS, 1984"
50 LOCATE 8,6:PRINT "-----"
60 LOCATE 1,10:INPUT "DISTANCIA ENTRE COLUMNAS (C<256) ";S
70 LOCATE 1,16:INPUT "DISTANCIA ENTRE FILAS (F<192) ";Z
80 LOCATE 11,22:PRINT "PULSE UNA TECLA"
90 AS=INPUT$(1)
100 CLS
110 SCREEN 2:CLS
120 FOR I=0 TO 191 STEP Z
130 LINE (0,I)-(255,I),10
140 NEXT I
150 FOR I=0 TO 255 STEP S
160 LINE (I,0)-(I,191),10
170 NEXT I
180 AS=INPUT$(1)
190 SCREEN 0:COLOR 1,15,15
200 LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END

```

Lista de variables:

---

I = Variable índice  
S = Separación entre columnas  
Z = Separación entre líneas

Descripción del programa:

---

Línea 10-50 : Título

Línea 60-70 : Introducción de la distancia entre líneas y entre columnas

Línea 80-100 : Espera una entrada desde el teclado y borra la pantalla

Línea 110 : Conmutación al modo gráfico de alta resolución

Línea 120-140 : Trazado de las líneas horizontales

Línea 150-170 : Trazado de las líneas verticales

Línea 180-190 : Bucle de espera y conmutación al modo de texto

Línea 200 : Final del programa



### 9.3 La recta

---

#### Presentación del problema

---

El programa siguiente sirve para dibujar cualquier recta en un eje de coordenadas cuyo origen se sitúa en el centro de la pantalla (127,95).

#### Análisis del problema

---

La posición de una recta queda determinada por dos parámetros, la ordenada en el origen A y la tangente del ángulo que forma con el eje de ordenadas B, según la siguiente ecuación:

$$Y_i = A + B \cdot X_i$$

Dado que el origen de coordenadas debe estar situado en (127,95), resulta que:

$$Y_i = 95 + A + B \cdot (X_i - 127)$$

En todo caso, debemos tener en cuenta que en contraposición al eje de coordenadas normal, los valores de Y empiezan en la parte superior de la pantalla (la línea número 0 es la línea superior de la pantalla).

Por esta razón, debe modificarse la función de la forma siguiente:

$$\begin{aligned} Y_i &= 127 - (95 + A + B \cdot (X_i - 127)) \\ &= 95 - A - B \cdot (X_i - 127) \end{aligned}$$

```

10 REM #MII/2# -RECTA-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT:PRINT " -PROGRAMA PARA TRAZAR UNA LINEA RECTA."
40 PRINT:PRINT:PRINT " PROF. DR. W. VOSS, 1984"
50 LOCATE 8,6:PRINT "-----"
60 LOCATE 6,10:INPUT "ABCISA EN EL ORIGEN : ";A
70 LOCATE 6,16:INPUT "PENDIENTE DE LA RECTA : ";B
80 LOCATE 11,22:PRINT "PULSE UNA TECLA"
90 AS=INPUT$(1)
100 SCREEN 2:CLS
110 DRAW "BMO,95R255BM127,0D191"
120 FOR X=0 TO 255
130 Y=95-A-B*(X-127)
140 IF Y<=0 OR Y>=191 THEN 160
150 PSET (X,Y),15
160 NEXT X
170 AS=INPUT$(1)
180 LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END

```

#### Lista de variables:

---

A = Abcisa en el origen  
B = Tangente del ángulo de pendiente  
I = Contador  
X = Valores sobre el eje X  
Y = Valores de la función

Descripción del programa:  
-----

Línea 10-50 : Título

Línea 60-70 : Introducción de los parámetros de la recta.

Línea 80-100 : Espera y conmuta a modo gráfico

Línea 110 : Trazado de los ejes de coordenadas

Línea 120-160 : Trazado de la recta sobre todo el ancho de la pantalla (si los valores de Y son no válidos, se salta la instrucción)

Línea 170 : Bucle de espera

Línea 180 : Final del programa

9.4 El círculo  
-----

Presentación del problema  
-----

Siguiendo el mismo patrón del punto anterior, donde se desarrolló un programa para trazar una recta, de nuevo presentamos un programa para dibujar todo tipo de círculos. Aquí aparecen problemas ya tratados, de modo que seremos breves.

Análisis del problema  
-----

Nos enfrentamos aquí al mismo problema que en el punto anterior, con la diferencia de que aquí tomamos como base la ecuación del círculo:

Esta se expresa de la forma siguiente:

$$(x - a)^2 + (y - b)^2 = r^2$$

r es el radio del círculo, x e y son las coordenadas de la línea circular y a y b son las coordenadas del centro del círculo.

Si despejamos  $y$  (resolución de una ecuación cuadrática), obtenemos las ecuaciones para el semicírculo superior e inferior:

$$(y - b)^2 = r^2 - (x - a)^2 = D$$

$$y - b = \pm \sqrt{D}$$

$$y = \pm b \pm \sqrt{D}$$

Con ello hemos aclarado los aspectos fundamentales de este programa:

Requiere como informaciones de entrada las coordenadas del centro del círculo  $a$  y  $b$  (que a partir de ahora llamaremos  $T$  y  $Z$ ) y el radio  $r$  (en BASIC :  $R$ ). A partir de aquí podrán determinarse los valores  $Y1$  e  $Y2$  correspondientes a diferentes valores de  $X$  y después trazar los puntos  $X, Y1$  y  $X, Y2$ .

Debe prestarse atención a que  $X$  se mueva únicamente en el ámbito que va de  $T-R$  a  $T+R$ , porque fuera de esta zona no hay valores de  $Y$  definidos dentro del campo real (el valor auxiliar  $D$  sería negativo, y al tratar de extraer la raíz del mismo provocaría un mensaje de error y una interrupción del programa.

```

10 REM #MII/3# -CIRCULO-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT:PRINT "PROGRAMA PARA TRAZAR UNA CIRCUNFERENCIA"
40 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
50 LOCATE 8,6:PRINT "-----"
60 LOCATE 2,10:INPUT "COORDENADA X DEL ORIGEN : ";T
70 LOCATE 2,14:INPUT "COORDENADA Y DEL ORIGEN : ";Z
80 LOCATE 2,18:INPUT "RADIO DE LA CIRCUNFERENCIA : ";R
90 LOCATE 11,22:PRINT "PULSE UNA TECLA"
100 AS=INPUT$(1)
110 SCREEN 2:CLS
120 DRAW "BMO,95R255BM127,0D191"
130 LINE (T,0)-(T,191),2
140 LINE (0,Z)-(255,Z),2
150 A=T-R:B=T+R
160 IF A<0 THEN A=0
170 IF B>255 THEN B=255
180 FOR X=A TO B
190 D=R*R-(X-T)^2
200 IF D<=0 THEN Y=Z:GOTO 230
210 Y=Z+SQR(D)
220 IF Y<0 OR Y>191 THEN 240
230 PSET (X,Y),13
240 NEXT X
250 FOR X=B TO A STEP -1
260 D=R*R-(X-T)^2
270 IF D<=0 THEN Y=Z:GOTO 300
280 Y=Z-SQR(D)
290 IF Y<0 OR Y>191 THEN 310
300 PSET (X,Y),13
310 NEXT X
320 AS=INPUT$(1)
330 SCREEN 0:COLOR 15,1,1:CLS:LOCATE 11,12:PRINT "FIN DEL P
ROGRAMA":END

```

#### Lista de variables

A = Punto extremo izquierdo del círculo  
B = Punto extremo derecho del círculo  
D =  $D = R - (X-T)$   
I = Índice variable  
R = Radio  
T = Columna del centro del círculo  
X = Valores de coordenada X  
Y = Valores de coordenada Y  
Z = Línea del centro del círculo

#### Descripción del programa

Línea 10-50 : Título y aclaraciones  
Línea 60-110 : Informaciones y modo gráfico  
Línea 120 : Eje de coordenadas  
Línea 130-140 : Eje de coordenadas auxiliar  
Línea 150-170 : Fijación del ámbito de valores de X  
Línea 180-240 : Trazado de un semicírculo  
Línea 250-310 : Trazado del otro semicírculo  
Línea 320-330 : Final del programa

#### 9.5 La senoide

#### Presentación del problema

En este último ejemplo de gráfico de alta resolución vamos a representar una oscilación sinusoidal, como la que ya nos encontramos en el capítulo "Física". El procedimiento a seguir es el mismo que en los dos puntos precedentes.

#### Análisis del problema

El análisis del problema es muy sencillo, y no se presentan cuestiones nuevas. Únicamente debe procurarse adaptar la periodicidad de la oscilación (frecuencia) y su amplitud (valor máximo) de forma que la representación gráfica en la pantalla sea ópticamente correcta, es decir, que el espacio disponible sea aprovechado de forma hábil.

```
10 REM      #MII/4#      -SEND-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT:PRINT " PROGRAMA PARA TRAZAR UNA SINUSOIDE."
40 PRINT:PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
50 LOCATE 8,8:PRINT "-----"
60 LOCATE 2,12:PRINT "FRECUENCIA (en función de ";CHR$(227
);" ) : ";LOCATE 34,12:INPUT P
70 LOCATE 2,16:INPUT "AMPLITUD      (DE 0 A 95)      : ";A
80 LOCATE 11,22:PRINT "PULSE UNA TECLA"
90 AS=INPUT$(1)
100 SCREEN 2:CLS
110 DRAW "BMO,95R255"
120 FOR X=0 TO 255
130 J=X/50:J=J*(1/P)
140 Y=SIN(J):Y=Y*A
150 Y=95-Y
160 PSET (X,Y),B
170 NEXT X
180 AS=INPUT$(1)
190 SCREEN 0:COLOR 15,1,1:CLS:LOCATE 11,12:PRINT "FIN DEL P
ROGRAMA":END
```

Lista de variables:

---

A = Amplitud  
I = Índice variable  
J = Valor X transformado  
P = Frecuencia  
X = Valor de la abcisa X  
Y = Valor de la ordenada Y

Descripción del programa

---

Línea 10-50 : Título

Línea 60-100 : Informaciones INPUT y modo gráfico

Línea 110 : Trazado de una línea central horizontal

Línea 120-170 : Determinación de los valores de la función y trazado después de transformar de acuerdo con P y A

Línea 180-190 : Impresión del texto y final del programa

## 9.6 Tabla matemática 1

### Presentación del problema

---

En el programa siguiente se imprimen los cuadrados y las raíces cuadradas de todos los números naturales del 1 al 100.

Este programa debe ser entendido casi como "programa de consulta".

### Análisis del problema

---

Siendo tan sencillo el problema, pensamos que resulta superfluo cualquier análisis.

```
10 REM 25-TABLA 1
20 CLS
30 PRINT " PROGRAMA PARA GENERAR LOS CUADRAD-"
40 PRINT " DOS Y LAS RAICES CUADRADAS DE TO-"
45 PRINT " DOS LOS NUMEROS NATURALES (1-100)."
```

Lista de variables:

---

I = Índice variable (números naturales 1-100)  
 J = Índice variable  
 Q = Número cuadrado  
 W = Raíz cuadrada

Descripción del programa

---

Línea 10-50 : Título

Línea 60 : Salto a la subrutina 2000 para la espera

Línea 70-110 : Cálculo e impresión de los valores en cuestión

Línea 100: Después de cada 15 líneas salto a la subrutina 2000 para la espera

Línea 120 : Final del programa principal

---

Línea 1000-1020 : Subrutina para la impresión de la cabecera de la tabla

---

Línea 2000-2050 : Subrutina para la espera

Línea 2020 : Impresión de un mensaje y espera.

Línea 2040-2050 : Borrado de la pantalla, impresión de una nueva cabecera mediante el salto al UP 1000 y después retorno al programa principal

9.7 Tabla matemática 2

---

Presentación del problema

---

Este programa sirve para ofrecer una tabla de valores de las funciones

- seno (SIN)  
 - coseno (COS)

Estas funciones angulares se representan en dependencia del parámetro (llamado pi en el programa). Por esto, en la tabla resultante aparecen también valores en grados angulares.

Análisis del problema

---

Tampoco aquí resulta necesario el análisis del problema, dado que no se presentan especiales dificultades.

```

10 REM      #MII/5#      -TABLA 1-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 LOCATE 0,5:PRINT "PROGRAMA PARA DETERMINAR SENOS Y COSE
NDS"
40 PRINT:PRINT:PRINT "      PROF. DR. W. VOSS, 1984"
50 LOCATE 8,10:PRINT "-----"
60 LOCATE 11,22:PRINT "PULSE UNA TECLA"
70 AS=INPUT$(1)
80 GOSUB 220
90 FOR J=0 TO 65
100 I=J/10:I=FIX(I*100+.5)/100
110 A=(360*I)/(2*3.141592654#):A=FIX(A+.5)
120 B=SIN(I):B=FIX(B*1000+.5)/1000
130 C=COS(I):C=FIX(C*1000+.5)/1000
140 LOCATE 4,2:PRINT I:LOCATE 13,2:PRINT A:LOCATE 22,2:PRIN
T B:LOCATE 31,2:PRINT C
150 Z=Z+2
160 IF Z>20 THEN Z=4
170 IF J=0 THEN 190
180 IF J/B=FIX(J/B) THEN LOCATE 11,22:PRINT "PULSE UNA TECLA"
A":AS=INPUT$(1):GOSUB 220
190 NEXT J
200 LOCATE 11,22:PRINT "PULSE UNA TECLA":AS=INPUT$(1):CLS
210 LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
220 CLS:Z=4:PRINT "      "; CHR$(227);"      "; "GRADOS";"
"; "SIN";"      "; "COS"
230 PRINT
240 FOR X=0 TO 39
250 PRINT "-";
260 NEXT X
270 RETURN

```

Lista de variables:

---

A = Angulo (en grados)  
B = Seno  
C = Coseno  
I = J/10 (=abcisa de la función angular)  
J = Índice variable (aquí 0-40)

Descripción del programa:

---

Línea 10-50 : Título

Línea 60-80 : Bucle de espera y salto a la subrutina de impresión de la cabecera.

Línea 90-190 : Cálculos e impresión de los resultados

Se determinan 66 valores (J=0 TO 65) siendo aplicadas las funciones angulares sobre I = J/10 (línea 102)

Línea 110 : Transformación en grados

Línea 120-130 : Determinación del SIN y del COS

Línea 140 : Impresión



Voss	Capítulo	9 : Matemáticas II	Página
	Punto	7 : Tabla 2	312
Colegio			

Línea 150-190 : Bucle contador del número de líneas que se imprimen. (Si se imprimen más de 8 líneas se borra la pantalla, se salta a la subrutina de impresión de la tabla y se siguen los cálculos).

Línea 200-210 : Final del programa principal

---

Línea 220-270 : Subrutina para la impresión de la cabecera de la tabla

---

Voss	Capítulo	9 : Matemáticas II	Página
	Punto	8 : Tabla 3	313
Colegio			

### 9.8 Tabla matemática 3

---

#### Presentación del problema

---

El programa siguiente sirve para imprimir los logaritmos neperianos  $\ln(X)$  dentro del ámbito de valores de  $X$  que va de 1 a 100.

#### Análisis del problema

---

Debido a la sencillez del problema también prescindimos aquí de cualquier análisis.

```

10 REM      #MII/6#      -TABLA 2-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 LOCATE 0,3:PRINT "--PROGRAMA PARA DETERMINAR LOS VALORES
DE"
40 LOCATE 7,7:PRINT "LOS LOGARITMOS NEPERIANOS."
50 PRINT:PRINT:PRINT:PRINT:PRINT "          PROF. DR. W. VOS
S, 1984"
60 LOCATE 8,14:PRINT "-----"
70 LOCATE 11,22:PRINT "PULSE UNA TECLA"
80 AS=INPUT$(1)
90 GOSUB 200
100 FOR I=1 TO 100
110 L=LOG(I)
120 L=FIX(L*10000+.5)/10000
130 LOCATE 9,2:PRINT I:LOCATE 23,2:PRINT L
140 Z=Z+2
150 IF Z>20 THEN Z=4
160 IF I/8=FIX(I/8) THEN LOCATE 11,22:PRINT "PULSE UNA TECL
A":AS=INPUT$(1):GOSUB 200
170 NEXT I
180 LOCATE 11,22:PRINT "PULSE UNA TECLA":AS=INPUT$(1):CLS
190 LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
200 CLS:Z=4:PRINT "          "; "NUMERO"; "          "; "LOGARITM
0"
210 PRINT
220 FOR X=0 TO 39
230 PRINT "--";
240 NEXT X
250 RETURN

```

Lista de variables :

---

I = Índice variable (aquí 1-100)

L = Logaritmo neperiano de I

Descripción del programa:

---

Línea 10-60 : Título

Línea 70-80 : Bucle de espera.

Línea 90 : Salto a la subrutina de impresión de la tabla.

Línea 100-170 : Cálculo de los logaritmos e impresión, después de cada 8 líneas de la tabla esperar

Línea 180-190 : Final del programa principal

---

Línea 200-250 : Subrutina para la impresión de la cabecera de la tabla

---

### 9.9 Elipse

#### Presentación del problema


El programa dibuja una elipse. Para ello deben introducirse los siguientes valores.

T: Columna del centro de la elipse (0-319)

Z: Línea del centro de la elipse (0-159)

A: Primer semieje de la elipse

B: Segundo semieje de la elipse



#### Análisis del problema

La ecuación de la elipse en la llamada forma central (centro de la elipse en el centro del eje de coordenadas):

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$$

Para poder dibujar una función como esta mediante un programa BASIC, debemos despejar y, resultando:

$$y_{1/2} = \pm \frac{b}{a} \sqrt{a^2 - x^2}$$

Si tenemos en cuenta que una elipse no sólo puede tener su punto central en la coordenada (0,0), sino también en una cualquiera (T,Z), resultan necesarias correcciones mediante traslación del mismo (ver la siguiente descripción de programa).

```

10 REM      #MII/7#      -ELIPSE-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " ALGORITMO PARA EL TRAZADO DE ELIPSES"
40 PRINT:PRINT "      PROF. DR. W. VOSS, 1984"
50 LOCATE 8,3:PRINT "-----"
60 PRINT:PRINT:PRINT " #COORDENADAS DEL CENTRO DE LA ELIPSE
E:"
70 PRINT:PRINT:INPUT "      -Coordenada x (columna) : ";I
80 PRINT:INPUT "      -Coordenada y (fila) : ";Z
90 PRINT:PRINT:PRINT " #LONGITUD DE LOS EJES DE LA ELIPSE
:"
100 PRINT:PRINT:INPUT "      - 1. Semieje      : ";A
110 PRINT:INPUT "      - 2. Semieje      : ";B
120 LOCATE 11,22:PRINT "PULSE UNA TECLA"
130 AS=INPUT$(1)
140 SCREEN 2:CLS
150 REM      EJES CARTESIANOS
160 DRAW "BMO,95C8R255BM127,OD191"
170 REM      ELIPSE
180 C=T-A:D=I+A
190 IF C<0 THEN C=0
200 IF D>255 THEN D=255
210 FOR X=C+1 TO D-1
220 DS=(B/A)*SQR(A*A-(X-T)^2)
230 Y=2+DS
240 IF Y<0 OR Y>191 THEN 260
250 PSET (X,Y),15
260 NEXT X
270 FOR X=D-1 TO C+1 STEP -1
280 DS=(B/A)*SQR(A*A-(X-T)^2)
290 Y=2-DS
300 IF Y<0 OR Y>191 THEN 320
310 PSET (X,Y),15
320 NEXT X
330 AS=INPUT$(1)
340 SCREEN 0:CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA"

```

Lista de variables:

---

A = Primer semieje de la elipse  
B = Segundo semieje de la elipse  
C = Extremo izquierdo de la elipse  
D = Extremo derecho de la elipse  
DS = Valor auxiliar tomado de la ecuación de la elipse  
I = Índice variable  
T = Columna del punto central de la elipse  
X = Abscisa  
Y = Ordenada  
Z = Línea del punto central de la elipse

Descripción del programa

---

Línea 10-50 : Título

Línea 60-130 : Fijación de las coordenadas del punto central y de los semiejes

Línea 140 : Modo gráfico y borrado de pantalla

Línea 150-160 : Trazado de los ejes de coordenadas

Línea 170-260 : Trazado de la mitad inferior de la elipse

Línea 270-320 : Trazado de la mitad superior de la elipse

Línea 330-340 : Final del programa

### 9.10 Distribución normal de Gauss

---

Presentación del problema:

---

Este programa representa gráficamente la distribución de Gauss.

Tiene la función de densidad

$$f(x) = \frac{1}{SS \cdot \sqrt{2\pi}} \cdot e^{-\frac{1}{2} \left( \frac{x - MM}{SS} \right)^2}$$


MM y SS son los parámetros de posición caracterizadores (ver el esquema siguiente):

MM designa el valor medio de la distribución  
 SS designa la dispersión de la distribución  
 (desviación estandar = distancia entre la línea central y cualquiera de los dos puntos de inflexión de la distribución).

En este programa seleccionado los siguientes valores :

$$MM = 160$$

$$SS = 40$$



Análisis del problema

---

Nuevamente prescindible.

```

10 REM      #MII/B#      -GAUSS-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " ALGORITMO PARA REPRESENTAR GRAFICAMENTE"
40 PRINT " LA DISTRIBUCION NORMAL SEGUN EL METODO "
50 PRINT " DE GAUSS."
60 PRINT:PRINT "      PROF. DR. W. VOSS, 1984"
70 LOCATE 8,7:PRINT "-----"
80 LOCATE 0,10:PRINT "Este programa traza la distribución
nor-"
90 PRINT "mal con el valor medio 127 y la desvia-"
100 PRINT "ción estándar 40."
110 PRINT:PRINT "Estos parámetros pueden modificarse para"
120 PRINT "cada caso (línea 160)."
130 LOCATE 11,22:PRINT "PULSE UNA TECLA"
140 AS=INPUT$(1)
150 SCREEN 2:CLS
160 MM=127:SS=40
170 DRAW "BMO,191C15R255BM127,OD191"
180 REM      DISTRIBUCION NORMAL
190 A=1/(SS*SQR(2*3.141592654#))
200 FOR X=0 TO 255
210 B=((X-MM)/SS)^2
220 B=.5*B*(-1)
230 Y=A*EXP(B)
240 Y=(600000/SS)*Y
250 Y=191-Y
260 IF Y<0 OR Y>191 THEN 280
270 PSET (X,Y),B
280 NEXT X
290 AS=INPUT$(1)
300 SCREEN 0:CLS
310 LOCATE 11,12:PRINT "FIN DEL PROGRAMA"

```

Lista de variables:

---

A = Primer término de la ecuación funcional  
B = Exponente en la ecuación funcional  
I = Índice variable  
MN = Valor medio de la distribución normal  
SS = Desviación estándar de la distribución normal  
X = Abscisa  
Y = Ordenada

Descripción del programa

---

Línea 10-120 : Título y aclaraciones

Línea 130-140 : Espera

Línea 150-160 : Definición de los parámetros de la función de distribución y modo gráfico

Línea 170 : Trazado de unos ejes de coordenadas

Línea 180-280 : Cálculo de los valores de ordenada de la distribución normal, normalización (línea 240) y trazado de la curva

Línea 290-310 : Final del programa

### 9.11 Regresión


#### Presentación del problema

El cálculo de regresiones constituye una cuestión importante dentro del campo de la estadística matemática. Se trata de describir una distribución estadística bivariable mediante una función matemática.

Una distribución estadística bivariable se produce, por ejemplo, cuando estudiamos simultáneamente dos características de una población, es decir, si consultamos a una serie de personas acerca de su tamaño y de su peso al mismo tiempo.

El objetivo de una investigación como ésta, es descubrir si existe alguna relación entre ambas variables.

Una distribución bivariable puede representarse gráficamente en un sistema de coordenadas bidimensional por medio de una "nube de puntos", como puede apreciarse en la página siguiente.



Para describir el tipo de relación existente entre ambas variables, o para determinar la tendencia lineal en una serie de tiempos, que desde el punto de vista matemático es el mismo problema, se intenta encajar de la mejor forma posible una función matemática lineal entre los puntos.

Del mismo modo, existen directrices de cálculo no lineales, pero que aquí no serán tratadas.

Análisis del problema

La función lineal o no lineal que tratamos de dibujar se determina, por regla general, con el método de los mínimos cuadrados.

Este método nos da unas directrices de cálculo para determinar los parámetros de la función. En el caso de una recta, se trata de la ordenada en el origen A y de la pendiente B.

El método de los mínimos cuadrados nos exige determinar los parámetros de forma que la suma de los cuadrados de las desviaciones entre los valores de ordenada y (coordenadas verticales) observados y los valores de ordenada de la recta y deben ser minimizados, es decir:

$$\sum (y_i - \hat{y}_i)^2 = \min.$$

Los valores de y se calculan del modo siguiente con la ecuación de la recta:

$$\hat{y}_i = a + b \cdot x_i$$

Con la regla de minimización antes citada resulta:

$$\sum (y_i - a - b \cdot x_i)^2 = \min.$$

El mínimo de esta función se halla realizando una derivada parcial según a y b e igualando estas dos derivadas parciales a cero.

De esta forma se obtienen dos ecuaciones con dos incógnitas (a y b), en las que éstas se habrán de despejar. Se obtienen así las dos ecuaciones:

$$b = \frac{n \cdot \sum x_i y_i - \sum x_i \sum y_i}{n \cdot \sum x_i^2 - (\sum x_i)^2}$$

$$a = \frac{\sum y_i}{n} - b \cdot \frac{\sum x_i}{n}$$

El programa BASIC a desarrollar sirve para calcular estos parámetros según el método de los mínimos cuadrados para cualquier par de valores x, y que se introduzca y para a continuación imprimir los resultados.



```

10 REM      #MII/9#      -REGRESION-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA DETERMINAR LOS PARAMETROS"
40 PRINT " DE UNA FUNCION LINEAL DE REGRESION O DE"
50 PRINT " UNA RECTA DE TENDENCIA."
60 PRINT:PRINT "          PROF. DR. W. VOSS, 1984"
70 LOCATE 8,7:PRINT "-----"
80 LOCATE 0,11:PRINT " Este programa requiere pares de val
ores"
90 PRINT " como informaciones de entrada (P.E. va-"
100 PRINT " lores X e Y 5 valores de tiempo y carac"
110 PRINT " terística, etc... "
120 LOCATE 11,22:PRINT "PULSE UNA TECLA"
130 AS=INPUT$(1)
140 CLS
150 LOCATE 4,12:INPUT "CUANTOS PARES DE VALORES : ";N
160 DIM X(N),Y(N)
170 CLS
180 LOCATE 5,2:PRINT "INTRODUZCA LOS PARES DE VALORES"
190 FOR I=1 TO N
200 LOCATE 13,7:PRINT "PAR NUMERO ";I
210 LOCATE 10,12:PRINT "VALOR DE X.....:      ":LOCATE
30,12:INPUT Z1
220 LOCATE 10,16:PRINT "VALOR DE Y.....:      ":LOCATE
30,16:INPUT Z2
230 X(I)=Z1:Y(I)=Z2
240 NEXT I
250 GOSUB 560
260 FOR I=1 TO N
270 LOCATE 5,P:PRINT I:LOCATE 17,P:PRINT X(I):LOCATE 31,P:P
RINT Y(I)
280 IF P=16 THEN LOCATE 11,22:PRINT "PULSE UNA TECLA":AS=IN
PUT$(1):GOSUB 560
290 P=P+2
300 NEXT I
310 LOCATE 11,22:PRINT "FIN DEL CONTROL"
320 AS=INPUT$(1):CLS
330 FOR I=1 TO N
340 S1=S1+X(I)*Y(I)
350 S2=S2+X(I):S3=S3+Y(I)

```

```

360 S4=S4+X(I)*X(I)
370 NEXT I
380 B=(N*S1-S2*S3)/(N*S4-S2*S2)
390 A=S3/N-B*S2/N
400 B=FIX(B*10000+.5)/10000
410 A=FIX(A*10000+.5)/10000
420 LOCATE 15,0:PRINT "RESULTADO"
430 LOCATE 3,8:PRINT "ORDENADA EN EL ORIGEN A = ";A
440 LOCATE 3,12:PRINT "PENDIENTE          B = ";B
450 LOCATE 11,22:PRINT "PULSE UNA TECLA"
460 AS=INPUT$(1)
470 CLS
480 LOCATE 0,5:PRINT " CON ESTOS PARAMETROS PUEDE LLAMARSE
AL"
490 LOCATE 0,8:PRINT " PROGRAMA PARA EL TRAZADO DE UNA RECT
A,"
500 LOCATE 0,11:PRINT " SI DESEA OBTENER TAMBIEN UNA REPRES
EN-"
510 LOCATE 0,14:PRINT " TACION GRAFICA DEL RESULTADO."
520 LOCATE 11,22:PRINT "PULSE UNA TECLA"
530 AS=INPUT$(1)
540 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
550 REM          SUBROUTINA
560 CLS:LOCATE 17,0:PRINT "CONTROL"
570 P=6
580 LOCATE 0,3:PRINT "          No.          Valor X          Valor Y"
590 FOR I=0 TO 39:LOCATE I,4:PRINT "--";:NEXT I
600 RETURN

```

Lista de variables:

---

A = Ordenada en el origen de la recta  
 B = Pendiente de la recta  
 I = Índice variable  
 N = Número de pares de valores  
 S1 =  
 S2 = } Sumas auxiliares para calcular  
 S3 = } B y A  
 S4 = }  
 X = Abcisa  
 Y = Ordenada  
 Z1 = Campos auxiliares para  
 Z2 = la introducción de datos

Descripción del programa:

---

Línea 10-140 : Título, aclaraciones y espera  
 Línea 150-160 : Indicación del número de pares de valores de la característica y dimensionado  
 Línea 170-240 : Introducción de los valores de la característica mediante INPUT  
 Línea 250 : Impresión de la tabla  
 Línea 260-300 : Impresión de control de los pares de valores y pausa después de 16 pares  
 Línea 310-320 : Espera  
 Línea 330-370 : Formación de las cuatro sumas necesarias en las fórmulas para A y para B  
 Línea 380-410 : Cálculo de los parámetros de la recta A y B  
 Línea 420-470 : Impresión de los resultados  
 Línea 480-540 : Indicación para bifurcar y final del programa  


---

 Línea 550-600 : Subrutina de imprimir la tabla  


---

9.12 Permutaciones

---

Presentación del problema

---

El cálculo de permutaciones responde a la cuestión de cuántas formas distintas pueden ordenarse diferentes elementos.

Análisis del problema

---

Si por ejemplo tenemos tres elementos distintos a, b, y c, pueden distinguirse las siguientes posibilidades de ordenación:

abc, acb, bac, bca, cab, cba

Es decir, que con tres elementos se obtienen seis posibilidades distintas de ordenación.

Generalmente se cumple la siguiente regla:

Para el primero de los tres elementos existen tres posibilidades : puede estar en primer lugar, en segundo lugar o en tercer lugar;

entonces, para el segundo elemento existen solamente dos posibilidades cuando se ha decidido ya dónde debe colocarse el primer elemento.

Esto significa, no obstante, que los dos primeros juntos tienen  $3*2=6$  posibilidades.

Para el tercer elemento ya sólo queda una posibilidad y para los tres elementos  $3*2*1 = 6$  posibilidades distintas de ordenación.

De la misma forma, cuando se tienen n elementos distintos, resulta:

$$\text{Número de posibilidades de ordenación} = n*(n-1)*(n-2)*(n-3)*...*3*2*1$$

Esta cadena de productos se resume mediante n! (n factorial), y se dice que n elementos pueden ordenarse de n factorial formas distintas, o :

que la permutación de n elementos distintos es n-factorial (n!).

Debe tenerse en cuenta que para valores de n elevados la expresión n! crece muy rápidamente. Es más, crece tan rápido, que el ordenador ya no puede representar el resultado de 34! a partir de n = 34.

```

10 REM #MII10# -PERMUTACIONES-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA DETERMINAR EL NUMERO DE"
40 PRINT " PERMUTACIONES DE N ELEMENTOS TOMADOS DE"
50 PRINT " N EN N."
60 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
70 LOCATE 8,7:PRINT "-----"
80 LOCATE 0,10:PRINT " Este programa calcula el factorial
(n!)"
90 PRINT " del valor indicado por el usuario."
100 LOCATE 6,15:INPUT "NUMERO DE ELEMENTOS n : ";N
110 P=N
120 IF N>48 OR N<0 THEN 100
130 IF N=0 OR N=1 THEN P=1:GOTO 170
140 FOR I=N-1 TO 1 STEP -1
150 P=P*I
160 NEXT I
170 LOCATE 8,18:PRINT "EL FACTORIAL DE ";N;" ES :":LOCATE 9
,20:PRINT P
180 LOCATE 11,22:PRINT "OTRO CALCULO (S/N)"
190 AS=INKEY$
200 IF AS="S" OR AS="s" THEN CLS:GOTO 10
210 IF AS="N" OR AS="n" THEN 220 ELSE 190
220 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA"

```

Lista de variables:

---

A\$ = Variable de cadena para las respuestas  
I = Contador  
N = Número de elementos  
P = Producto intermedio y final

Descripción del programa:

---

Línea 10-90 : Título y aclaraciones

Línea 100-110 : Demanda de la información INPUT (número de elementos a permutar N) el campo del producto P toma el valor del campo, como primer factor de la cadena de productos a realizar

Línea 120-130 : Comprobación de que no se exceden los límites de cálculo del ordenador.

Línea 140-160 : Formación de la cadena de productos en el campo P

Línea 170 : Impresión del resultado

Línea 180-220 : Final del programa, a menos que se desee continuar calculando (entonces seguir en la línea 10 una vez borrada la pantalla)

### 9.13 Combinaciones

#### Presentación del problema

El problema es similar al del programa anterior, calculándose la cantidad de formas distintas en que pueden seleccionarse  $k$  elementos de entre  $n$  elementos distintos.

La fórmula de cálculo es la siguiente:

$$\text{Número} = \frac{n!}{k!(n-k)!}$$

A esta magnitud se le llama coeficiente binomial y se abrevia de la forma siguiente:

$$\binom{n}{k}$$

Por otra parte, éste equivale a los factores que resultan en el desarrollo de los llamados binomios:

$$\begin{aligned} (a+b)^2 &= a^2 + 2ab + b^2 \\ &= 1a^2b^0 + 2a^1b^1 + 1a^0b^2 \\ &= \binom{2}{0}a^2b^0 + \binom{2}{1}a^1b^1 + \binom{2}{2}a^0b^2 \end{aligned}$$

0:

$$\begin{aligned} (a+b)^3 &= a^3 + 3a^2b + 3ab^2 + b^3 \\ &= 1a^3b^0 + 3a^2b^1 + 3a^1b^2 + 1a^0b^3 \\ &= \binom{3}{0}a^3b^0 + \binom{3}{1}a^2b^1 + \binom{3}{2}a^1b^2 + \binom{3}{3}a^0b^3 \end{aligned}$$

Como puede observarse, todo binomio se descompone en sumandos. Siempre hay un sumando más que el indicado por el exponente del binomio (el exponente en el último ejemplo era 3; el número de sumandos 4).

Cada sumando consta de 3 factores :

- una constante (el coeficiente binomial que aquí nos ocupa)
- factor "a elevado..." con un exponente, que partiendo del exponente del binomio (aquí 3) baja hasta 0;
- factor b "b elevado a..." con un exponente, que partiendo de 0 sube hasta el exponente del binomio (aquí 3).

En el coeficiente binomial que nos ocupa, el mayor valor es el exponente del binomio (aquí 3); el valor menor es un número entero que va ascendiendo de derecha a izquierda desde 0 hasta el exponente del binomio.

El programa presentado determina cualquier coeficiente binomial, permitiéndonos calcular fácilmente también binomios más complejos (p.e.  $(a+b)^{27}$ ).

Pero este programa puede utilizarse también para cálculos, como se demostrará en los próximos programas.

Finalmente, un programa de este tipo, permite responder p.e. a la siguiente cuestión:

" 8 corredores están a punto de comenzar una carrera de 100 m lisos. Cuántas posibilidades diferentes existen de reparto de las tres primeras posiciones?"

Respuesta:

$$\binom{8}{3} = \frac{8!}{3!(8-3)!} = \frac{8!}{3!5!} =$$

$$= \frac{8 \cdot 7 \cdot 6}{3 \cdot 2 \cdot 1} = 56$$

```

10 REM #MII11# -COMBINACIONES-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA DETERMINAR EL NUMERO DE"
40 PRINT " COMBINACIONES DE N ELEMENTOS TOMADOS DE"
50 PRINT " K EN K."
60 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
70 LOCATE 8,7:PRINT "-----"
80 LOCATE 5,12:INPUT "NUMERO DE ELEMENTOS N : ";N
90 LOCATE 5,17:INPUT "NUMERO SELECCIONADO K : ";K
100 Y=N-K
110 IF K=0 OR K=N THEN BK=1:GOTO 210
120 IF K>N/2 THEN H=Y:Y=K:K=H
130 GOSUB 300
140 PX=A:A=N
150 IF PX=1 THEN PN=N:GOTO 200
160 FOR I=N-1 TO N-K+1 STEP -1
170 A=A*I
180 NEXT I
190 PN=A
200 BK=PN/PX
210 CLS
220 PRINT " #S O L U C I O N#"
230 PRINT " -----"
240 LOCATE 3,5:PRINT "EL NUMERO DE COMBINACIONES DE ";N
250 LOCATE 2,8:PRINT "ELEMENTOS TOMADOS DE ";K;" EN ";K;" E
S:"
260 LOCATE 14,12:PRINT BK
270 LOCATE 11,20:PRINT "PULSE UNA TECLA"
280 AS=INPUT$(1)
290 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
300 REM SUBROUTINA
310 A=K
320 IF K=0 OR K=1 THEN A=1:GOTO
330 FOR I=K-1 TO 1 STEP -1
340 A=A*I
350 NEXT I
360 RETURN

```

Lista de variables:

---

A = Factorial (producto intermedio y final)  
 BK = Coeficiente binomial  
 H = Valor auxiliar en el cambio de K por Y  
 I = Índice variable (factor en el cálculo del factorial)  
 K = Número de elementos a elegir  
 N = Número de elementos  
 PN = Primer factorial en el coeficiente binomial  
 PX = Segundo factorial en el coeficiente binomial  
 Y = Y-K

Descripción del programa:

Línea 10-70 : Título y aclaraciones

Línea 80-90 : Introducciones (número de elementos N y número de elementos a elegir K)

Línea 100 : Definición del valor auxiliar Y

Línea 110 : Cálculo simplificado para K=0 ó K=N, porque por definición se cumple que

$$\binom{N}{0} = \binom{N}{N} = 1$$

Línea 120 : Intercambio de K y (N-K) (=Y) para K>N/2. Esto es posible porque se cumple que:

$$\binom{N}{K} = \frac{N!}{K!(N-K)!} = \binom{N}{N-K}$$

De este modo se simplifican los cálculos siguientes.

Línea 130 : Cálculo de K! en la subrutina 300

Línea 140-190 : Cálculo de otro factorial, que queda después de reducir (ver p.e. problema de los corredores de antes).

Línea 200-290 : Cálculo e impresión del coeficiente binomial y final del programa principal.

---

Línea 300-360 : Cálculo de K! en una subrutina

---

### 9.14 Lotería

#### Presentación del problema

Vamos a confeccionar un programa que calcula las posibilidades de pre lo en la lotería (6 entre 49).

#### Análisis del problema

El programa calcula la probabilidad de conseguir 3, 4, 5 ó 6 aciertos en la lotería (6 de 49).

Las correspondientes probabilidades se calculan con la siguiente fórmula :

$$P(X) = \frac{\binom{6}{X} \binom{43}{6-X}}{\binom{49}{6}} = \frac{B1*B2}{B3}$$

X es el número de aciertos que debe indicar el usuario.

En suma, aquí tenemos que operar con tres coeficientes binomiales (B1, B2 y B3), que ya conocemos del capítulo anterior.

Por este motivo nos ahorramos ahora cualquier otra explicación.

```

10 REM      #M1112#      -LOTERIA-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT "PROGRAMA PARA DETERMINAR LA PROBABILI-"
40 PRINT "DAD DE ACERTAR 3,4,5 ó 6 NUMEROS EN LA"
50 PRINT "LOTERIA (6 DE 49)."

```



```
390 P2=P*B2=P1/P2
400 REM B3
410 FOR I=1 TO 5
420 P1=P1*(49-I)
430 P2=P2*(6-I)
440 NEXT I
450 B3=P1/P2:W=(B1*B2)/B3
460 CLS
470 IF X=2 THEN X=4
480 LOCATE 11,0:PRINT "R E S U L T A D O"
490 LOCATE 11,1:PRINT "-----"
500 LOCATE 0,7:PRINT "LA PROBABILIDAD DE ACERTAR ";X;" NUME
ROS"
510 LOCATE 0,11:PRINT "EN LA LOTERIA ES :"
520 LOCATE 9,15:PRINT W
530 LOCATE 11,22:PRINT "OTRO CALCULO (S/N)"
540 AS=INKEY$
550 IF AS="S" OR AS="s" THEN 10
560 IF AS="N" OR AS="n" THEN 570 ELSE 540
570 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
```

Lista de variables:

-----

A\$ = Variable de cadena para introducir respuestas  
B1 = Primer término en la fórmula de cálculo  
B2 = Segundo término en la fórmula de cálculo  
B3 = Tercer término en la fórmula de cálculo  
P = Producto en el cálculo de factoriales  
P1 = Primer factorial del coeficiente binomial  
P2 = Segundo factorial del coeficiente binomial  
W = Probabilidad  
X = Número de aciertos  
Y = 6-X

Descripción del programa:

---

- Línea 10-100 : Título y aclaraciones
- Línea 110-120 : Bucle de espera y borrado pantalla
- Línea 130 : Introducción del número de aciertos
- Línea 140 : Para  $X=6$  se cumple que  $B1=B2=1$  (ver programa Z10); seguir en la línea 400
- Línea 150-270 : Determinación de  $B1$  de acuerdo con las reflexiones hechas ya con referencia al programa Z10
- Línea 280-390 : Del mismo modo, determinación del segundo coeficiente binomial  $B2$
- Línea 400-440 : Lo mismo con  $B3$
- Línea 450 : Cálculo de la probabilidad  $W$
- Línea 460-520 : Impresión del resultado
- Línea 530-570 : Final del programa principal, a menos que se desee un nuevo cálculo

9.15 Distribución binomial

---

Presentación del problema

---

La distribución binomial es una distribución de probabilidades. Las distribuciones de probabilidades, a su vez, sirven para calcular -de la forma más sencilla posible- probabilidades, incluso en problemas complejos.

La distribución binomial es también muy importante porque sirve de "base" para toda una serie de procedimientos de cálculo de probabilidades.

La distribución binomial responde a la siguiente cuestión:

¿Cuál es la probabilidad de que en  $n$  intentos independientes entre sí se produzca  $x$  veces el suceso favorable, si en cada intento sólo son posibles dos resultados, y si la probabilidad de que se produzca este suceso favorable en un intento es del  $P\%$ .

Sirva el siguiente ejemplo para ilustrar esta cuestión :

¿Cuál es la probabilidad de que echando 10 veces una moneda nos salga p.e. 5 veces cara?

Este es un caso de aplicación de la distribución binomial, pues en cada intento son posibles dos sucesos ("cara" y "cruz") y los intentos (las tiradas) son independientes entre sí.

Otro problema de este tipo sería :

¿Cuál es la probabilidad de que de cada 5 nacimientos 2 sean niña?

#### Análisis del problema

Para analizar el problema en cuestión nos ocuparemos en primer lugar del ejemplo citado en último lugar :

Si de cada cinco nacimientos dos deben ser niñas, la siguiente "combinación de nacimientos" podría responder al problema :

NA NA NO NO NO (NA = niña, NO = niño)

También sería válida la siguiente combinación :

NO NO NO NA NA

y también muchas otras.

El número total de combinaciones "favorables" puede hallarse con ayuda del cálculo de permutaciones, tal como a comentamos.

Sin ser nuestra intención el demostrar este extremo, calculamos que el número de posibilidades de combinación de 2 niñas y 3 niños es:

$$\begin{aligned} \text{Número} &= \frac{5!}{2! * 3!} = \frac{5*4*3*2*1}{2*1 * 3*2*1} \\ &= \frac{5*4}{2*1} = 10 \end{aligned}$$

Generalmente, el número de disposiciones distintas de n elementos, de los que x son iguales (en el sentido de "favorables") y otros n-x son también iguales (en el sentido de "desfavorables"), es :

$$\text{Número} = \frac{n!}{x_1! * (n-x_1)!}$$

Voss	Capítulo	9 : Matemáticas II	Página
	Punto	15 : Distribución	350
Colegio		binomial	

Para seguir con el análisis del problema nos basaremos en que la probabilidad de que nazca una niña es  $P = 0.5$  - esta es la probabilidad de que se produzca un suceso favorable en un intento.

Contemplemos el resultado siguiente :

NA NA NO NO NO

¿Cuál es la probabilidad de que se produzca un resultado así en cinco intentos?

Contemplemos en primer lugar el nacimiento de una niña : La probabilidad de que se produzca es lógicamente 0.5.

La probabilidad de que los dos primeros nacimientos juntos sean niña es entonces  $0.5 * 0.5$  (en general  $P * P$ ).

La probabilidad de que nazca un niño es  $1 - 0.5$  (en general  $1 - P$ ). De esta manera, siguiendo el mismo procedimiento de arriba, la probabilidad de que el tercero, el cuarto y el quinto nacimiento sean niño es :

Voss	Capítulo	9 : Matemáticas II	Página
	Punto	15 : Distribución	351
Colegio		binomial	

$$(1 - 0.5) * (1 - 0.5) * (1 - 0.5)$$

o, en general :

$$(1 - P) * (1 - P) * (1 - P)$$

Para la combinación NA NA NO NO NO obtenemos en total la probabilidad siguiente (escrita de forma general) :

$$\begin{aligned}
 & P * P * P * (1 - P) * (1 - P) * (1 - P) \\
 &= P^3 * (1 - P)^3 \\
 &= P^{x_i} * (1 - P)^{n-x_i}
 \end{aligned}$$

En nuestro ejemplo obtenemos :

Probabilidad para NA NA NO NO NO =

$$0.5^2 * (1 - 0.5)^{5-2} = 0.5^5 = 1/32$$

Pero, en el ejemplo que nos ocupa hay 10 posibles combinaciones distintas NA,NO .

Así, a partir de 10 posibles combinaciones, con una probabilidad de 1/32 cada una, se obtiene :

Probabilidad de que de 5 nacimientos 2 sean niñas =  $10 * 1/32 = 10/32 = 31.25 \%$

Este ejemplo nos permite inducir la siguiente fórmula general :

La probabilidad de que en n intentos se produzca x veces el suceso favorable, cuando la probabilidad individual de producirse un suceso es P, se calcula del modo siguiente :

$$W = \frac{n!}{x! * (n-x)!} * P^x * (1-P)^{n-x}$$

Esta fórmula parece complicada, pero su programación resulta relativamente sencilla.

```

10 REM #MII13# -BINOMIAL-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT "PROGRAMA PARA CALCULAR LAS PROBABILIDA-"
40 PRINT "DES CON LA DISTRIBUCION BINOMIAL."
50 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
60 LOCATE 8,5:PRINT "-----"
70 LOCATE 0,8:PRINT "CON ESTE PROGRAMA, ES DECIR, CON LA D
IS-"
80 LOCATE 0,10:PRINT "TRIBUCION BINOMIAL, PUEDE RESPONDER
SE A"
90 LOCATE 0,12:PRINT "LA SIGUIENTE CUESTIION:"
100 LOCATE 0,15:PRINT "Cuál es la probabilidad de que se pr
oduz"
110 LOCATE 0,16:PRINT "ca x veces un suceso favorable, en N
in-"
120 LOCATE 0,17:PRINT "tentos, si la probabilidad en un int
entoes P?"
130 LOCATE 11,22:PRINT "PULSE UNA TECLA"
140 AS=INPUT$(1)
150 CLS
160 PRINT " DATOS DEL PROBLEMA"
170 PRINT "-----"
180 LOCATE 0,7:INPUT "NUMERO DE INTIENTOS..... : ";N
190 LOCATE 0,12:INPUT "NUMERO DE SUCEOS FAVORABLES... : ";
X
200 LOCATE 0,17:PRINT "PROBABILIDAD EN UN INTENTO..... : "
210 LOCATE 3,19:PRINT "(como número decimal)":LOCATE 34,17:
INPUT P
220 Z=X
230 REM COEFICIENTE BINOMIAL
240 Y=N-X:Q=1-P
250 IF X=0 OR X=N THEN BK=1:GOTO 350
260 IF X>N/2 THEN H=Y:Y=X:X=H:G=Q:Q=P:P=G:F=X
270 GOSUB 470
280 PX=A:A=N
290 IF PX=1 THEN PN=N:GOTO 340
300 FOR I=N-1 TO N-X+1 STEP -1
310 A=A*I
320 NEXT I
330 PN=A

```

```

340 BK=PN/PX
350 W=BK*P^X*Q^Y
360 CLS
370 LOCATE 11,0:PRINT "R E S U L T A D O"
380 LOCATE 11,1:PRINT "-----"
390 LOCATE 1,7:PRINT "LA PROBABILIDAD DE PRODUCIRSE EN ";N
400 LOCATE 0,11:PRINT "INTENTOS ";Z;" VECES EL SUCESO FAVOR
ABLE"
410 LOCATE 9,15:PRINT "ES : ";W
420 LOCATE 11,22:PRINT "OTRO CALCULO (S/N)"
430 AS=INKEY$
440 IF AS="S" OR AS="s" THEN 150
450 IF AS="N" OR AS="n" THEN 460 ELSE 430
460 CLS:LOCATE 11,12:PRINT "FIN DEL PROGRAMA":END
470 REM          SUBROUTINA
480 A=X
490 IF X=0 OR X=1 THEN A=1:GOTO 530
500 FOR I=X-1 TO 1 STEP -1
510 A=A*I
520 NEXT I
530 RETURN

```

Lista de variables:

---

A = Producto en el cálculo de factoriales  
A\$ = Variable de cadena para la introducción de respuestas  
BK = Coeficiente binomial  
F = Valor auxiliar (F=X)  
G = Valor auxiliar en el cambio de P y Q  
H = Valor auxiliar en el cambio de X e Y  
I = Índice variable  
N = Número de intentos  
P = Probabilidad de producirse el suceso favorable en cada intento  
  
PN = Primer término del coeficiente binomial  
PX = Segundo término del coeficiente binomial  
Q = 1-P  
W = Probabilidad  
X = Número de sucesos favorables  
Y = N-X  
Z = X

Descripción del programa

---

- Línea 10-120 : Título y aclaraciones
- Línea 130-210 : Espera e introducción de las  
Informaciones INPUT (número de  
Intentos, probabilidad).
- Línea 220-340 : Determinación del coeficiente bino-  
mial :
- 240 : Campos auxiliares;
- 250 : Si  $X=0$  ó  $X=1$  entonces el coe-  
ficiente se define como 1;  
seguir en 350;
- 260 : Si  $X > N/2$  entonces, para apro-  
vechar las posibilidades de  
reducción, cambiar  $X$  por  $N-X$   
y  $P$  por  $1-P$
- 280 : Definición del denominador  
( $=PX$ ) del coeficiente y re-  
definición de  $A$ , en caso de  
utilizar nuevamente var.  $A$   
(300);
- 290 : Si el denominador del coefi-  
ciente es = 1, entonces el  
numerador queda definido como  
 $N$ ; seguir en línea 340.
- 300-320 : Determinación del numera-  
dor del coeficiente  $BK$ ;
- 340 : Cálculo de  $BK$ .
- Línea 350 : Cálculo de la probabilidad buscada.

Línea 360-410 : Impresión del resultado del cálculo.

Línea 420-460 : Final del programa, a menos que no  
se quiera un nuevo cálculo (entonces  
volver a la línea 190 tras borrar la  
pantalla).

---

Línea 570-530 : Subrutina para calcular un factorial

---

### 9.16 Resolución de sistemas de ecuaciones

---

#### Presentación del programa

---

Cuando nos encontramos con problemas cuya resolución requiere el cálculo de un sistema de  $N$  ecuaciones con  $N$  incógnitas cada una, es conveniente tener un programa que nos permita librarnos de los tediosos y repetitivos cálculos a que nos llevaría un sistema de gran envergadura.

El algoritmo del programa se basa en aplicar el método de Gauss-Jordan a la matriz ampliada del sistema. Este método viene explicado en cualquier tratado de álgebra lineal existente en el mercado por lo que nos extenderemos en su explicación.

El programa pregunta en primer lugar la dimensión de la matriz ampliada del sistema que tenemos que resolver. A continuación se nos van pidiendo los coeficientes de cada ecuación (si falta alguna incógnita deberemos introducir un cero).

Una vez introducidos todos los coeficientes se pasa al cálculo de las soluciones.

Este programa no tiene en cuenta los posibles casos de incompatibilidad; instamos al lector a que modifique el programa teniendo en cuenta estos posibles casos.

```

10 COLOR 15,1,1:WIDTH 40:CLS:KEY OFF
20 FOR I=1 TO N
30 LOCATE 0,7:PRINT "ESTE PROGRAMA RESUELVE SISTEMAS DE EC
UA-"
40 PRINT "CIONES MEDIANTE LA APLICACION DEL METODO"
50 PRINT "DE GAUSS-JORDAN."
60 LOCATE 7,22:PRINT "Pulse una tecla por favor"
70 AS=INPUT$(1)
80 CLS
90 LOCATE 1,9:PRINT "DIMENSION DE LA MATRIZ DE COEFICIENTE
S":PRINT:PRINT "          DEL SISTEMA":LOCATE 25,11:INPUT
N
100 DIM A(N,N)
110 CLS
120 FOR I=0 TO N-1
130 PRINT "  COEFICIENTES DE LA ";I+1;CHR$(166);" ECUACIO
N":PRINT:PRINT:PRINT
140 FOR J=0 TO N-1
150 PRINT "    A("I+1","J+1")  =  ";:INPUT A(I,J):PRI
NT
160 NEXT J
170 INPUT "      Término cte. b =  ";A(I,N):CLS
180 NEXT I
190 COLOR 1,15,15:CLS:LOCATE 10,10:PRINT "C A L C U L A N D
O"
200 FOR I=0 TO N-2
210 FOR J=I+1 TO N
220 IN=A(J,I)/A(I,I)
230 FOR X=0 TO N
240 A(J,X)=A(J,X)-IN*A(I,X)
250 NEXT X
260 NEXT J
270 NEXT I
280 FOR I=0 TO N-1
290 IF A(I,I)=1 THEN 340
300 IN=1/A(I,I)

```



```

310 FOR X=0 TO N
320 A(I,X)=IN*A(I,X)
330 NEXT X
340 NEXT I
350 FOR I=N-1 TO 0 STEP -1
360 SU=0
370 FOR X=I+1 TO N-1
380 SU=SU+A(X,X)*A(I,X)
390 NEXT X
400 A(I,I)=A(I,N)-SU
410 NEXT I
420 COLOR 15,1,1:CLS
430 LOCATE 0,1:PRINT "      - LAS SOLUCIONES SON: ":PRINT:
PRINT:PRINT
440 FOR I=1 TO N
450 IF A(I-1,I-1)-FIX(A(I-1,I-1))/10000 > .99999 THEN W=A(I-1,I-1)+
1E-05:W=FIX((W*10000+.5)/10000):PRINT "      X";I;"=";"
";W:PRINT:GOTO 470
460 PRINT "      X";I;"=";" ";A(I-1,I-1):PRINT
470 NEXT I
480 IF N<9 THEN LOCATE 11,21:PRINT "FIN DEL PROGRAMA":END E
LSE PRINT "      FIN DEL PROGRAMA"
490 END

```

Descripción del programa:

---

- Línea 10-50 : Título y descripción del programa
- Línea 60-70 : Bucle de espera.
- Línea 80 : Borrado de la pantalla
- Línea 90 : Petición de la dimensión de la matriz ampliada del sistema.
- Línea 100 : Dimensionado de la matriz A con las dimensiones del sistema.
- Línea 110-180 : Petición de los coeficientes y término constante de cada ecuación
- Línea 190 : Impresión del mensaje "CALCULANDO"
- Línea 200-410 : Cálculo de la diagonalización de la matriz ampliada y de las soluciones del sistema.

### 9.17 Elecciones

#### Presentación del problema

El programa representa gráficamente, en forma de un diagrama de barras, los resultados de las elecciones.

Para ello, el usuario debe introducir el porcentaje de votos obtenido por los partidos señalados en el programa : PSOE, AP, CIU, PCE y otros.

#### Análisis del problema

Dada la sencillez del problema podemos prescindir del análisis del mismo.

```

10 REM #MIIIS# -ELECCIONES-
20 COLOR 15,1,1:SCREEN 0:WIDTH 40:KEY OFF:CLS
30 PRINT " PROGRAMA PARA LA REPRESENTACION GRAFICA"
40 PRINT " DE RESULTADOS ELECTORALES."
50 PRINT:PRINT " PROF. DR. W. VOSS, 1984"
60 LOCATE 8,5:PRINT "-----"
70 DIM F(5),G(5)
80 LOCATE 0,8:PRINT "INTRODUCE LOS RESULTADOS ELECTORALES
(%)"
90 LOCATE 12,12:INPUT "PSOE : ";X:F(1)=X
100 LOCATE 12,14:INPUT "AP : ";X:F(2)=X
110 LOCATE 12,16:INPUT "CDC : ";X:F(3)=X
120 LOCATE 12,18:INPUT "PCE : ";X:F(4)=X
130 LOCATE 12,20:INPUT "OTROS : ";X:F(5)=X
140 S=0
150 FOR I=1 TO 5
160 S=S+F(I)
170 NEXT I
180 IF S=100 THEN 190 ELSE CLS:LOCATE 7,9:PRINT "ERROR EN L
A INTRODUCCION":LOCATE 11,22:PRINT "PULSE UNA TECLA":A$=INP
UT$(1):CLS:GOTO 80
190 CLS
200 FM=0
210 FOR I=1 TO 5
220 IF F(I)>FM THEN FM=F(I)
230 NEXT I
240 FOR I=1 TO 5
250 G(I)=(13/FM)*F(I)
260 G(I)=FIX(G(I)+.5)
270 NEXT I
280 PRINT " R E S U L T A D O S"
290 PRINT "
300 LOCATE 12,5:PRINT "PSOE : ";F(1)
310 LOCATE 12,8:PRINT "AP : ";F(2)
320 LOCATE 12,11:PRINT "CDC : ";F(3)
330 LOCATE 12,14:PRINT "PCE : ";F(4)
340 LOCATE 12,17:PRINT "OTROS : ";F(5)
350 LOCATE 12,22:PRINT "PULSE UNA TECLA"
360 A$=INPUT$(1):CLS
370 REM GRAFICO

```

```
380 FOR J=1 TO 5
390 B=17:A=17-G(J)
400 FOR I=A TO B
410 LOCATE (J-1)*B+2,I:PRINT "*"
420 LOCATE (J-1)*B+3,I:PRINT "*"
430 LOCATE (J-1)*B+4,I:PRINT "*"
440 NEXT I
450 NEXT J
460 FOR I=0 TO 39
470 LOCATE I,17:PRINT "#";
480 NEXT I
490 LOCATE 0,19:PRINT " PSOE      AP      CDC      PCE      OTR
OS"
500 LOCATE 11,22:PRINT "FIN DEL PROGRAMA":END
```

Lista de variables:

---

- A = Línea de inicio del gráfico
- B = Línea final del gráfico
- F = Porcentajes
- FM = Valor máximo
- G = Porcentajes transformados
- I = Índice variable
- J = Índice variable
- S = Campo para sumas
- X = Campo para introducir datos

Descripción del programa:

---

- Línea 10-60 : Título y aclaraciones
- Línea 70 : Dimensionados
- Línea 80-130 : Introducción de las Informaciones  
INPUT
- Línea 140-180 : Prueba
- Línea 190 : Borrado de la pantalla
- Línea 200-230 : Búsqueda del porcentaje mayor
- Línea 240-270 : Conversión de los porcentajes de  
manera que el valor máximo aprove-  
che el espacio disponible en la  
pantalla
- Línea 280-350 : Impresión de las informaciones
- Línea 360-450 : Impresión del gráfico
- Línea 460-480 : Impresión de una línea horizontal
- Línea 490 : Impresión de los textos PSOE, AP, CDC,  
PCE, OTROS.
- Línea 500 : Final del programa.

Epílogo

---

En los capítulos precedentes hemos intentado mostrar cómo pueden resolverse problemas típicos de la escuela con programas BASIC relativamente sencillos.

Desearía insistir en que los ejemplos escogidos tienen únicamente carácter de muestra, y que no se le ha dado concedido especial importancia a los programas óptimos.

El lector debe comprender rápidamente el funcionamiento de los programas. De este modo será capaz de encontrar la vía de resolución de otros problemas, que frecuentemente se asemejarán a los que aquí presentamos.

Lo fundamental a la hora de utilizar un ordenador no es el ordenador en sí, sino la capacidad que demuestra el usuario para tratar determinados problemas, de forma que pueda ofrecerle al ordenador el tipo de resolución más adecuada para un programa.

No es necesario que un programa funcione perfectamente a la primera (aun cuando esto, naturalmente, sea muy positivo); el usuario debe aprender cómo estructurar mentalmente un problema dado, para que el ordenador pueda resolverlo :

Voss	Capítulo 10	: Epílogo	Página
	Punto	- : -	368
Colegio			

La causa es que el ordenador no sirve para nada si nosotros, los usuarios, no conocemos la vía de resolución con anterioridad.

Si alguna vez ocurriese que un programa no funciona tal y como nosotros lo habíamos previsto, informándonos mediante un mensaje de error y la interrupción del programa, siempre nos ayudará el echar una ojeada al manual del ordenador.

A aquellos lectores, a los que los ejemplos presentados les han parecido demasiado sencillos, les indicaremos que este libro ha sido concebido en primera instancia para principiantes del BASIC y que no hemos querido superar sus posibilidades a través de una complejidad excesiva de la obra. Para consuelo de los primeros, queremos anunciar la pronta publicación de un segundo tomo de esta obra, donde se estudiarán problemas más complejos, especialmente de tipo matemático.

Voss	Índice general	Página
		369
Colegio		

## A

ABS	25
Adición	23
Amortización de una hipoteca	262
Análisis del problema	14, 41
Argumento	24
Arquímedes	151
Arte por computadora	138, 143
ASCII	96
Asignación de valores	22

## B

BASIC	7, 17
Bifurcación	28
Bifurcación del programa	28
Biología	347
Bit	211
Bucle	32
Bucle del programa	32
Bucle sin fin	36

C

Cadena	11, 18, 30
Cálculo	18
Cálculo de Intereses	256
Cálculo de porcentajes	80
Cálculos estequiométricos	116
Campo	10
Capitales	238
Carácter especial	9
Caracteres gráficos	96
Cargar	34
Cassette	13
CHR\$	25, 36, 97
Cifra	18
Círculo	299
CIRCLE	136
Código ASCII	96
COLOR	134, 135
Comando	12, 20
Combinaciones	336
Comprobación de números primos	59
Comunidades Autónomas	238
Consulta	28
CONT	36
Contaminación ambiental	223
Corrección	24
Crecimiento exponencial	212
Crecimiento limitado	217
COS	25, 309
Coseno	309

D

Dado	85
DATA	99
Datos	8, 12
Descripción del programa	41
Diagrama de flujo	14, 41
DIM	30
Dimensionado	100
Dinamómetro	144
Diskette	13
Distribución binomial	347
Distribución de frecuencias	273
Distribución normal	320, 358
División	23
DRAW	137

E

E	74
Ecología	211
Economía	255
Ecuación de segundo grado	68
Ecuación química	109
Elecciones	376
Elemento químico	122
Elipse	316
END	18
Error en la introducción	12
Escritura en clave	203
Espacio de memoria	100
Estequiometría	116
EXP	25

Voss	Indice general	Página
		372
Colegio		

F

Fichero	11
FILES	35
Física	131
Forma exponencial	92
FOR...NEXT	32
Función	24, 25
Funcionamiento	12

G

Gauss	320
Geografía	231
GOSUB	180
GOTO	30
Grabar	34
Gráfico	132, 293
Gráfico de alta resolución	132
Gráfico de bloque	132, 158
Gráfico normal	132, 158

H

Hipoteca	262
Historia	231

Voss	Indice general	Página
		373
Colegio		

IF...THEN	
Importe bruto	80
Importe del Impuesto	80
Importe neto	80
Impresión	12
Impresión del resultado	17
Impresora	13
Indexado doble	101
Índice de columna	101
Índice de línea	101
Informaciones	8
INPUT	26, 98
INPUT\$(n)	143
Instrucción de programa	11, 12
Instrucción BASIC	95, 179
Instrucción gráfica BASIC	135
INT	26, 98
Interés	256
Introducción	12
Introducción de datos	99
Introducción de informacione s	26

Voss	Indice general	Página
Colegio		374

LEN	204
Lengua	179
Lengua extranjera	179
Lente	163
LET	23, 98
Letra	9
Ley de Ohm	173
LINE	136
Línea en blanco	18
LIST	20
Lista de variables	41
LOAD	34
LOCATE	21
LOG	25, 313
Logaritmo	313
Loop	32
Lotería	342

#### M

Matemáticas	39, 291
m.c.m.	50
M.C.M	50
Media aritmética	269
Memoria externa	13, 34
Menú	174
Modo de texto	132
Molécula de agua	103
Movimiento pendular	157
Multiplicación	23

Voss	Indice general	Página
Colegio		375

#### N

NEW	21
NEXT	32
Nombre	10
Nombre de la función	24
Nombre de la variable	18, 22
Número	18
Número elevado al cuadrado	306
Número aleatorio	85
Número de línea	28
Número e	74
Número primo	59

#### O

Operación de cálculo	23
----------------------	----

#### P

Países	245
Pantalla	13, 31
Paralelogramo de fuerzas	168
Paso de trabajo	41
Péndulo	157
Permutaciones	332
Pitágoras	41



Voss	Indice general	Página
Colegio		376

P

Poder económico	282
Posición de memoria	10
Potenciación	23
Precisión de los cálculos	92
Presentación del problema	41
PRESET	136
Principio EPS	42, 152
PRINT	18
Proceso de datos	8
Programa	41
Programación de gráficos	132
Programa de sorteo	207
PSET	136
Punto	297, 298

Química	95
---------	----

R

Raíz cuadrada	306
Raya	136
Reacción química	109
READ	99
Recta	296
Red	293
Registro	11

Voss	Indice general	Página
Colegio		377

R

READ	99
Regla de tres	81
Regresión	324
REM	37
RND	25, 85
Reproducción óptica	163
Resolución de reglas de tres	80
RESTORE	102
Resultado	41
RETURN	181
RUN	19

S

Salto	28
Salto condicional	28
Salto incondicional	28
SAVE	34
SCREEN	133, 134
Seno	303, 309
Sentencia	17
Símbolo	9
SIN	25
Sinusóide	303
Sistema operacional	12
Sistema periódico	122
Sistema de ecuaciones	358-362
SQR	25
Statement	17
STEP	33
STOP	36
Substracción	23

Voss	Índice general	Página
Colegio		378

T

Tangente	309
Tecla BREAK	36
Teclado	13, 96
Tecla RETURN	24
Técnica de los menús	174
Tendencia	325
Teorema de los límites	358
Test de vocablos	196
Test de vocablos ingleses	196
Tratamiento	12
Triángulo rectángulo	42

U

Unidad central	13
----------------	----

V

Valor	10
Valor inicial	32
Valor final	32
Variable	10, 22
Variable de cadena	22, 30, 101
Variable indexada	100
Variable numérica	22

ss	Índice general	Página
Colegio		379

V

Verbos	182
Verbos ingleses	182
Verbos irregulares	182
Vocablos	189
Vocablos ingleses	196
Vocablos franceses	189

Z

Zona de la pantalla	20
---------------------	----



ROBOTICA PARA SU COMMODORE 64, 230 pág.  
P.V.P. 2.800,- ptas.

En el libro de los robots se muestran las asombrosas posibilidades que ofrece el CBM 64, para el control y la programación, presentadas con numerosas ilustraciones e intuitivos ejemplos. El punto principal: Cómo puede construirse uno mismo un robot sin grandes gastos. Además, un resumen del desarrollo histórico del robot y una amplia introducción a los fundamentos cibernéticos. Gobierno del motor, el modelo de simulación, interruptor de pantalla, el Port-Usuario cómodo del modelo de simulación, Sensor de infrarrojos, concepto básico de un robot, realimentación unidad cibernética, Brazo prensor, Oír y ver.




TODO SOBRE EL FLOPPY 1541, 482 pág.  
P.V.P. 3.200,- ptas.

La obra Standard del floppy 1541, todo sobre la programación en disquettes desde los principiantes a los profesionales, además de las informaciones fundamentales para el DOS, los comandos de sistema y mensajes de error, hay varios capítulos para la administración práctica de ficheros con el FLOPPY, amplio y documentado Listado del Dos. Además un filón de los más diversos programas y rutinas auxiliares, que hacen del libro una lectura obligada para los usuarios del Floppy.



MANTENIMIENTO Y REPARACION DEL FLOPPY 1541,  
200 pág.  
P.V.P. 2.800,- ptas.

Saberse apañar uno mismo, ahorra tiempo, molestias y dinero, precisamente problemas como el ajuste del floppy o reparaciones de la platina se pueden arreglar a menudo con medios sencillos. Instrucciones para eliminar la mayoría de perturbaciones, listas de piezas de recambio y una introducción a la mecánica y a la electrónica de la unidad de disco, hay también indicaciones exactas sobre herramientas y material de trabajo. Este libro hay que considerarlo en todos sus aspectos como efectivo y barato.



PEEKs y POKES, 177 pág.  
P.V.P. 1.600,- ptas.

Con importantes comandos PEEK y POKE se pueden hacer también desde el Basic muchas cosas, para las que se necesitarían normalmente complejas rutinas en lenguaje máquina. Este libro explica de manera sencilla el manejo de PEEKs y POKEs. Con una enorme cantidad de POKEs importantes y su posible aplicación. Para ello se explica perfectamente la estructura del Commodore 64: Sistema operativo, interpretador, página cero, apuntadores y stacks, generador de caracteres, registros de sprites, programación de interfaces, desactivación del interrupt. Además una introducción al lenguaje máquina. Muchos programas ejemplo.



EL MANUAL DEL CASSETTE, 190 pág.  
P.V.P. 1.600,- ptas.

Un excelente libro, que le mostrará todas las posibilidades que le ofrece su grabadora de cassettes. Describe detalladamente, y de forma comprensible, todo sobre el Datassette y la grabación en cassette. Con verdaderos programas fuera de serie: Autostart, Catálogo (¡busca y carga automáticamente!), backup de y a disco, SAVE de áreas de memoria, y lo más sorprendente: un nuevo sistema operativo de cassette con el 10-20 veces más rápido FastTape. Además otras indicaciones y programas de utilidad (ajuste de cabezales, altavoz de control).




MSX PROGRAMAS Y UTILIDADES, 1985, 194 pág.  
P.V.P. 2.200,- ptas.

El libro contiene una amplia colección de importantes programas que abarcan, desde un desensamblador hasta un programa de clasificaciones deportivas. Juegos superemocionantes y aplicaciones completas. Los programas muestran además importantes consejos y trucos para la programación. Estos programas funcionan en todos los ordenadores MSX, así como en el SPECTROVIDEO 318 328. EXTRACCIÓN DEL CONTENIDO: Volcado memoria hexadecimal. Editor gráficos. Editor de sonido. Escritura de ordenador. Lista referencia de variables. Calendario. Desensamblador. ADMINISTRACIÓN de una colección de discos L.P. HOLLOW - JUEGO DE LAS CEREZAS. DIAGRAMAS DE BARRAS. TABLAS DEPORTIVAS.



ZX SPECTRUM CONSEJOS Y TRUCOS, 211 pág.  
P.V.P. 2.200,- ptas.

Una interesante colección de sugestivas ideas y soluciones para la programación y utilización de su ZX ESPECTRUM. Aparte de muchos peeks, pokes y USRs hay también capítulos completos para, entre otros, entrada de datos asegurado sin bloqueo de ordenador, posibilidades de conexión y utilización de microdrives y lápices ópticos programas para la representación de diagramas de barra y de tarta, el modo de utilizar óptimamente ROM y RAM.




CPC-464 EL LIBRO DEL COLEGIO  
P.V.P. 2.200,- ptas.

Escrito para alumnos de los últimos cursos de EGB y de BUP, este libro contiene muchos programas para resolver problemas y de aprendizaje, descritos de una forma muy compleja y fácil de comprender. Teorema de Pitágoras, progresiones geométricas, escritura cifrada, crecimiento exponencial, verbos irregulares, igualdades cuadráticas, movimiento pendular, estructura de moléculas, cálculo de interés y muchas cosas más.




ZX SPECTRUM EL MANUAL ESCOLAR  
P.V.P. 2.200,- ptas.

Escrito para alumnos de los últimos cursos de EGB y de BUP, este libro contiene muchos programas para resolver problemas y de aprendizaje, descritos de una forma muy completa y fácil de comprender. Teorema de Pitágoras, progresiones geométricas, escritura cifrada, crecimiento exponencial, verbos irregulares, igualdades cuadráticas, movimiento pendular, estructura de moléculas, cálculo de interés y muchas cosas más.



MSX\_EL MANUAL ESCOLAR  
P.V.P. 2.800,- ptas.

Escrito para alumnos de los últimos cursos de EGB y de BUP, este libro contiene muchos programas para resolver problemas y de aprendizaje, descritos de una forma muy completa y fácil de comprender. Teorema de Pitágoras, progresiones geométricas, escritura cifrada, crecimiento exponencial, verbos irregulares, igualdades cuadráticas, movimiento pendular, estructura de moléculas, cálculo de interés y muchas cosas más.



CPC-464 CONSEJOS Y TRUCOS  
P.V.P. 2.200,- ptas.

Ofrece una colección muy interesante de sugerencias, ideas y soluciones para la programación y utilización de su CPC-464: Desde la estructura del hardware, sistema de funcionamiento - Tokens Basic, dibujos con el joystick, aplicaciones de ventanas en pantalla y otros muchos interesantes programas como el procesamiento de datos, editor de sonidos, generador de caracteres, monitor de código máquina hasta listados de interesantes juegos.



MSX GRAFICOS Y SONIDOS, 250 pág.  
P.V.P. 2.800,- ptas.

Las computadoras MSX no sólo ofrecen una relación precio/rendimiento sobresaliente, sino que también poseen unas cualidades gráficas y de sonido excepcionales. Este libro expone las posibilidades de los MSX de forma completa y fácil. El texto se completa con numerosos y útiles programas ejemplo.



METODOLOGIA DE LA PROGRAMACION  
P.V.P. 2.200,- ptas.

El primer libro recomendado para escuelas de enseñanza de informática y para aquellas personas que quieren aprender la programación. Cubre las especificaciones del Ministerio de Educación y Ciencia para Estudios de Informática. Realizado por un alto mando del ejército Español, un Dr. Ingeniero y Diplomado en Informática y profesor de la UNED y por un oficial técnico especialista en informática de gestión. Utilizado en todos los institutos politécnicos del ejército español. Es un seguro candidato a ediciones en lengua inglesa, alemana y francesa. Es el primer libro que introduce a la lógica del ordenador. Es un elemento de base que sirve como introducción para la programación en cualquier otro lenguaje. No se requieren conocimientos de programación ni siquiera de informática. Abarca desde los métodos de programación clásicos a los más modernos.



MANUAL ESCOLAR PARA SU COMMODORE 64, 351 pág.  
P.V.P. 2.800,- ptas.

Este libro, escrito especialmente para escolares de grado medio y superior, contiene muchos interesantes programas de aprendizaje para solucionar problemas, descritos detalladamente y de manera fácilmente comprensible. Facilitan un aprendizaje intensivo y ameno, con, entre otros, los siguientes temas: Teorema de pitágoras, progresiones geométricas, palanca mecánica, crecimiento exponencial, verbos irregulares, ecuaciones de segundo grado, movimientos de péndulo, formación de moléculas, aprendizaje de vocablos, cálculo de interés y su capitalización. Una corta repetición de los elementos BASIC más importantes y una introducción a los rasgos esenciales del análisis de problemas, entre otros, completan el conjunto.



64 EN EL CAMPO DE LA TECNICA Y LA CIENCIA, 296 pág.  
P.V.P. 2.800,- ptas.

Ofrece un campo fascinante y amplio de problemáticas científicas. Para esto el libro contiene muchos listados interesantes: Análisis de Fourier y síntesis, análisis de redes, exactitud de cálculo, formateado de números, cálculo del valor PH, sistemas de ecuaciones diferenciales, modelo ladrón presa, cálculo de probabilidad, medición de tiempo, integración, etc.



LENGUAJE MAQUINA PARA  
COMMODORE 64, 1984, 201 pág.  
P.V.P. 2.200,- ptas.

¡Por fin una introducción al código máquina fácilmente comprensible! Estructura y funcionamiento del procesador 6510, introducción y ejecución de programas en lenguaje máquina, manejo del ensamblador, y un atractivo muy especial: ¡un simulador de paso a paso escrito en BASIC!



LENGUAJE MAQUINA PARA AVANZADOS  
CBM 64, 1984, 206 pág.  
P.V.P. 2.200 ptas.

¿Ud. ha logrado iniciarse en código máquina? Entonces el «nuevo English» le enseñará cómo convertirse en un profesional. Naturalmente con muchos programas ejemplo, rutinas completas en código máquina e importantes consejos y trucos para la programación en lenguaje máquina y para el trabajo con el sistema operativo.

**RESPUESTA  
COMERCIAL**

F.D. Autorización 6975  
(B.O. de Correos N.º 80 de 26-7-85)

**HOJA PEDIDO  
DE LIBRERIA**

NO NECESITA  
SELLOS

A franquear  
en destino

**FERRE MORET, S.A.**

Apartado N.º 551. F.D.  
08080 BARCELONA

**RESPUESTA  
COMERCIAL**

F.D. Autorización 6975  
(B.O. de Correos N.º 80 de 26-7-85)

**HOJA PEDIDO  
DE LIBRERIA**

NO NECESITA  
SELLOS

A franquear  
en destino

**FERRE MORET, S.A.**

Apartado N.º 551. F.D.  
08080 BARCELONA

## **EL CONTENIDO:**

Escrito para alumnos de los últimos cursos de EGB y de BUP, este libro contiene multitud de programas para resolver problemas y de aprendizaje, descritos de una forma muy completa y fácil de comprender. Proporcionan un aprendizaje intensivo, a la vez que divertido, con el MSX.

Del contenido:

- Teorema de Pitágoras
- Progresiones geométricas
- Escritura cifrada
- Crecimiento exponencial
- Verbos irregulares
- Igualdades cuadráticas
- Movimiento pendular
- Estructura de moléculas
- Repaso de vocablos
- Cálculo de intereses

Una pequeña panorámica sobre los fundamentos del tratamiento informático, un somero repaso a los principales elementos del BASIC y una introducción en los aspectos básicos del análisis de problemas completan el libro.

## **ESTE LIBRO HA SIDO ESCRITO POR:**

Werner Voss, profesor de estadística en la Universidad de Bochum. Multitud de publicaciones dentro del campo de la estadística y del tratamiento informático.